
前言

2014年年底的某一天，有一个网名叫潇子的人通过Excel效率手册交流群与我私聊，让我做她的老师。这么多年来叫我老师的人很多，但我实际上并没有收任何人为徒弟，也正因为如此，我委婉地拒绝了她。不过潇子对自己很有信心，说我一定会收她为徒。通过跟她的交流，我对她越来越有好感，所以就同意了，她是我目前唯一的徒弟。可惜说来惭愧，一直以来都很少教她Excel技能，反倒把她引上另外一条不归路：理财之路。那时我除了是Excel爱好者外，同时也是理财爱好者，每天都在接触理财，给她灌输了一些理财知识。在这期间也断断续续地教了她一些Excel技能，但是比较零散。太过零散的知识学起来很费劲，看似学了很多，但实际上学过以后就忘记了。其实学习Excel需要关联性学习，就是同一个系列的知识一起学习，然后举一反三，同时学会后通过练习来进行巩固，这样才能学得牢。为此，我准备花半年的时间整理总结，逐步教潇子Excel中的函数与公式，做一个名正言顺的老师。

潇子是一个刚毕业的女大学生，非常聪明。很多技能只要稍微提一下，她就能很好地掌握，领悟能力很好，我这个做老师的就不用发愁，教起来比较轻松。

我是卢子，是Excel效率手册丛书的作者，同时也是Excel与理财爱好者。

读者对象

本书适用于各类需要使用Excel函数和公式高效办公的职场人士，包括：


	需要处理大量数据的如市场、销售、财务、人力资源等方面的分析人员；

	各类市场研究人员、金融研究人员、管理顾问等专业人士；

	希望能高效、快速解决工作表格数据的办公室白领人士；

	为进入职场做准备的学生朋友。



从软件掌握程度而言，本书需要读者对Excel具有初级到中级的掌握程度。读者应该具有基本的Excel知识，知道如何创建一个图表，Excel软件各功能操作也要有所了解。

阅读指南

本书内容来自笔者个人的经验、经历和观点，因此，笔者在本书中选择用自己的语言来讲述个人以及很多网友关于Excel函数和公式方面遇到的实际问题。笔者力图以轻松、自然的语言和案例使行文简洁易读，减少读者对专业知识的理解难度，增加读者学习的兴趣和热情。全书共包括四篇，各篇内容按照读者学习流程循序渐进、环环相扣。

第一篇　初体验（1～6章）。函数其实并不难，比你想象中的更容易，即使不会英语，照样可以学好函数。在本篇中，你可以学习到Excel常用函数的用法。

第二篇　深了解（7～11章）。函数与公式是非常神奇的东西，第一篇已经讲解了单个函数的用法，本篇将介绍不同函数之间的嵌套组合，来解决各种各样的疑难。

第三篇　活运用（12～14章）。本篇将给出实际工作中大量可以用Excel函数与公式方便解决的问题。通过本篇根据实际情况来解决具体方案的练习，能给你提供解决问题的正确思路。

第四篇　解疑难（15～16章）。笔者甄选在各大Excel论坛、Excel交流群、淘宝上等帮人解答的经典Excel函数与公式的问题，让你的工作更加高效、省时、流畅，不加班，早回家！

如何使用本书？

1．疑难和案例
 　本书在每章开始都引出生活、工作中简单的疑难和案例，并给出用Excel函数与公式快速解答的妙法，让零基础的读者了解相关知识，顺利入门。

2．原理分析
 　针对每章介绍的函数和公式原理进行分析，帮助读者更直观地理解函数和公式的实际意义。

3．知识扩展
 　每章会提供对核心知识点的内容扩展，让读者了解用其他方法如何解决本章问题或难度升级的实际工作问题，让读者能举一反三。

4．为了便于读者阅读，全书还穿插着一些注意、提示等小贴士，如下所示：


提示：
 通常是一些贴心的提醒，让读者加深印象，提供建议或者解决问题的方法。


注意
 　提出学习过程中需要特别注意的一些知识点和内容，或者相关信息。


5．课后练习
 　每章最后会提供专门的测试习题，供读者检验所学知识是否牢固掌握。

版本说明：本书的操作版本为Excel 2013，本书介绍的内容主要通过Excel常用功能实现，对于使用最新版本和旧版本的读者不用担心。

第一篇　初体验

Excel中内置的函数就有几百个，很多人看到这么多的函数头都大了，其实只要掌握十分之一就是高手，但对于普通人来说，学会二十个左右即可完成绝大部分的工作。抓住重点，才能事半功倍。

函数难学吗？

不会英语能学好函数吗？

函数其实并不难，比你想象中更容易，即使不会英语照样可以学好函数。为什么这样说呢？因为笔者本身就是从菜鸟走过来的，当初什么都不懂，还有笔者本身英语水平就很差，在笔者的印象中很少英语考试能及格。像笔者这种水平的人都能学好函数，你有什么理由学不好呢？只要你对自己多一点信心，就一定可以学会！来，让我们一起来学习函数，学完本篇，相信你会有意想不到的收获。在本书中，你不但可以学习到Excel常用函数的用法，还可以学到一些理财的知识！



第1章　求和函数


 1.1　统计总销售量（SUM函数）


疑难：
 这是某集团公司2014年各分公司每月销售量，如何统计总销售量呢？

[image: ]



妙解：
 下面的情景相信很多人见过：

拿起计算器一个个地敲进去进行求和，然后把结果输入在Excel中。

[image: ]


当然有的人也懒得拿计算器，把Excel当成了计算器，输入以下的公式：


    =C2+C3+C4+C5+C6+C7+C8+C9+C10+C11


[image: ]


这样确实能够解决问题，但如果有几千个数据，你是不是得敲晕，所以聪明一点的人就选择了下面这种做法。选择区域C2:C11单元格，然后查看状态栏的求和，最后将结果输入在Excel中，这样明显比上面2种方法高明，即使再多的数据也能瞬间统计完。

[image: ]


不过不管是哪种方法，只要其中的数据改变了，就得重新统计，相当麻烦。其实Excel预先帮我们定义好了一个求和函数SUM，它可以让汇总变得更加轻松。

单击C12单元格，在“公式”选项卡里，单击“自动求和”按钮，在默认情况下就帮你选好区域进行求和，简单快捷。这样即使你修改其中的数据也可以立即更新汇总结果。

[image: ]


原理分析

本例中的求和区域是C2:C11单元格，使用SUM函数对销售量进行求和。

SUM函数语法如下：


    SUM(数字1,[数字2],…)


参数中数字可以是262之类的数字，也可以是C2之类的单元格引用或C2:C11之类的单元格范围，参数1为必须参数，参数2为可选参数。

如下面的公式就是对C2和38进行求和。


    =SUM(C2,38)


知识扩展

我们将鼠标放在“自动求和”按钮上，会出现一个提示，也就是说求和其实可以用【Alt】+【=】快捷键实现快速对上方连续单元格区域汇总。

课后练习

在做基金定投的时候，笔者会记录下每个月的净值、定投金额跟份额，下图是一个先跌后涨的趋势表，如何计算到最后总共有多少金额？总金额就是最后一个净值（0.9）乘以所有份额的累计数。这样一来就知道你的基金定投一年盈亏多少钱了。

 1.2　统计上海公司的销售量（SUMIF函数）


疑难：
 现在只是想知道上海公司的销售量，如何统计呢？

[image: ]


SUM函数是对区域进行求和，现在多了一个公司名称条件，这样SUM函数就暂时派不上用场。我们首先想到的方法就是用眼睛看，然后逐个相加：


    =C5+C9+C10

妙解：
 这样做显然不太妥当，数据一多肯定容易出错，即使不出错，效率也非常低。可能你也会想到借助筛选进行求和。


STEP 01
 　单击A1单元格，在“数据”选项卡中单击“筛选”按钮。

STEP 02
 　单击“公司名称”的筛选按钮，取消“（全选）”的复选框。

STEP 03
 　勾选“上海公司”，单击“确定”按钮。选择筛选出来的区域，在状态栏即可看到对上海公司销售量的求和。

这种方法明显比第一种好很多，而且准确率更高。其实Excel中提供了一个对条件求和的函数SUMIF，借助这个函数就能轻易对数据进行条件求和。

在F2单元格输入如下公式，按【Enter】键结束。


    =SUMIF(B:B,E2,C:C)


[image: ]


用这个公式有一个好处就是不管对多少公司求和，只要下拉公式即可轻松汇总起来。

原理分析

本例中的条件区域为B列，条件为E2单元格，求和区域为C列，借助SUMIF函数进行单条件求和，求出上海公司的销售量。

SUMIF函数语法如下：


    SUMIF(条件区域,条件,[求和区域])


假如现在要对C列大于500的销售量进行求和，正常使用下面的公式：


    =SUMIF(C:C,">500",C:C)


这时会发现一个问题，如果条件区域跟求和区域是一样的，在这种情况下，求和区域可以省略，可将公式改成：


    =SUMIF(C:C,">500")


知识扩展

对于刚接触函数的朋友来说，并不知道有这么一个函数的存在，那怎么才能知道使用什么函数来进行条件求和呢？

在“公式”选项卡中单击“插入函数”按钮，弹出“插入函数”对话框，在“搜索函数”的文本框中输入求和，单击“转到”按钮。这时会出现相关函数，比如第一个SUMIF，下面就有对这个函数的介绍：“对满足条件的单元格求和”，刚好符合我们的要求，假如不符合就接着看下一个函数，直到找到合适的函数为止。

如果想知道这个函数的详细语法，就单击“确定”按钮，弹出“函数参数”对话框。这时可以看到对第一参数Range的说明：“要进行计算的单元格区域”。

将鼠标放在第二参数Criteria，这时可以看到对第二参数Criteria的说明：“以数字、表达式或文本形式定义的条件”。

接着将鼠标放在第三参数Sum_range，这时可以看到对第三参数Sum_range的说明：“用于求和计算的实际单元格。如果省略，将使用区域中的单元格”。

看到这里我们应该对这个函数有了初步的了解，如果还不是很懂的话，可以单击“有关该函数的帮助”的超链接。在弹出的SUMIF函数帮助中，有对这个函数的详细说明，正常的话，单个函数多看几遍就能够了解这个函数的语法和使用方法。

“Excel帮助”其实就相当于一本字典，对基本功能都有一个介绍，哪里不懂的话，就随时进行搜索学习。

课后练习

如下图所示，每一家基金公司旗下都有不同基金，如何求宝盈公司旗下所有基金产品的总规模呢？也就是只要包含“宝盈”这2个字，就把规模累加进去。

 1.3　统计销售量在500至700的和（SUMIFS函数）


疑难：
 前面说过借助SUMIF函数可以求销售量大于500的和，现在是500至700这个区间，该如何做呢？


妙解：
 其实也就是要符合两个条件，一个条件是>=500，另一个条件是<=700。我们知道SUMIF函数只能单条件求和，不能多条件求和。

如果你是一个懂得举一反三又细心的人，应该很快就懂得如何操作，因为刚刚在查看SUMIF函数的Excel帮助时，Excel帮助中已经有提到多条件求和，Excel帮助其实也挺贴心的。

[image: ]



注意
 　若要根据多个条件对若干单元格求和，请参阅SUMIFS函数。


单击“SUMIFS函数”这个超链接，就可以看见SUMIFS函数的Excel帮助。

也就是说借助SUMIFS函数就可以搞定这个问题，在单元格输入以下的公式：


    =SUMIFS(C:C,C:C,">=500",C:C,"<=700")

原理分析

在本例中借助SUMIFS函数可以实现多条件求和的特点，设置条件1为">=500"和条件2为"<=700"，对C列的销售量进行求和。

SUMIFS函数语法如下：


    SUMIFS(求和区域,条件区域1,条件1,[条件区域2],[条件2],…)


SUMIFS跟SUMIF的参数顺序有点不同，SUMIFS求和区域在第一参数，而SUMIF函数求和区域在第三参数，这个需要特别注意一下。

知识扩展

SUMIFS函数虽然可以对多个条件进行求和，但条件太多的时候还是很容易混乱。如现在对符合月份、销售人员、城市、商品4个条件进行销售量求和，如果使用SUMIFS公式为：


    =SUMIFS(E:E,A:A,G2,B:B,H2,C:C,I2,D:D,J2)

在这种情况下DSUM函数就派上用场了，在单元格输入如下公式：


    =DSUM(A1:E20,5,G1:J2)


这样看起来非常简洁，又不容易出错。

DSUM函数语法如下：


    DSUM（单元格区域,第几列号(名),条件区域）


也可以改成：


    =DSUM(A1:E20,K1,G1:J2)


课后练习

基金产品可以分成股票基金、货币基金等，如下图所示，现在要统计货币基金且金额大于1000的总金额是多少？

 1.4　统计总销售金额（SUMPRODUCT函数）


疑难：
 根据销售明细表如何统计总销售金额？

妙解：
 我们首先想到的就是在F列建立一个辅助列，先获取金额小计，最后再求和。


STEP 01
 　在F2单元格输入以下公式，并下拉至F11单元格。


    =D2*E2

STEP 02
 　在F12单元格输入以下求和公式。


    =SUM(F2:F11)


在这里会多一个辅助列，其实还有更好的办法，就是直接借助SUMPRODUCT函数进行统计：


    =SUMPRODUCT(D2:D11,E2:E11)


原理分析

借助SUMPRODUCT函数对两个区域先乘积，后求和。

SUMPRODUCT函数语法：


    SUMPRODUCT(区域1, 区域2,...)

知识扩展

SUMPRODUCT函数的作用远远不止如此，它是一个非常强大的函数，前面的几个例子都可以通过SUMPRODUCT函数来完成。

统计总销售量，当只有一个参数的时候，作用就跟SUM函数一样。


    =SUMPRODUCT(C2:C11)


统计上海公司的销售量


    =SUMPRODUCT(($B$2:$B$11=E2)*$C$2:$C$11)


统计销售量在500至700的和


    =SUMPRODUCT((C2:C11>=500)*(C2:C11<=700)*C2:C11)


你也许会奇怪，这个语法Excel帮助没有提到啊。

Excel帮助只是告诉我们最基础的用法，很多函数的高级用法都是在运用过程中慢慢摸索出来的。在以后的章节你会看到更多Excel帮助中没有提到的用法，甚至看到Excel帮助也有出错的时候。现在你只要记住下面一个通用公式即可。

SUMPRODUCT函数条件求和的通用公式：


    =SUMPRODUCT((条件1)*(条件2)*…*求和区域)


在学习函数的过程中，你会发现其实很多问题都有N种解决的办法，就应了那句“条条大路通罗马”。学习过程中你就尽量想出越多种方法越好，实际工作中你就选择最熟悉的方法，以确保准确无误。

课后练习

我们在买基金之前，通常会关注很多只基金进行观察，但不一定有买，如果没买就是金额为零（不是数字的0），如下图所示，统计股票基金总共有多少金额？

第2章　计数函数


 2.1　统计上班天数（COUNTA函数）


疑难：
 这是某公司考勤记录表，有上班就打勾（√），没上班就空着，如何统计每个人的上班天数？

妙解：
 在B34单元格输入以下公式，并向右复制。


    =COUNTA(B3:C33)/2


原理分析

借助COUNTA函数统计非空单元格个数，因为每天上班包含上下午，一个√算半天，所以要求出总数再除以2。

COUNTA函数语法如下：


    COUNTA(值1,[值2],…)


COUNTA函数语法跟SUM函数一样。

知识扩展

现在要统计每个人有多少天没上班的话，该如何做？

在单元格输入公式，并向右复制。


    =COUNTBLANK(B3:C33)/2


COUNTBLANK函数语法跟COUNTA函数一样，作用就是统计区域中空单元格的个数，没上班就是空着，也就是空单元格。跟这2个函数类似的还有COUNT函数，就是统计区域中数字的个数。

如图所示现在要统计有多少个数字，就可以用下面的公式：

    =COUNT(A1:A7)


函数有的时候还是挺好学习的，因为同一类型的函数语法都差不多或者是一样，学一个就能举一反三。

课后练习

正常的话每一家基金公司成立3年以上就有3年的标准差，5年以上就有5年的标准差，时间不足的就没有标准差，标准差越小表现越稳定。如下图所示，判断是否全部满足3年或者5年，全部满足显示“是”，不满足则显示“否”。

 2.2　考勤统计（COUNTIF函数）


疑难：
 上一个例子的考勤表只有上班跟空白两种情况，而实际上可以细分为正常上班、迟到、事假等，现在如何一次统计每个人正常、迟到、事假的次数呢？

妙解：
 COUNT跟COUNTA两个函数都是统计个数，但没有按条件统计。我们知道SUM函数就是求和，SUMIF函数就是条件求和，也就是SUM+IF就能条件求和。同理，COUNT+IF也能条件计数，也就是COUNTIF函数。

在J2单元格输入以下公式，并向右和向下复制公式。


    =COUNTIF($B2:$I2,J$1)


原理分析

COUNTIF函数就是条件计数函数，跟SUMIF函数有点类似，就是少了一个求和区域而已。

COUNTIF函数语法：


    COUNTIF(条件区域,条件)


知识扩展

COUNTIF函数还有很多常用的例子，参数1的DATE就是区域，下面的用法为ExcelHome整理出来的，可以参考学习。

课后练习

股神巴菲特年化率收益大概24%，我们普通人能够做到年化率收益15%就已经不错了，牛市除外。如下图所示，2014年刚好是牛市初期，判断年化率收益大于15%的基金公司有几家。

 2.3　统计性别为男、学历为大专的人数（COUNTIFS函数）


疑难：
 下图为员工信息表，如何统计性别为男、学历为大专的人数呢？

妙解：
 前面的例子只有一个条件可以用COUNTIF，现在是2个条件暂时用不了。我们知道多条件求和可以用SUMIFS函数，也就是说多条件计数可以用COUNTIFS函数。

在单元格输入以下公式：


    =COUNTIFS(C2:C11,"男",E2:E11,"大专")


原理分析

COUNTIFS函数可以实现多条件计数，函数语法如下：


    COUNTIFS(条件区域1,条件1, 条件区域2,条件2,…)


知识扩展

COUNTIFS函数可以实现多条件计数，但有些情况下却会出错，如下面的例子。

员工考勤表是实际工作中需要做的事情，下图为10月份员工的一个考勤表，分别统计每人10月份的出勤情况。

其中的考勤符号：“A”正常上班；“K”旷工；“B”病假；“S”事假；“C”迟到；“W”外出出差。

在C14单元格输入以下公式，发现按下回车键后居然出错。


    =COUNTIFS($B$5:$B$10,$B14,$C$5:$Z$10,C$13)


碰到这种情况该如何解决呢？

我们在前面学习了SUMPRODUCT函数，知道这个函数可以进行条件求和，其实它还能进行条件计数，真有这么神奇吗？一起来见识下吧！

在C14输入以下公式，并向右和向下复制公式。


    =SUMPRODUCT(($B$5:$B$10=$B14)*($C$5:$Z$10=C$13))

在Excel2003中没有COUNTIFS跟SUMIFS函数，只要出现多条件必然想到SUMPRODUCT函数，所以这个函数非常受欢迎。在高版本的Excel软件中只有一些特殊情况下会用到。

SUMPRODUCT函数条件计数的通用公式为：


    =SUMPRODUCT((条件1)*(条件2)*…*(条件n))


课后练习

在挑选基金的时候，会比较在意基金经理、基金成立时间和基金管理人上任时间。正常的话会选择成立时间比较长，而且基金经理管理的时间也比较长的基金，老基金有以往业绩可以参考，比新基金稳妥点。如下图所示，判断有多少只基金的成立时间跟基金管理人上任时间是同一天。

第3章　逻辑函数


 3.1　猴子分桃（IF函数）


案例：
 宋国有一位养猴的老人，他可以理解猴子的意思，猴子也可以理解老人的心意。养猴的老人宁可减少他与家人的食物也要满足猴子的需求。不久，他家里的粮食缺乏了，他想限定猴子食物的数量，但又怕猴子不顺从自己，就先欺骗猴子说：“给你们桃子，早上三个，晚上四个，够吗？”猴子们都站了起来并且十分恼怒。他又说：“给你们桃子，早上四个，晚上三个，够了吧？”猴子们都非常高兴，然后一个个都趴在地上。

从这里就可以看到2个判断：

1．如果是早上就给3个桃子，否则（是晚上）就给4个。

2．如果是早上就给4个桃子，否则（是晚上）就给3个。

将数据输入Excel中，我们就可以借助IF函数替我们判断。

在C2单元格输入如下公式，并向下复制，即可得到每个时间段应该分多少桃子。


    =IF(B2="早上",3,4)
    =IF(B2="晚上",4,3)


原理分析

IF函数的作用就是进行逻辑判断，语法如下：

IF(条件,条件符合时返回的值,条件不符合时返回的值)

如第1个公式，条件就是判断是否等于早上，满足了就显示3，否则（等于晚上）就显示4。因为只是早晚需要分桃的问题而已，不是早上，必然是晚上。

为了更好地理解这个函数，再举一个简单的例子，假如国庆节只放假7天我就回家，没有的话我就待在公司。

假设F1单元格为放假的天数，就可以设置如下的公式。


    =IF(F1=7,"回家","待在公司")


知识扩展

猴子只要吃2顿饭就好，但人是要吃3顿饭才行。我们经常在外面吃饭，早饭花5元，午饭花15元，晚饭花10元。现在如果单独用1个IF函数没办法判断这么多条件，该如何处理呢？

我们先用IF函数做可以做到的事，先判断是否等于早上，如果是，显示5，不是，就先用一个其他的文本代替。


    =IF(B2="早上",5,"其他")


其他这个文本我们可以用IF函数进行判断获取。


    =IF(B2="中午",15,10)


在进行数学公式的时候，可以用替代法进行计算，比如数学公式为：


    Y

=X

+3
    X

=2


公式中的X我们就可以用2替代，从而计算出Y的值。同样的道理，Excel中的公式也可以用这样替代法，将其他的公式替代进去，从而获得新的公式。


    =IF(B2="早上",5,IF(B2="中午",15,10))


课后练习

在购买股票的时候仓位管理很重要，控制不好仓位，可能会导致损失惨重。雪球大V小小辛巴提出了5档买入法，也就是一只股票每下降一定的百分比或者金额时，就依次买入。假如东阿阿胶购买的起始价格为48元，占仓位的20%，每下降10元，补仓20%，也就是说38元的时候仓位为40%，到8元的时候仓位为100%。如何设置公式，获取每个价格的投资比例呢？

 3.2　旅游必备条件（AND函数）


案例：
 还没参加工作的时候，有的是时间，总想到处去玩，去云南、北京、西安……但这时因为没钱，只能是想想而已。参加工作以后，钱就有了，可是天天上班，没有时间出去玩。也就是说要去外面玩的机会还是比较少，需要同时满足2个条件：1.有钱；2.有时间。

现在只有同时满足有钱、有时间才能去外面玩，否则就不玩。

在C2单元格输入以下公式，并向下复制。


    =IF(AND(A2="有钱",B2="有时间"),"出去玩","不玩")


原理分析

AND函数的作用就是同时满足的意思，语法如下：


    AND(条件1,条件2,…)

只有当所有条件同时满足的情况下才显示TRUE，只要其中一个条件不满足都显示FALSE。

再举一个例子来说明，我们现在年轻人出门必带3件宝：手机、钱包、钥匙，只要少一样，心里都不会踏实，有的时候即使走到半路发现忘记手机都会回家去拿手机。


    =IF(AND(E2="手机",F2="钱包",G2="钥匙"),"安心出门","不自在")


知识扩展

有的时候我们会看到类似下面的公式，IF函数的第一参数是1或者0，而不是TRUE跟FALSE。他们究竟有什么联系呢？


    =IF(1,"男","女")
    =IF(0,"男","女")


在Excel中输入下面的公式：


    =--TRUE
    =--FALSE


分别得到1跟0，也就是说通过运算，逻辑值可以转换成数字1跟0。其实我们还可以直接进行逻辑值之间的乘法运算。


    =TRUE*TRUE


得到1


    =FALSE*FALSE


得到0

通过上面的运算，我们发现其实可以直接用*代替AND函数。


    =IF((A2="有钱")*(B2="有时间"),"出去玩","不玩")
    =IF((E2="手机")*(F2="钱包")*(G2="钥匙"),"安心出门","不自在")


全部满足的时候，乘积就是1，只要其中1个条件不满足，乘积就是0。1就返回第2参数，0就是返回第3参数。

课后练习

选好股票就是选好公司，首先是行业要好，有发展前景，然后是好公司，会赚钱。其次就是价格要低，这样持有才能挣更多的钱。中石油是世界500强，这个是好公司绝对没错，但如果你在最高点买入，那你到现在七八年过去了，还翻不了身。而巴菲特在中石油价格接近最低的时候买入，挣了好多倍，这就是差异。还有就是单独价格低也不行，有的公司发展不好，价格虽然低，但很可能继续往下跌。

如果同时满足，行业好、价格低就显示买入，否则不显示，不要用借助AND函数跟*来完成。


 3.3　让女朋友开心的条件（OR函数）


案例：
 现实生活中一定会发现经常遇到女朋友没有理由的生气，或者感觉女朋友无理取闹，那么在这个时候你应该怎么办呢？是晾着不管还是想办法哄她开心呢。这个时候如果你选择不管，那一定会进一步激化矛盾，所以你最好的办法就是尽快找到合适的方法哄她开心。

下面的方法都可以让女朋友开心：吃大餐、送礼物、安慰她、讲趣事、去散心、陪伴……

哄女朋友开心的方法有很多，但只要你用了1种或者多种方法女朋友都会开心，而如果你什么都不做，女友肯定会很不开心。

在G2单元格输入以下公式，并向下复制。


    =IF(OR(A2="吃大餐",B2="送礼物",C2="安慰她",D2="讲趣事",E2="去散心", F2="陪伴"),"开心","不开心")


原理分析

OR函数就是或者的意思，语法如下：


    OR(条件1,条件2,…)


只要其中有一个条件满足就显示TRUE，全部不满足就显示FALSE。在广东这边每天不是喝粥就是吃白米饭，不管吃什么都算吃过。学会了AND函数，再来学这个OR函数就变得简单很多。

知识扩展

AND函数可以用*代替，OR函数可以用+代替，前面案例的公式可以改成以下这样：


    =IF((A2="吃大餐")+(B2="送礼物")+(C2="安慰她")+(D2="讲趣事")+(E2="去散心")+(F2="陪伴"),"开心","不开心")


课后练习

怎么样才能在中国股市永不亏钱呢？大概有三种人：有内幕的、大量研究财务报告的、不炒股的。炒股7亏2平1挣，挣钱的比例非常小，而普通人正常都不会有内幕，除了大量研究每个公司的财务报告，专心挑选好公司，最好的办法就是不炒股，不炒就不会亏，确保安全。

让符合条件的显示不亏钱，否则显示空白。

第4章　查找引用函数


 4.1　根据姓名查询快递单号（VLOOKUP函数）


疑难：
 现在我们买东西都是在网上购买，经常都会收到各种快递，并通过快递单号查询物件什么时候到。现在有一份快递清单，因为姓名的顺序乱了，如何根据姓名查找韵达快递单号呢？

查找数据我们首先想到的就是借助Excel中的查找功能，复制姓名，利用【Ctrl】+【F】快捷键调出“查找和替换”功能对话框，查找姓名所在的位置，然后复制粘贴韵达快递单号。每次查找一个就重复一遍上述的操作。如果数据只有几个的话，用这种方法其实也挺快的，但假如有1000个甚至更多的时候，这种方法明显不行，非常吃力。

妙解：
 其实Excel中的函数也有这个查找的功能。查找数据首先想到的就是VLOOKUP函数，这是一个非常神奇的函数，学会它可以让你事半功倍。

在E2单元格输入以下公式，并向下复制公式。


    =VLOOKUP(D2,A:B,2,0)


正常的话，输入公式就能查找到对应值，但这时出现了一个很奇怪的现象，就是查找出来的结果都是1.90092E+12，这是怎么回事呢？

在Excel中超过11位的数字都会显示科学计数法，也就是类似E+12这种形式出现。选择E列，用【Ctrl】+【1】快捷键调出“设置单元格格式”对话框，在“分类”选项卡选择“数值”格式，小数位数改为0，单击“确定”按钮。通过设置单元格格式，数字就显示正常了。

原理分析

本例借助VLOOKUP函数查找D2单元格在区域A:B两列中的第2列的对应值，也就是韵达单号，函数语法如下：

VLOOKUP(查找值,查找区域,返回查找区域的第几列,精确还是模糊查找)

参数4：如果为0就是精确查找，1就是模糊查找。

这个函数参数有点多，如下面示意图所示应该会更好理解。

如果还不能理解，不妨看看微博高手们如何对这个函数进行通俗易懂的说明。

=Vlookup(找啥，在哪块儿找，找到后要它身上哪个地儿的东西，是准确找还是近似找找)

东北话版：=Vlookup（你瞅啥，向哪儿瞅，瞅着了是卸胳臂还是卸腿，来真格的还是扯犊子）

河南话版：=Vlookup(信谁咧，搁哪儿信，信卓特跟特要啥，非得信卓特还是长哩差不多都中)

知识扩展

在学习函数到一定程度的时候，很多人都喜欢使用简写，但对于新手而言这种简写让人很摸不着头脑，明明看起来“一样”，但结果却差了十万八千里。

高手这样写公式：


    =VLOOKUP(D2,A:B,2,)


菜鸟这样写公式：


    =VLOOKUP(D2,A:B,2)


粗看还真一样，对于菜鸟而言，很多时候都记不住函数有几个参数，就依然画葫芦照搬，但如果仔细观察，还真能够发现这2个公式存在一点点的差别。那就是第一个公式最后有个逗号（,），而第二个公式没有。这个逗号在这里起什么作用呢？

用逗号占位的，其实也就是省略了0，依然就是精确查找，这2个公式的作用是一样的。


    =VLOOKUP(D2,A:B,2,)
    =VLOOKUP(D2,A:B,2,0)


而没有逗号占位的，其实就是模糊查找，跟最后参数设置为1的作用是一样的。


    =VLOOKUP(D2,A:B,2)
    =VLOOKUP(D2,A:B,2,1)


说到模糊查找，应该很多人会好奇这个用在哪里呢？

模糊查找就是获取一个区间的对应值，比如下面的例子，查找天数在期间的日费率。

借助一个辅助列，创建每个区间的下限，然后用VLOOKUP函数进行模糊查询，当查找不到对应值时，会返回比查找值小，且最接近查找值的值。比如查找值为70，查找不到对应值，就查找比它小的值，0、31、61都比它小，而61最接近70，也就是查找61的对应值。


    =VLOOKUP(D12,$E$4:$G$9,3)


课后练习

在购买基金前要了解的信息有很多，比如这只基金的基金经理是谁，任职了多少年，任职期间涨了多少。如下图所示，根据基金简称，用VLOOKUP函数查询每只基金的所有对应值。

 4.2　根据客户号逆向查询客户姓名（LOOKUP函数）


疑难：
 现在有一份客户姓名跟客户号清单，想根据客户号查找客户姓名。

前面我们说过查找数据用VLOOKUP函数，现在试试是否行得通？

在E2单元格输入以下公式，并向下复制公式，发现查询的结果全部是错误。


    =VLOOKUP(D2,A:B,1,0)


[image: ]


怎么会出现这种情况呢？

原来查找值“客户号”必须在首列才能查找到对应值，现在是“客户姓名”在首列，所以会出现错误。

有的人可能会想到，将客户号复制，然后在A列重新插入复制的单元格，这样就变成“客户号”在首列了。

在E2单元格输入以下公式，并向下复制公式，所有结果就都出来了。


    =VLOOKUP(E2,A:C,2,0)

妙解：
 除了这种方法，在不改变数据源的情况下还有其他办法吗？

答案是肯定的，VLOOKUP函数有一个兄弟叫LOOKUP函数，查询数据的时候不区分方向，即使是逆向也可以查询，用在这里刚好合适。

在E2单元格输入以下公式，并向下复制公式。


    =LOOKUP(1,0/($B$2:$B$34=D2),$A$2:$A$34)


原理分析

LOOKUP函数经典查找模式的通用公式如下：


    LOOKUP(1,0/((条件1)*(条件2)*…*(条件n)),返回区域)


通用公式如何获取的，在后续章节会详细介绍。有了这个通用公式，即使是多条件查询对应值也可以轻易做到，这就比VLOOKUP函数好用多了。

如根据品名和加工单号两个条件查询出库数量。

在G2单元格输入以下公式，并向下复制公式。


    =LOOKUP(1,0/(($A$2:$A$11=E2)*($B$2:$B$11=F2)),$C$2:$C$11)


知识扩展

基本上VLOOKUP函数可以做的事，LOOKUP函数都可以做到。如前面说的模糊查找对应值。

模糊查找就是获取一个区间的对应值，比如下面的例子，查找在天数期间的日费率。

借助一个辅助列，创建每个区间的下限，然后用LOOKUP函数进行模糊查询，当查找不到对应值时，会返回比查找值小，且最接近查找值的值。比如查找值为70，查找不到对应值，就查找比它小的值，0、31、61都比它小，而61最接近70，也就是查找61的对应值。

下面2个公式都可以查找到对应值。


    =LOOKUP(D12,$E$4:$G$9)
    =LOOKUP(D12,$E$4:$E$9,$G$4:$G$9)


课后练习

在购买基金前要了解的信息有很多，比如这只基金的基金经理是谁，任职了多少年，任职期间涨了多少。如下图所示，根据基金简称，用LOOKUP函数查询每只基金的所有对应值。

 4.3　查询生肖的排位（MATCH函数）


疑难：
 这里有一份12生肖对应表，要根据生肖，查询自己的排位。

妙解：
 VLOOKUP函数与LOOKUP函数都是查询对应值，如果再添加一列“排位”可以用这2个函数，但现在只有一列，用这2个函数明显不合适。有一个函数跟VLOOKUP函数很像——MATCH函数。这个函数的语法跟VLOOKUP函数类似，只是少了第三参数“返回第几列”，其他都一样。不返回第几列的对应值，也就是只查询它的排位而已，比如兔就是排在12生肖的第4位。

在D2单元格输入以下公式：


    =MATCH(C2,A2:A13,0)


原理分析

MATCH函数的语法如下：

MATCH(查找值,查找的区域,精确或者模糊查找)

第三参数0的时候为精确查找，1为模糊查找。

知识扩展

MATCH函数单独存在的作用很小，如果加上INDEX函数，合作威力会非常大。MATCH函数是获取排位，而INDEX函数是获取排位的对应值。

如现在根据MATCH函数知道兔的排位是4，根据这个排位，我们就可以设置以下的公式获得兔对应的英文：


    =INDEX(B2:B13,E2)


E2的公式是：


    =MATCH(D2,A2:A13,0)


公式有一个很好玩的功能就是函数嵌套，跟组合积木一样，通过组合你可以得到你想要的模型。如果将INDEX函数的第二参数E2换成MATCH函数，就可以直接得到对应值，不用再经过中间这个过渡就能通过中文查询英文。


    =INDEX(B2:B13,MATCH(D2,A2:A13,0))


其实INDEX函数也有另外一种用法，就是可以有3个参数，如刚刚的例子就可以将公式改成：


    =INDEX(A2:B13,MATCH(D2,A2:A13,0),2)


2就是区域A2:B13单元格的第2列，即B列，如果是要返回A列的对应值，可以写成以下这样：


    =INDEX(A2:B13,MATCH(D2,A2:A13,0),1)


INDEX函数的第三参数与第二参数都是可以通过MATCH函数获取，也就是说要获取区域中任意一列都可以很轻松完成。

INDEX函数语法说明：

语法1：


    INDEX(区域,行号)


语法2：


    INDEX(区域,行号,列号)


查询对应值的通用公式：


    INDEX(区域,MATCH,MATCH)


课后练习

在购买基金前要了解的信息有很多，比如这只基金的基金经理是谁，任职了多少年，任职期间涨了多少。如下图所示，根据基金简称，用INDEX+MATCH函数查询每只基金的所有对应值。

 4.4　路程导航（OFFSET函数）


疑难：
 我们去陌生的地方就经常会问朋友，我现在在某某地方，如何去某某地方。或者用百度地图查询。比如我现在的起点是中国银行，终点是潮州市政府。

这时先往后走630米到潮州大道和枫春路这个十字路口，再往右边走200米就到潮州市政府了。

那现在卢子要去潇子家，该怎么走呢？

妙解：
 有2种方案：

1．往右边走3站到姐姐家，再往下边走5站到潇子家。

2．往下边走5站到班长家，再往右边走3站到潇子家。

只往右边走3站，不向下走。


    =OFFSET(B2,0,3)


只往下边走5站，不向右走。


    =OFFSET(B2,5,0)


不管是哪种方案，都必须向下5站，向右3站。


    =OFFSET(B2,5,3)


原理分析

OFFSET函数的语法如下：


    OFFSET(起点,偏移多少行,偏移多少列)


向下（右）用正数表示，向上（左）用负数表示。虽然两点之间的距离最短，但OFFSET并不支持对角线走，因为对角线没路可走。

比如现在潇子要去卢子家，就是向上5行，向左3行。


    =OFFSET(E7,-5,-3)


知识扩展

OFFSET函数其实有五个参数，只是后两个参数用得比较少，完整的函数语法如下：

OFFSET(起点,偏移多少行,偏移多少列,行高,列宽)

如图所示是学生语文、数学及英语三科的成绩表，现在想获取语文的成绩。

在单元格输入以下公式，回车后居然出现错误，怎么回事呢？


    =OFFSET(A1,1,1,7)


按理来说这个公式没错，A1偏移1行1列就是B2单元格，行高为7就是代表语文这一列的成绩。但实际上少考虑了一个问题，1个单元格怎么能容纳7个单元格的内容，就比如你平常吃1碗饭，现在让你吃7碗饭肯定会出现问题，同样的道理。

那怎么查看这个公式到底有无问题呢？

这时就得借助一个神器——【F9】键，因为谐音的缘故，有人把它称为独孤九剑。专门来解读公式用的，哪里不会抹黑哪里。在编辑栏用鼠标选择公式，Excel自动会帮你抹黑，也就是背景颜色变了。

按【F9】键，就出现了语文这一列的成绩，也就是说我们的公式想法其实是正确的，但没办法直接体现，需要借助其他函数间接体现。

现在要获取语文的总成绩，就在函数外面嵌套一个SUM函数进行求和即可。


    =SUM(OFFSET(A1,1,1,7))


可用下面的公式验证结果。


    =SUM(B2:B8)


通过验证，再一次确定公式是正确的。

如果现在是想获得所有成绩的总和，可用下面的公式。


    =SUM(OFFSET(A1,1,1,7,3))


验证的公式：


    =SUM(B2:D8)


看到这里你也许会觉得，用OFFSET在这里体现不出任何价值，只会更加麻烦，但为什么还要使用这个函数呢？在某些情况下，行高跟列宽两个是可以随时变动的，因为不确定，所以无法直接使用SUM函数，这也就是OFFSET函数在这里的价值。

课后练习

现在根据基金的明细表获取某一列的所有值，F1这个单元格只要一改变，下面的对应值也要随着改变。

 4.5　间接引用多个表的合计（INDIRECT函数）


疑难：
 如下图所示是1月到4月机台的产量，4个表格的格式一模一样，如何引用所有表格的合计呢？

正常我们的做法就是依次引用每个表格的合计。


    =1月!B11
    =2月!B11
    =3月!B11
    =4月!B11



妙解：
 在表格少的情况下，这种方法也挺快捷的，但表格多的时候，容错出错，而且效率低下。直接引用不行，这时可以考虑它的对立面，间接引用。

先举个简单的例子来说明什么是间接引用？

正常的话要获取A2这个单元格的值可以直接输入以下公式：


    =A2


而间接引用是通过第三方而得到A2这个单元格的值：


    =INDIRECT(A1)


比如你喜欢一个人，可以直接向她告白，也可以通过别人的转述向她告白，从别人转述出来的就是间接引用，这样就比较好理解。

这个例子懂了后，最开始的引用多表的合计就变得很简单了，依样画葫芦。在B2单元格输入以下公式，并向下复制。


    =INDIRECT(A2&"!B11")

原理分析

INDIRECT函数的语法如下：


    INDIRECT(单元格引用)


公式中的&就是将字符连接起来，变成了“1月!B11”，也就是变成“工作表名!单元格”的格式。

这个&就相当于月老的红绳，可以将相爱的两个人牵在一起，如将小红跟小明撮合在一起。

知识扩展

有的时候可以看见下面的公式，用R1C1这种格式显示，这是什么意思呢？


    =INDIRECT("R2C1",0)


在Excel中有两种样式，一种就是我们最常用的A1，另外一种就是R1C1样式。R是ROW的缩写，代表行；C是COLUMN的缩写，代表列。也就是第N行第N列的对应值，这种用得很少，不直观。

单击“文件”选项卡→“选项”→“使用公式”，勾选“R1C1引用样式”，再单击“确定”按钮。

返回到工作表界面，单击任意单元格，如下图所示可以看到在名称框出现的R1C1引用样式。

这种样式始终别扭，如果要重新返回A1引用样式，可单击“文件”选项卡→“选项”→“使用公式”，取消勾选“R1C1引用样式”，再单击“确定”按钮。

课后练习

理财除了基金、股票，还有P2P，我投资过的平台很多，如人人聚财、有利网、陆金所。目前就陆金所后台比较硬，背靠中国平安。2014年在每一家公司都投资了一部分金额，如何将明细表的投资金额累计，计算每一家公司的投资金额呢？

第5章　文本函数


 5.1　姓名分离（LEFT/RIGHT函数）


疑难：
 字符合并与分离是很常见的事，我们知道&可以将字符合并在一起，那用什么函数可以将字符分离呢？A列的姓名都是3个字，分别在B列得到姓，C列得到名。

妙解：
 在B2单元格输入以下公式，并向下复制公式。


    =LEFT(A2,1)


在C2单元格输入以下公式，并向下复制公式。


    =RIGHT(A2,2)


原理分析

LEFT函数的语法如下：


    LEFT(字符,N)


从左边提取字符N
 位。

RIGHT函数的语法如下：


    RIGHT(字符,N)


从右边提取字符N位。

因为姓都是1位，就是从左边提取1位；名都是2位，就是从右边提取2位。

知识扩展

正常情况下，姓名除了3位，还有2位的，针对这种情况，又该如何提取名呢？

用RIGHT函数的话，需要判断是2位的姓名还是3位的姓名，比较麻烦。其实，它们还有一个兄弟叫MID函数。这个函数比它们灵活很多，可以从任意位置提取任意字符。不管名是多少位，都是从第2位开始提取，只要提取2位就能够保证将名提取出来。

在C2单元格输入以下公式，并向下复制公式。


    =MID(A2,2,2)


可能你会产生疑问，为什么名只有1个字符也能提取出来？有疑问很正常，我们打开MID函数的帮助。备注提到一句话，当要提取的字符数超过文本的长度，则MID只返回至多直到文本末尾的字符。也就是说，如果你现在要提取的字符为2位，而实际上只有1位，就按实际的1位进行提取。

课后练习

做投资理财并不是看到哪种产品收益率高就投，还需要考虑自己的承受能力，正所谓风险跟收益成正比。如果自己的风险承受力比较差，还是买一些货币基金和银行理财，这种产品可以说是零风险。如果你的风险承受力非常强，就可以尝试买P2P和股票这种高风险的产品，不过这很可能最后本金都受到亏损。还是那句话，投资有风险，购买需谨慎。根据收益区间，获取每款产品的最小收益和最大收益。

 5.2　获取省份（FIND函数）


疑难：
 在一份快递单里，省份的字符数不确定，如何将省份提取出来呢？


妙解：
 在B2单元格输入以下公式，并向下复制。


    =LEFT(A2,FIND
    ("省",A2)-1)

原理分析

FIND函数的语法如下：


    FIND(要查找的字符,文本)


从文本中查找字符出现的位置。

FIND("省",A2)获取省字出现的位置3，省份不包含省字，就是出现的位置减去1，也就是FIND("省", A2)-1，因为是从左边提取，所以用LEFT函数。

知识扩展

2015年新广告法禁止出现与“一”有关的词，现在用FIND函数判断的时候，不包含“一”全部显示错误，该如何解决？

FIND函数如果查找到，就返回在文本中的位置（也就是数字，如2），而查找不到就返回错误值，IF函数是无法直接判断错误值的，这时可以借助另外一个函数来判断是数字还是其他。


    =ISNUMBER(FIND("一",A2))


如果是数字，就返回TRUE，错误值就返回FALSE。

最后再嵌套IF函数，就大功告成了。


    =IF(ISNUMBER(FIND("一",A2)),"违规","正常")


课后练习

基金的分类有很多：股票型、指数型、混合型、债券型，只要里面包含了关键字眼就划分成某个类型。如申万菱信中证军工指数分级，就属于指数型，现在对所有的基金进行分类。

 5.3　区分手机与电话号码（LEN函数）


疑难：
 在一堆杂乱的号码中，如何快速区分这些是手机还是电话号码？

妙解：
 在D2单元格输入以下公式，并向下复制。


    =IF(LEN(C2)=11,"手机","电话")


原理分析

LEN函数的语法如下：


    LEN(字符)


计算字符中有多少位数。

手机号码是11位，电话号码是不等于11位的。用LEN函数判断是否等于11位，如果是就显示手机，否则就显示电话。

知识扩展

读书和写文章的时候，经常有规定，写一篇800字以上的文章或者写一篇1000字以上的文章。那时候为了要知道有多少个字，都是一个个地数的。假如这段话是在Excel中，要计算多少个字就变得很简单了。

STEP 01
 　在B2单元格中输入以下公式，并向下复制公式，计算每一段的字符数。


    =LEN(A1)



STEP 02
 　在B7单元格中输入以下公式，计算总字符数。


    =SUM(B1:B6)


用两步就可以轻松统计字符数。

当然，统计字符数用Word更专业，只要直接复制到Word中就知道有多少个字符，这里只是提供一种解决问题的思路而已。

课后练习

投资理财除了基金、P2P、股票外，其实还有保险，现在重视保险的人越来越多了。

“保险发展到今天，已不是买不买的问题，而是买多少的问题，是自己的亲人里还有谁没有买的问题”。

在上段话中，一共出现了多少个买字？

 5.4　手机号加密（REPLACE函数）


疑难：
 为了保护中奖者的隐私，电视台在公布中奖号码时通常都是恭喜138尾号是1234的朋友中了一等奖，也就是中间部分号码不公布。如何将手机号码的4～7位用星号（*）表示，达到加密的效果。

妙解：
 在D2单元格中输入以下公式，并向下复制。


    =REPLACE(C2,4,4,"****")


原理分析

REPLACE函数的语法如下：


    REPLACE (字符,开始位置,替换的字符个数,替换成什么)


将字符中某一段字符替换成新字符。4～7位替换掉，就是开始位置为4，替换掉4个字符数，替换成****。

知识扩展

在输入规格的时候，为了方便输入，将×输成*，现在如何将*替换成×？

REPLACE函数需要知道开始位置，但由于每个规格中的*位置不确定，这时派不上用场。SUBSTITUTE函数的作用也为替换，就是将旧字符替换成新字符，但可以不用知道字符的起始位置，只要知道为某一个具体字符即可。

在B2单元格输入以下公式，并向下复制公式。


    =SUBSTITUTE(A2,"*","×")


SUBSTITUTE函数语法如下：


    SUBSTITUTE(文本,旧字符,新字符)

课后练习

在做投资理财的时候，经常需要到网站上下载一些表格，但下载到的表格很多时候并不是标准的格式。比如在P2P平台下载到的日期是以“2014.2.1结清”这种格式，这种日期对后期统计不方便，如何将日期变成标准的“2014/2/1”这种格式？

第6章　日期和时间函数


 6.1　日期运算


疑难：
 我们经常可以看到类似的问题：这个产品再过10天就可以交货了，今天是10/5，距离过年还有多少天的倒计时。在Excel中如何处理这些问题呢？

妙解：
 问题1　知道开始日期和相隔的天数，如何计算结束日期呢？

在C2单元格输入以下公式，并向下复制。


    =A2+B2


问题2　知道开始日期和结束日期，如何计算间隔的天数呢？

在C8单元格输入以下公式，并向下复制。


    =B8-A8


原理分析

我们都知道数字可以直接进行四则运算，其实日期也属于数字。

将单元格的日期格式改成常规格式。

这样我们就可以看到日期的本质，其实就是数字。

[image: ]


知识扩展

既然日期可以进行四则运算，时间是否也可以进行四则运算呢？现在知道开始时间和结束时间，如何得到持续时间？

在C2单元格输入以下公式，并向下复制。


    =B2-A2


其实时间也属于数字，只是日期是整数，时间是小数而已。将单元格设置为常规，即可看到时间的本质。

12:00对应的是0.5，也就是说，24小时对应的是1。

课后练习

在做投资理财时必须要经常学习，想要不劳而获是不太可能的。如何统计上半年学习的总时间？

 6.2　今天的日期及时间（TODAY/NOW函数）


疑难：
 我们经常会听到这样的问题：今天几号了？现在是几点了？当听见这些时，第一反应就是拿起手机，查看日期和时间。如果你现在坐在电脑前，也可以借助Excel帮你快速获取这些信息。

妙解：
 今天的日期用下面的公式：


    =TODAY()


也可以使用【Ctrl】+【;】快捷键。

现在的时间用下面的公式：


    =NOW()


也可以使用【Ctrl】+【Shift】+【;】快捷键。

原理分析

这2个函数是Excel中最简单的函数，没有任何参数。TODAY翻译过来就是今天，NOW翻译过来就是现在，只需记住这2个的意思即可。

知识扩展

有时候我们问日期的用意是想知道今天是星期几，就知道还有几天就放假；问时间就是想知道还有多久就下班。

那怎么将今天的日期转换成星期几？


    =WEEKDAY(A2,2)


WEEKDAY函数就是返回一周中的第几天，参数2有很多种表示法，但针对中国而言，是采用数字1（星期一）到7（星期日）这种表示方式。

课后练习

笔者报名参加了理财班的学习，每周的星期天都有理财课，为了更清晰地知道星期天是什么日期，需要将星期天的日期填充成黄色的背景色，该如何做呢？

 6.3　获取间隔N个月的日期（DATE函数）


疑难：
 已知开始日期和间隔月份，如何获取结束日期？

在C2单元格输入以下公式，并向下复制。


    =DATE(YEAR(A2),MONTH(A2)+B2,DAY(A2))原理分析


使用YEAR函数获取年份，MONTH函数获取月份，DAY函数获取日。

YEAR、MONTH、DAY三个函数的语法如下：


    YEAR(日期)
    MONTH(日期)
    DAY(日期)


DATE函数返回表示特定日期的连续序列号，语法如下：


    DATE(年,月,日)


先依次获取年月日，然后加上间隔的月份，即可知道结束日期。

知识扩展

DATE函数虽然可以解决问题，但显得很烦琐，其实借助EDATE函数会显得更简单。


    =EDATE(A2,B3)


EDATE函数返回表示某个日期的序列号，该日期与指定日期相隔（之前或之后）指示的月份数。使用EDATE函数可以计算与发起日处于一月中同一天的到期日的日期。

EDATE函数语法如下：


    EDATE(日期,间隔月份)


间隔月份为正数就是之后的日期，0就是本月，负数就是之前的日期。


    =EDATE(A2,-1)


表示指定日期之前的1个月。

课后练习

在学习理财的时候，由于老师的名气不高，再加上很多人对理财兴趣不足，第1期的学员可以学习2年，以这个特殊的条件来吸引人。第2期开始就全部恢复正常，学习的时间都是1年。根据报名日期和第几期，获取学习的截止日期。

 6.4　计算工龄（DATEDIF函数）


疑难：
 如何通过入职日期与离职日期计算工龄？工龄以×年×个月的形式显示。

妙解：
 在C2单元格输入以下公式，并向下复制。


    =DATEDIF(A2,B2+1,"y")&"年"&DATEDIF(A2,B2+1,"ym")&"个月"


原理分析

DATEDIF函数的语法如下：


    DATEDIF(开始日期,结束日期,间隔参数)


第三参数类型：


        “Y”时间段中的整年数。
        “M”时间段中的整月数。
        “D”时间段中的天数。
        “MD”日期中天数的差。忽略日期中的月和年。
        “YM”日期中月数的差。忽略日期中的日和年。
        “YD”日期中天数的差。忽略日期中的年。


这个函数的第三参数比较复杂，借助两个图来加强理解（图来自网络）。

下图是一个时间轴，两头分别是开始日期和结束日期，分别用S和E表示。

在中间有两个时间点。从S点开始，到A点，表示最大的整年的时间点，剩下的日期段AE是不能形成整年的剩余日期。

从S点开始，到B点，表示最大的整月的时间点，剩下的日期BE是不能形成整月的剩下的日期。

我们来看其中的y、m、d三个参数，这三个参数都很容易理解。

从下图再来理解ym、yd、md三个参数。

ym表示不再计算已经形成整年的SA段，在剩下的日期段AE中，计算能形成整月的月数，因此这个结果肯定是小于12的。

md表示不再计算已经形成整月的SB段，在剩下的日期段BE中，计算能形成整天的天数。

yd表示不再计算已经形成整年的SA段，在剩下的日期段AE中，计算能形成整天的天数。

理解了上述时间轴的含义后，几个字母的组合就都容易记忆了：

两个字母组合的都是指忽略第一个字符的“整”（年或月）部分，计算第二个字母的“整”（月或天）部分。

理解了第三参数的含义后，就很容易理解以下公式。


    DATEDIF(A2,B2+1,"y")&"年"


这部分就是获取相差的年份，DATEDIF函数在计算工龄的时候会相差1天，如2014/1/1到2014/12/31这样本来是1年，而DATEDIF函数认为这样不足1年，只有2014/1/1到2015/1/1这样才算1年，所以在结束日期后面+1，就刚好避免了这种误差。


    DATEDIF(A2,B2+1,"ym")&"个月"


这部分是获取相差的月份。

年份和月份最后连接起来就是×年×个月的形式。

知识扩展

DATEDIF的第三参数有点复杂，再通过一个例子来说明。每天都用黄色背景色提示还有30天合同就到期。

计算2个日期之间的天数差可用：


    =DATEDIF(TODAY(),B2,"d")


在30天内提示，也就是天数≤30，可表示为：


    =DATEDIF(TODAY(),B2,"d")<=30


用背景色提示需要借助条件格式。

STEP 01
 　选择B2:B16区域，单击“条件格式”→“新建规则”。

STEP 02
 　在弹出的“新建格式规则”对话框中，单击“使用公式确定要设置格式的单元格”，输入公式，再单击“格式”。

STEP 03
 　在弹出的“设置单元格格式”对话框中单击“填充”，设置黄色背景色，单击“确定”按钮。

[image: ]



STEP 04
 　单击“确定”按钮，符合条件的就全部显示为黄色背景色。

课后练习

购买银行产品基本上都是单利的，也就是说，不会将利息滚动计算进去。而像P2P或者股票这些都是复利的，会将利息滚动进去投资。现在有三款产品，根据投资时间、月利率、本金，计算最后的本利和。本利和计算的公式为：本利和=本金×(1+利率)^期数

第二篇　深了解

函数与公式是非常神奇的东西，通过不同函数之间的嵌套可以解决各种各样的疑难。很多看起来不可能完成的难题，通过高级函数组合或者数组公式都可以得到解决。前面已经学习了单个函数的用法，现在学习多个函数的组合会变得轻松很多。



第7章　字符分离与合并


 7.1　文字与数字分离（LEN函数）


疑难：
 在一堆字符串中，左边是文字，右边是数字，字数长度又不确定，如何将两者分离开呢？

妙解：
 在B2单元格输入以下公式，并向下复制。


    =LEFT(A2,LENB(A2)-LEN(A2))


在C2单元格输入以下公式，并向下复制。


    =RIGHT(A2,2*LEN(A2)-LENB(A2))


原理分析

LEN函数用于统计字符数，它有一个带B的兄弟——LENB函数，其语法与LEN函数一样，但作用略有差异。LENB函数是统计字节数，文字按2字节算，数字按1字节算。根据这个特点就可以对公式进行设置。

文字的个数就是总字节减去总字符数，比如广东电信10000，总字节数就是13，而总字符数是9，因为每个文字都多出一个字符，13-9=4就刚好是4个文字。也就是：


    =LENB(A2)-LEN(A2)


因为文字在左边，就用LEFT函数提取。

而数字的个数就是总字符数减去文字的字符数，也就是：


    =LEN(A2)-文字数=LEN(A2)-(LENB(A2)-LEN(A2))=2*LEN(A2)-LENB(A2)


其实很多时候计算都跟数学差不多，如果稍微有点数学基础，学习函数嵌套会更有帮助。

知道了数字个数，因为是从右边提取，就可以直接嵌套RIGHT函数。

知识扩展

在“规格”中间都有一些数字，如何将这些数字提取出来呢？

在C2单元格输入以下公式，并向下复制。


    =MIDB(B2,3,LENB(B2)-4)


这些数字看似没规律，但只要细心观察还是有一些规律的。

1．规格开头都是1×，也就是从第3位开始提取。

2．规格后面都是单位，文字单位为1个字符，而英文单位为2个字符。英文其实跟数字一样，字节数都是1，也就是说，2个字符的英文等同1个文字，所以说单位的字节数也是一样。

从中间提取可以用MID函数，但这个函数与字符数有关，而不是字节。要根据字节提取，需要用到MIDB函数，其实说白了，凡是带B的函数都与字节有关。中间开始的位置为3，而要提取的总长度，就是总字节数减去开头的1×和单位，开头和单位都是2字节，加起来就是4字节，也就是：


    =LNEB(B2)-4


课后练习

现在不是要提取“规格”中的数字，而是要提取“产品名称”中的数字，又该如何做？产品名称都是带有单位g或者kg的。

 7.2　提取金额（MIDB函数）


疑难：
 某位朋友有记账的习惯，会将每月的各项消费详细地记录下来，如何将这些金额提取出来，方便以后统计呢？

妙解：
 在B2单元格输入以下公式，并向下复制。


    =MIDB(A2,SEARCHB("?",A2),2*LEN(A2)-LENB(A2))


原理分析

在这里，数字的位置是不确定的，唯一可以利用的条件就是字节。因为数字都是1字节，而文字是2字节，只要判断1字节的开始位置就可以了。

首先可以想到的是借助通配符问号（?）来查询，看看在什么位置，而查找字符的位置一般都是用FIND函数，现在是用字节，也就是用FINDB函数。但是输入公式后却显示错误值，怎么回事呢？

调出FINDB函数的帮助，可以看到下面一段话：

这里需要借助SEARCHB函数才可以，这时再调出SEARCHB函数的帮助，发现语法跟FINDB函数一样，里面还有一段备注：

看到这么一段话，基本上确定用SEARCHB函数来解决问题，在B2单元格输入公式，得到15，跟预想中的一样，文字是双字节，2个就是4，接下来就是数字，刚好就是第5位。

前面文字与数字的分离，就提到了计算数字的个数。

而数字的个数就是总字符数减去文字的字符数，也就是：


    =LEN(A2)-文字数=LEN(A2)-(LENB(A2)-LEN(A2))=2*LEN(A2)-LENB(A2)


现在开始位置确定了，要提取的数字长度确定，只需在外面嵌套一个MIDB函数即可，因为MID函数是根据字符提取，而MIDB函数才是根据字节提取。

其实写公式有时候就是一个猜测、推断的过程。先猜测用某个函数解决，顺利的话就一次解决，不过有的时候猜测错误，就重新再猜测，看Excel帮助了解相关函数之间的特点。了解后，再进一步测试，直到成功为止。

知识扩展

有的人因为工作需要会保留着不止一个联系号码，但通常情况下我们只需要其中一个即可，现在要获取最后面的号码，该如何做呢？这个例子跟前面有点类似，只是里面的文字换成了英文字母，而字母与数字是没法借助字节和字符这些特点来提取的。也就是带B的函数一概派不上用场。

现在全部是单字节，不能借助字节数和字符数的特点来提取，这时就是见证奇迹的时刻。

在B2单元格输入以下公式，并向下复制。


    =-LOOKUP(1,-RIGHT(A2,ROW($1:$15)))


ROW函数现在我们已经很熟悉了，就是获取行号，ROW($1:$15)获取1到15。现在看看RIGHT的语法：


    =RIGHT(文本,提取右边N位)
    =RIGHT(A2,1)得到4
    =RIGHT(A2,2)得到54
    =RIGHT(A2,3)得到454
    =RIGHT(A2,4)得到3454
    ……
    =RIGHT(A2,11)得到13486123454
    ……
    =RIGHT(A2,15)得到EL3：13486123454
    =RIGHT(A2,ROW($1:$15))


也就是获取右边1到15位，要看一个公式的运算结果，可以在编辑栏用鼠标选择公式，这时背景色会变化，俗称抹黑。

按【F9】键，即可在编辑栏中看到运算结果。


    ={"4";"54";"454";"3454";"23454";"123454";"6123454";"861234 54";"486123454";"3486123454";"13486123454";"：13486123454";"L：13486123454";"EL：13486123454";"TEL：13486123454"}


因为这是多个结果，所以不能直接在一个单元格显示，需要借助F9查看才行。


    =-RIGHT(A2,ROW($1:$15))


就是将数字变成负数，文本变成错误值，同样在编辑栏选择公式抹黑，按【F9】查看结果。


    ={-9;-9;-7 0 9;-6 7 0 9;-5 6 7 0 9;-556709;-5556709;-35556709;-735556709; -3735556709;-13735556709;#VALUE!;#VALUE!;# VALUE!;#VALUE!}


现在用1在这些由负数和错误值组成的数组中查询对应值，因为LOOKUP函数会忽略错误值，再加上每个数字都会比1小，从而返回最后一个满足条件的值，这样就会提取最后一个数字“-13735556709”。


    =LOOKUP(1,-RIGHT(A2,ROW($1:$15)))


既然将数字变成负数，就得想办法将它复原，再加一个负号就可以，负负得正。


    =-LOOKUP(1,-RIGHT(A2,ROW($1:$15)))

为什么要提取1到15位而不是提取1到更多呢？

Excel允许的最大数字刚好是15位，提取再多也没有意义，只要保证能提取全部数字就行。其实手机号码是11位，将15换成11也可以得到正确结果。

课后练习

根据前面所学技能，至少用3种方法提取月份中的数字。


 7.3　提取百分比数字（TEXT函数）


疑难：
 如何提取区间左边的百分比数字，也就是下限呢？

妙解：
 在B2单元格输入以下公式，并向下复制。


    =TEXT(LEFT(A2,FIND("-",A2&"-")-1),"0%")


原理分析

区间前几个都有包含“-”字符，而最后一个没有，如果直接用FIND函数查找的话，不包含“-”就会显示错误值。

A2&"-"的作用就是保证每个都包含“-”，这样用FIND函数查找就不会出现错误值，因为FIND函数默认情况下就是查找第一个“-”的位置，所以即使在最后面连接“-”，也不会影响前面查找到的位置。

本来知道“-”的位置，然后减去1，用LEFT函数从左边提取就应该没问题，但实际上最后一个出现了一个小小的问题，变成了1.01。

其实百分比的实质也是数字，101%就是1.01。TEXT函数的作用就是将数字转换成百分比。

TEXT函数的语法如下：


    TEXT(数字,要显示的格式)


知识扩展

当然有的东西也不一定非得用函数与公式解决，其实这种下限的提取用技巧非常容易办到。

方法1　借助Excel 2013新功能——快速填充。


STEP 01
 　在B2单元格输入1%。


STEP 02
 　向下复制，单击“自动填充选项”，选择“快速填充”。

瞬间就完成了下限的提取。

方法2　针对通用版本，采用分列。


STEP 01
 　选择区域A2:A5单元格，切换到“数据”选项卡，单击“分列”，在弹出的“文本分列向导”窗口中保持默认不变，单击“下一步”按钮。

STEP 02
 　在“其他”的分隔符号输入“-”，就可以从数据预览看到效果，单击“下一步”按钮。

STEP 03
 　目标区域改成$B$2，因为我们只需要下限，不需要上限，所以选择第2列，设置“不导入此列（跳过）”，单击“完成”按钮。

课后练习

在字符串中含有百分比，但百分比位置不确定，如何提取出来呢？

 7.4　将数字拆分到每个单元格（COLUMN函数）


疑难：
 有一列数字，需要将这些数字分别拆开到每一个单元格，该怎么做呢？

妙解：
 在B2单元格输入以下公式，并向右和向下复制。


    =MID($A2,COLUMN(A1),1)


原理分析

拆分数字就是依然取第1位、第2位、...、第8位，也就是不断地改变MID的第2参数。


    =MID($A2,1,1)
    =MID($A2,2,1)
    ……
    =MID($A2,8,1)


如果第2参数用手工更改会比较麻烦，而且还可能出错，这时借助COLUMN函数就可以获取1到N的序号。

知识扩展

拆分数字很多时候并不都是固定8位，像这种情况该如何拆分到每一个单元格呢？

在B2单元格输入以下公式，并向右和向下复制。


    =MID(REPT("",8-LEN($A2))&$A2,COLUMN(A1),1)


数字不足8位直接提取会显示有异常，需要想办法全部统一成8位。统一成8位最好的办法就是在数字前面填充空格。

填充空格的个数就是：


    =8-LEN($A2)


REPT函数可以重复某个字符N次，语法如下：


    REPT(字符,N)


现在要重复空格的次数，就可以用：


    =REPT("",8-LEN($A2))


其他用法就跟刚才的例子一样。

课后练习

如何将金额拆分到每一个单元格，最前面的一个单元格带￥符号呢？

 7.5　提取最后一个分割符的内容（TRIM函数）


疑难：
 在源数据中的都是以“/”作为分隔符号，如何提取最后一个分割符合“/”后面的内容呢？

妙解：
 在B2单元格输入以下公式，并向下复制。


    =TRIM(RIGHT(SUBSTITUTE(A2,"/",REPT(" ",99)),99))


原理分析

在最后面就用RIGHT函数提取，但提取的字符数不确定，所以不能直接提取。因为是以“/”作为分隔符号，我们可以将这个分割符号替换成99个空格，这样从右边提取99位，就能够保证提取最右边的所有字符。

为了方便说明，我们将99换成10，这样就可以清晰地看到这部分SUBSTITUTE (A2,"/",REPT(" ",10))在编辑栏用F9键抹黑解读的结果。每个分隔符现在变成了10个空格。

最后面的字符数+10个空格，用RIGHT函数从右边提取10位，因为最后面的字符数少于10，必然能够全部提取，同时也会提取一部分空格。

为什么要将空格数和提取的字符写成99？

因为我们要提取的字符数不确定，不过正常的话，字符数不会超过99，写99就是保证能将所有的字符提取。如果已知只有10个以内的字符，我们写10也是完全可以的。

因为提取出来的字符还包含空格，需要去除才可以。TRIM函数就是去除多余空格用的，语法如下：


    TRIM(字符)


知识扩展

如果现在要将调色配方按空格分别提取到每个项目上，该如何做呢？

在C2单元格输入以下公式，并向右和向下复制。


    =TRIM(MID(SUBSTITUTE($B2," ",REPT(" ",99)),COLUMN(A1)*99-98,99))


SUBSTITUTE($B2," ",REPT(" ",99))这部分跟前面的例子一样，就是为了将空格替换成足够多的空格。

提取项目1可以用：


    =TRIM(MID(SUBSTITUTE($B2," ",REPT(" ",99)),1,99))


也就是从第1位开始提取99位，然后用TRIM函数去除多余的空格。

提取项目2可以用：


    =TRIM(MID(SUBSTITUTE($B2," ",REPT(" ",99)),100,99))


左边的字符+99个空格，从第100个字符开始提取，也就是不会包含最左边的字符，这样提取了99位，字符前后必然都有一部分空格，再用TRIM函数去除。

同样道理，从199开始，提取99位就是项目3的内容。

为了保证提取正确，99一定不能比原来的总字符少，如果看到字符很长的话，你可以用999或者其他比较大的数字代替。


    COLUMN(A1)*99-98这部分就是为了依次获得1,100,199。


课后练习

数据中用不同的分隔符号分成7个数据，但每个分隔出来的数据都是2位数，如何依次提取出来呢？

 7.6　将多列内容合并成一列（CONCATENATE函数）


疑难：
 ABC三列分别是省份、城市、区县，如何将这三列合并成一列，变成地址呢？

妙解：
 在D2单元格输入以下公式，并向下复制。


    =A2&B2&C2


或者


    =CONCATENATE(A2,B2,C2)


原理分析

文本合并符&语法如下：


    字符1&字符2&…&字符n


CONCATENATE函数的语法如下：


    CONCATENATE(字符1,字符2,…,字符n)


作用都是将字符连接起来，组成一个新的字符串。

知识扩展

不管用哪一个，当需要连接的字符非常多时，都非常麻烦，很容易出错。比如现在要将A~F的内容合并在一起。

    =A1&B1&C1&D1&E1&F1


对于多个文本的合并，有一个很好用的函数PHONETIC。

在G1单元格输入以下公式，并向下复制。


    =PHONETIC(A1:F1)


PHONETIC函数的语法如下：


    PHONETIC(单元格区域)


这个函数在使用的时候需要注意以下两点：

1．单元格区域必须连续，不能是“A1:B1,E1:F1”这样的区域。

2．使用这个函数只能连接文本，不能连接数字，如果连接数字，就会被自动忽略。

3．不能将用公式产生的结果连接起来，如下图所示，红色填充色的单元格是用公式产生的。

需要特别注意的是，PHONETIC 函数的Excel帮助跟实际情况会有差异，容易让人产生误解。

PHONETIC 函数提取文本字符串中的拼音 （furigana）字符。该函数只适用于日文版。

课后练习

如何将城市连接起来，中间用分隔符分开？

 7.7　将标题与内容合并在一个单元格换行显示（CHAR函数）


疑难：
 有一份员工工资明细表，现在要制作类似于工资条的形式，上面是标题，下面是内容。唯一不同的就是，合并后的内容在一个单元格内换行显示而已。

妙解：



STEP 01
 　在另外一个表格的A1单元格中输入以下公式，并向右和向下复制。


    =Sheet1!A$1&CHAR(10)&Sheet1!A2



STEP 02
 　选择区域A:E，单击“开始”选项卡中的“自动换行”。

STEP 03
 　A列的银行账号比较长，将列宽拉宽一点。

原理分析

换行符在计算机中的代码是10，所以在公式中使用CHAR(10)后，再设置单元格格式为自动换行，即可实现结果换行显示。

CHAR函数的语法如下：


    CHAR(代码)


代码只需记住几个常用的即可，比如生成大写字母A，就是65；生成√就是41420。

知识扩展

对于不常用的代码，我们如果要使用的话，如何得知呢？现在要知道“我爱你”对应的代码。

CHAR函数是根据代码获取对应的值，而CODE函数刚好相反，是根据值获取相对应的代码。

课后练习

如何生成大写A～Z和小写a～z的序列？

[image: ]




第8章　查找系列


 8.1　查找单位与金额


疑难：
 公司每天都会收到其他单位的付款金额，现在想查询最后一天是哪家单位付款及付款金额？

妙解：
 在E2单元格输入以下公式。


    =LOOKUP("座",B:B)


在F2单元格输入以下公式。


    =LOOKUP(9E+307,C:C)


或者直接在E2单元格输入下面任意一个公式，并向右复制，一次生成2个结果。


    =LOOKUP(TODAY(),$A:$A,B:B)
    =LOOKUP(TODAY(),$A:B)


原理分析

为了更好地了解公式的含义，我们先看看下面几条公式。


    =LOOKUP(10,{1;2;3;9;5;6;7;8;4})


返回4。


    =LOOKUP(100, {1;2;3;9;5;6;7;8;4})


返回4。


    =LOOKUP(1000, {1;2;3;9;5;6;7;8;4})


返回4。

也就是说，LOOKUP函数查找到最后一个满足条件的值，在数字不确定的情况下，查找的值越大越能保证查找到的值得准确性。9E+307是一个很大的数字，Excel允许最大的数字不能超过15位，而9E+307是9乘以10的307次方，比最大值还要大，查找最后一个值是相当保险。“座”是一个接近最大的文本，当然类似于々这种生僻字比座还大，但正常情况下不会出现，所以就用“座”来查找最后一个文本。

同理用TODAY()也就是今天的数字查找过去发生的日期，肯定也可以查找到最后一个对应值。

知识扩展

如图所示，“类型”使用了合并单元格，虽然看起来好看，但是后期操作会很麻烦，如何在辅助列将“类型”填充成每一个单元格都有内容呢？

在B2单元格输入以下公式，并向下复制。


    =LOOKUP(1,0/($A$2:A2<>""),$A$2:A2)


当类型比较多的时候，不能保证一定就是文本，可能会是数字，这时如果用“座”来作为查询条件的时候可能会出错。而不管是什么内容，肯定是非空，也就是<>""。

现在以B13单元格的公式进行解释。在编辑栏用鼠标选择$A$2:A13<>""，按F9键进行解读公式。通过比较，公式将符合条件的转换成TRUE，也就是非空的值，不符合条件的转换成FALSE。

在编辑栏再次用鼠标选择0/({TRUE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;TRUE;TRUE})，按【F9】键进行解读。

0/(条件)将TRUE转换成0，FALSE转换成错误值#DIV/0!，构成一个由0跟#DIV!/0组成的数组。利用LOOKUP函数查找不到对应值，返回小于或等于的最后一个值特点进行匹配。因为LOOKUP函数在查找的时候忽略错误值，用1来查找0，也就是可以查找最后一个0的位置，然后返回区域$A$2:A13的对应值。

A13这个单元格为非空单元格，所以B13单元格的对应值就是A13单元格。A14这个单元格为空单元格，所以B14单元格的对应值就是上一个非空单元格的值，也就是A13单元格。

课后练习

前面的案例是查找最后满足条件的值，如果现在是查找最大数量对应的不良项目，又该如何做呢？

 8.2　查找数量最后及最早出现的时间


疑难：
 日期是乱序的情况下如何查询每个数量最后一次出现的时间（日期最大）及最早出现的时间（日期最小）呢？

妙解：



STEP 01
 　选择A列，单击“数据”选项卡中的[image: ]
 ，也就是对日期进行升序排序，在弹出的“排序提醒”对话框中，保持默认不变，再单击“排序”按钮。

STEP 02
 　查找最后日期，在E2单元格输入以下公式。


    =LOOKUP(1,0/(B:B=D2),A:A)


查找最早日期，在F2单元格输入以下公式。


    =VLOOKUP(D2,IF({1,0},B:B,A:A),2,0)



STEP 03
 　直接用公式产生的结果是数字的格式，需要将E:F两列的单元格格式设置为短日期格式。

[image: ]


或者在公式外嵌套TEXT函数，转换成日期格式。


    =TEXT(LOOKUP(1,0/(B:B=D2),A:A),"e/m/d")
    =TEXT(VLOOKUP(D2,IF({1,0},B:B,A:A),2,0),"e/m/d")


原理分析

通过前面的例子知道LOOKUP函数可以在查找的时候返回最后一个对应值，但因为日期是乱序，直接查找的话，最后一个对应值并不是最大的日期，所以这时需要对日期进行升序排序才能找到。

而VLOOKUP函数查找的时候是返回最早一个对应值，也就是首次出现。对日期升序还不能直接查询，我们知道VLOOKUP函数的查找值需要在最左列，而不能从右往左找。在最开始认识这个函数的时候，提到辅助列方法，其实也可以利用一个函数进行构造一个新区域。

对于IF({1,0},B:B,A:A)进行构造的区域，先分成几步，慢慢说明。

在单元格输入以下2条公式，参数1为1的时候，返回左边，参数1为0的时候，返回右边。


    =IF(1,"左边","右边")
    =IF(0,"左边","右边")


在单元格中重新输入以下2条公式，然后在编辑栏用【F9】键进行解读。{1,0}这种形式，还是按原来的顺序显示，而{0,1}这种形式会将顺序调换。也就是说通过改变1跟0的位置，可以调换内容的前后位置。


    =IF({1,0},"左边","右边")
    =IF({0,1},"左边","右边")


将参数2跟参数3换成区域，就可以将两列的数据调换好位置，构成一个新区域。我们现在需要B列在前，A列在后，所以可以这样写公式。


    =IF({1,0},B:B,A:A)
    =IF({0,1},A:A,B:B)


VLOOKUP函数在新区域中，是返回第2列的对应值。


    =VLOOKUP(D2,新区域,2,0)


将新区域替换成上述构成的公式即可。


    =VLOOKUP(D2,IF({1,0},B:B,A:A),2,0)
    =VLOOKUP(D2,IF({0,1},A:A,B:B),2,0)


最后说明一点，用整行整列的话看起来简洁一点，但运算速度会稍微慢点，如果数据量比较大的时候，建议使用一个很小的区域。


    =VLOOKUP(D2,IF({0,1},$A$2:$A$18,$B$2:$B$18),2,0)

知识扩展

IF函数因为参数的限制，只能有2个区域，不是非常灵活。而CHOOSE函数在这里就显得好用很多。同样是逆序查找，可以使用下面的公式。


    =VLOOKUP(D2,CHOOSE({2,1},$A$2:$A$18,$B$2:$B$18),2,0)


CHOOSE语法如下：


    CHOOSE(数字,值1,值2,值3…值N)


数字必须是介于1到254之间的数字，或是包含1到254之间的数字的公式或单元格引用。

值为从中选择一个数值或一项要执行的操作。参数可以是数字、单元格引用、定义的名称、公式、函数或文本。

当数字为1时返回值1，数字为2时返回值2，以此类推。

现在有数量1~数量3三列，现在想获取第2列值的和。


    =SUM(CHOOSE(2,B2:B18,C2:C18,D2:D18))


如我们在玩剪刀石头布游戏的时候，也可以借助CHOOSE函数来玩一玩，在单元格输入以下公式后，按F9键可刷新结果。


    =CHOOSE(RANDBETWEEN(1,3),"石头","剪子","布")


RANDBETWEEN函数为随机数，语法如下：

RANDBETWEEN(起始值,终止值)

课后练习

根据数字判断性别，奇数为男，偶数为女？

 8.3　多条件查找对应的数量


疑难：
 在明细表中有四列数据，在查询表用日期、组别、型号这三个条件，如何查询到对应的数量？

妙解：



STEP 01
 　为了方便查询，我们在明细表中创建一列辅助列，将三个条件组合起来。

STEP 02
 　在查询表的D2单元格输入以下公式，并向下复制。


    =VLOOKUP(A2&B2&C2,明细表!D:E,2,0)


原理分析

查询数据首选VLOOKUP函数，但VLOOKUP函数在正常的条件下，只能根据一个条件进行查询。如果直接查询是行不通的，这时借助一个辅助列就将三个条件合并成一列，然后再进行查询。

知识扩展

查询对应值我们想到的就是借助查询函数，但有时是可以例外的。因为我们现在查找后返回的是数值，其实也可以借助求和函数进行查找。单条件就用SUMIF函数，多条件就用SUMIFS函数，查找的时候非常方便。

现在以插入辅助列的数据作为数据源，但不使用辅助列进行说明。

在E2单元格输入以下公式，并向下复制。


    =SUMIFS(明细表!E:E,明细表!A:A,A2,明细表!B:B,B2,明细表!C:C,C2)


如果太久没用，忘记语法了，没关系，再重新复习一遍。

SUMIFS函数语法如下：


    SUMIFS(求和区域,条件区域1,条件1,[条件区域2],[条件2]……)


用求和函数还有一个好处，就是当没有对应值的时候显示0，而不是错误值。如现在将B2的内容改成卢子，用VLOOKUP函数查找会得到#N/A，而用SUMIFS函数查找会得到0。

如果查询返回的是文本，除了借助辅助列还有其他办法吗？

其实查找多个对应值，用LOOKUP函数也同样可以。

在F2单元格输入以下公式，并向下复制。


    =LOOKUP(1,0/((明细表!$A$2:$A$19=A2)*(明细表!$B$2:$B$19=B2)*(明细表!$C$2:$C$19=C2)),明细表!$E$2:$E$19)


LOOKUP函数查询对应值有一个通用的公式：


    LOOKUP(1,0/((条件1)*(条件2)*…*(条件N

)),返回区域)


有了这个公式，查询对应值变得非常简单。

课后练习

当查找不到对应值的时候显示错误值，看起来不是很美观，如何让错误值显示成空白呢？

 8.4　根据产品编号查询产品名称及单价


疑难：
 参数表是事先将产品的各项信息录入好，查询表是每天要输入的信息。如何只输入产品编号，产品名称及单价就自动查询出来，而不用手工输入呢？

妙解：


在C10单元格输入以下公式，并向右和向下复制。


    =VLOOKUP($B10,$A$2:$C$6,COLUMN(B1),0)


原理分析

VLOOKUP函数可以通过改变第3参数而返回各项对应值，现在要获取第2,3列的对应值。


    =VLOOKUP($B10,$A$2:$C$6,2,0)
    =VLOOKUP($B10,$A$2:$C$6,3,0)


如果项目少，更改几次参数也没什么，但项目多了，肯定不方便。产生序列号可以用ROW函数跟COLUMN函数，因为是要获取列号，所以用COLUMN函数。COLUMN(B1)就是表示从第2列开始，往右复制公式就变成从第3列开始……

知识扩展

按原来顺序排序是理想状态，但有的时候，项目的排列顺序是跟原来有差异的，如现在单价在产品名称前面，又该如何查询呢？

在C10单元格输入以下公式，并向右和向下复制。


    =VLOOKUP($B10,$A$2:$C$6,MATCH(C$9,$A$2:$C$2,0),0)


同样道理，这里需要通过改变第3参数才能获取所有对应值。因为不按原来顺序，COLUMN函数就派不上用场，而MATCH函数可以获取指定值在数据区域中的排位，刚好就适合用在这里。

单价在原来区域中的排位是第2位。


    =MATCH("单价",$A$2:$C$2,0)


产品名称在原来区域中的排位是第3位。


    =MATCH("产品名称",$A$2:$C$2,0)


不管两者顺序如何改变，在原来区域中的排位是不会改变。

课后练习

假如现在参数表的产品编号是在最后一列，在查询表中根据产品编号，如何依次查询到产品名称及单价呢？

 8.5　根据姓名或者工号查询工资


疑难：
 在查询工资的时候，有的人是报姓名，有的人是报工号，现在如何根据其中任意条件，查询到工资呢？

妙解：
 在F2单元格输入以下公式，并向下复制。


    =IFERROR(VLOOKUP(E2,A:C,3,0),VLOOKUP(E2,B:C,2,0))


或者用Excel2013新函数IFNA代替IFERROR函数。


    =IFNA(VLOOKUP(E2,A:C,3,0),VLOOKUP(E2,B:C,2,0))


原理分析

单独用一个VLOOKUP函数的时候，只能查询到一部分。比如用VLOOKUP(E2,A:C,3,0)只能查询到姓名对应的工资，而工号对应的工资就查不到。相反用VLOOKUP(E2,B:C,2,0)只能查询到工号对应的工资，而姓名对应的工资就查不到。而两者结合起来，就刚好可以找到所有工资。

VLOOKUP函数查询对应值的时候，如果查找不到对应值会显示错误值#N/A，我们可以借助错误值这个特点来进行两个公式合并。IFERROR函数可以让错误值显示成任意值，不是错误值则显示其本身。

IFERROR函数语法如下：


    IFERROR(值,错误值要显示的结果)


也就是说如果VLOOKUP(E2,A:C,3,0)查询的时候没错误值，就用VLOOKUP(E2,A:C,3,0)进行查询，有错误值就用VLOOKUP(E2,B:C,2,0)进行查询。

IFERROR是针对所有错误值，而IFNA只是针对#N/A这种错误，因为VLOOKUP查询不到对应值都是返回#N/A这种错误值，也就是可以用IFNA取代IFERROR。

知识扩展

对于低版本的朋友来说，使用最多的是IS类函数来进行屏蔽错误值处理。

Excel中IS类函数共有9个函数，可以用来检验数值的类型并根据参数取值返回TRUE或FALSE。


    1． ISBLANK(value)
    2． ISERR(value)
    3． ISERROR(value)
    4． ISLOGICAL(value)
    5． ISNA(value)
    6． ISNONTEXT(value)
    7． ISNUMBER(value)
    8． ISREF(value)
    9． ISTEXT(value)


IS类9个函数的参数都一样，Value为需要进行检验的数值。分别为空白（空白单元格）、错误值、逻辑值、文本、数字、引用值或对于以上任意参数的名称引用。

IS类函数在用公式检验计算结果时十分有用。当它与IF函数结合在一起使用时，可以提供用来在公式中查出错误值。

刚刚的例子用ISNA函数的话，可用：


    =IF(ISNA(VLOOKUP(E2,A:C,3,0)),VLOOKUP(E2,B:C,2,0),VLOOKUP(E2,A:C,3,0))


使用低版本的话，会多写一个VLOOKUP函数，显得烦琐一点，还是高版本比较简洁。

课后练习

公司员工的姓名全部在A列，现在根据输入的姓名，判断是否为公司的员工？也就是说，如果在A列查找不到姓名就返回“不是”，查找到就返回“是”。

 8.6　根据姓名及月份查询工资


疑难：
 下面的汇总表是各人员一到四月份的工资，现在根据姓名简称及月份两个条件查询工资，该如何做呢？

妙解：
 在B12单元格输入以下公式。


    =VLOOKUP("*"&B10&"*",A1:E8,MATCH(B11,A1:E1,0),0)


原理分析

因为使用的是简称，所以不能直接查询，需要跟通配符一起连用才可以。

通配符说明：*代表所有字符，?代表一个字符。

如“*今朝*”只要数据源包含今朝两个字都能查到，汇总表只有一个包含今朝，也就是可以查找到笑看今朝。

返回第几列的值，这个由MATCH函数决定，这个函数就是获取某值在区域中的排位。

知识扩展

有一部分人经常将模糊查找跟按通配符查找混为一谈，其实这两者是不同的。模糊查找是查找跟本身一样或者小于它的值，前提还要数据源升序，两者缺一不可，所以下面的公式就不能查询到正确的结果。


    =VLOOKUP(B10,A1:E8,MATCH(B11,A1:E1,0))


其实汉字也可以进行模糊查找，如根据汉字与字母对应表，进行查找姓名中每个汉字的首字母。

在B2单元格输入以下公式，并向下复制公式。


    =LOOKUP(MID(A2,1,1),$D$1:$E$24)&LOO KUP(MID(A2,2,1),$D$1:$E$24)&LOOKUP(MID (A2,3,1),$D$1:$E$24)&LOOKUP(MID(A2,4,1 ),$D$1:$E$24)


逐个地用MID函数将姓名拆开，然后用LOOKUP函数查询，最后将获得的字母用&连接起来。当然这个并不是很准确，比如有的汉字是多音字，这种根本没办法区分，仅当娱乐。

课后练习

查询以“河”作为结尾并且只有2个字符的营业部对应的金额，也就是相当于查找天河的对应金额。

 8.7　多表查询各地区店名的工程造价


疑难：
 有三个地区，分别是无锡、响水和苏州，各自的信息记录在三个表格中，每个表格的格式都是一样的。如何根据地区与店名在三个表格中查询工程造价呢？

妙解：
 在查询表的C2单元格输入以下公式，并向下复制。


    =VLOOKUP(B2,INDIRECT(A2&"!B:C"),2,0)


原理分析

正常跨表查询都是这样写公式的：


    =VLOOKUP(B2,无锡!B:C,2,0)
    =VLOOKUP(B2,响水!B:C,2,0)
    =VLOOKUP(B2,苏州!B:C,2,0)


参数2是“工作表名称!区域”这样的形式，通过不断地改变工作表名称可以查询任意表格。因为查询表已经提供了工作表名称，这时我们重新再用&连接区域即可。


    =A2&"!B:C"


可是当我们直接这样引用并输入公式的时候，却提示了错误。

其实直接用&得到的只是一个文本，要通过INDIRECT函数间接引用才能转换成一个真正的区域。


    =INDIRECT(A2&"!B:C")


知识扩展

在查询表有了地区的名字会比较好找，假如没有地区的名字，根据店名又该如何在这三个表格中查询工程造价呢？

在查询表的C2单元格输入以下公式，并向下复制。


    =SUM(SUMIF(INDIRECT({"无锡","响水","苏州"}&"!B:B"),B2,INDIRECT({"无锡","响水","苏州"}&"!C:C")))


因为查询的是数值，前面有提到，查询数值用SUM函数家族来处理最好。

因为是三个表格，所以先构成这三个表的区域文本。


    ={"无锡","响水","苏州"}&"!B:B"


接着嵌套INDIRECT函数进行间接引用，变成真正的区域。


    =INDIRECT({"无锡","响水","苏州"}&"!B:B")


根据SUMIF函数的语法，设置为以下公式：


    =SUMIF(INDIRECT({"无锡","响水","苏州"}&"!B:B"),B2,INDIRECT({"无锡", "响水","苏州"}&"!C:C"))


这时奇怪的现象发生了，除了无锡这个表的对应值是对的，其他都是错的。

这究竟是怎么一回事呢？

其实用SUMIF函数产生了三个数据，而不止一个，因为无锡这第一个表，正常的话一个单元格只能显示第一个数据，也就是无锡是对的。而响水跟苏州是第2，第3个数据，显示不出来，所以错误。

有疑惑时最好的办法就是在编辑栏用鼠标选中公式抹黑，然后用F9键解读。

因为是三个数据，需要在最外面嵌套SUM函数进行求和才能变成一个值，这样才能得到准确的结果。

课后练习

只有三个表格的时候可以手工输入工作表名，但现在有很多个工作表名，如8个。该如何进行查询“项目”在各个表格中的对应数量？

 8.8　员工信息查询，且带图片


疑难：
 员工信息表记录着每位员工的详细信息，如何在查询表中根据姓名查询各种信息，且包括照片呢？

妙解：
 在查询表的B2单元格输入以下公式，并向右复制。


    =VLOOKUP($A2,员工信息表!$A:$E,COLUMN(),0)


其他内容都可以正常显示，唯独照片不可以。

STEP 01
 　复制一个空单元格，粘贴成图片。

STEP 02
 　使用名称引用动态图片，定义一个叫图片的名称。单击“公式”选项卡中的“定义名称”按钮。在“名称”文本框输入“图片”，在“引用位置”输入下面的公式，然后单击“确定”按钮。


    =INDEX(员工信息表!$E:$E,MATCH(查询表!$A$2,员工信息表!$A:$A,0))

STEP 03
 　单击图片，在编辑栏输入公式，回车就自动生成相应的照片。=图片

原理分析

查询图片的公式其实跟常规查询差不多，如果我们要查询内容也用下面的公式。


    =INDEX(员工信息表!$E:$E,MATCH(查询表!$A$2,员工信息表!$A:$A,0))


只是因为图片不是单元格的内容，所以需要借助名称来实现而已。其中的复制粘贴成图片也可以用插入任意一张图片来实现。

知识扩展

粘贴成的图片还有一种是动态的，当数据源改变的时候，图片也会有相应的变化。

复制A1:E2单元格的内容，粘贴的时候选择链接的图片。

现在我们改变原始的数据源，会立即看到图片也随着改变。

[image: ]


图片为什么会改变呢？

其实只要点击图片，就可以看到编辑栏有一个引用地址，也就是图片始终得到这个地址的内容。这个地址是可以改变的，只要一改变，图片也可以随着改变。

以前在核对数据的时候，因为是不同表格，我就经常用这种方法。如果是复制原来表格的结果过来核对，可能需要再重新排版，而如果是用图片，怎么摆放都可以，比较方便，而且数据源变化，也能随着变化。

课后练习

前面一直介绍了VLOOKUP跟LOOKUP两个函数之间的查找，现在借助INDEX+MATCH函数的组合，进行查询。

汇总表是各员工一到四月份的工资，现在要根据姓名简称及月份两个条件查询工资，该如何做呢？

 8.9　取最接近100的数值


疑难：
 在一堆数值中，如何获取最接近100的数值呢？

妙解：
 在C2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，输入后公式会自动产生大括号，这个是数组公式跟普通公式的差异。


    =INDEX(A2:A10,MATCH(MIN(ABS(100-A2:A10)),ABS(100-A2:A10),0))


如果所有数值都小于100的话，可以用下面的公式。都小于100，就是需要求最大值，也就是最接近100的值。


    =MAX(A2:A10)


原理分析

这是我们接触到的第一个数组公式，数组公式的作用就是可以减少使用辅助列。

常规方法就是在B列创建一个辅助列，获取每个值跟100的差异，因为有的数值可能会比100大，得到的是负数，这时嵌套ABS函数。ABS函数就是获取绝对值，负数跟正数的绝对值都是正数。


    =ABS(100-A2)


接着再查找最小差异值的排位，然后获取对应值。


    =INDEX(A2:A10,MATCH(MIN(B2:B10),B2:B10,0))


解读数组公式，其实跟解读普通公式的方法是一样的。

在编辑栏用鼠标选择ABS(100-A2:A10)，按【F9】键，就可以得知，这里是获取每个值跟100的差值。

在编辑栏用鼠标选择MIN({27;48;59;9;63;73;72;32;69})，按【F9】键，就可以得知，这里是获取这些数值中的最小值，也就是跟100差异最小，最接近100的值。

解读公式就是哪里不会就抹黑哪里，看运算结果，多用【F9】键几次，基本上就能明白每部分的含义。

需要提醒的是：当你对公式按【F9】键进行求值后，返回的时候记得按【Esc】键，或者单击编辑栏左侧的“取消”按钮。否则公式就变成你求值后的样子了。

MATCH函数部分就是获取最小值，在差异值中的排位；嵌套INDEX函数就是得到最小值对应的值。

知识扩展

第一次接触到数组可能你会对它是怎么运算很感兴趣，现在就来聊聊数组公式的计算。

而数组公式是一次生成的，不用进行下拉，直接选择区域C2:C10单元格，输入公式后，按【Ctrl】+【Shift】+【Enter】三键结束。

其实数组还有一个区域自动扩展的功能，现在如果用一个值跟某列进行运算，这个值会自动扩展成与这一列同样的区域。选择区域H2:H10单元格，输入下图中的公式后，按【Ctrl】+【Shift】+【Enter】三键结束。

也就是说下面2个公式的作用是一样的。


    ={100;100;100;100;100;100;100;100;100}-A2:A10
    =100-A2:A10


单行单列的时候，区域也会自动扩展。

但并不是每种区域都能自动扩展的，就像这种3行与5行的乘积，因为Excel不懂要扩展成什么内容，就返回错误。

课后练习

求全部产品的总金额？

 8.10　根据城市查询省份


疑难：
 标题是各个省份，省份下面是所属的城市，现在根据城市如何查询相对应的省份呢？

妙解：
 在B9单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制公式。


    =INDEX($1:$1,MAX(IF($A$2:$D$4=A9,COLUMN(A:D))))


原理分析

省份跟城市是同一列，只要判断城市在第几列就可以。因为有4个省份，所以需要有{1,2,3,4}。生成列号用COLUMN函数，一共4列，就是COLUMN(A:D)。用COLUMN函数有一个好处就是即使有很多列，比如26列，也可以用COLUMN(A:Z)生成列号。


    $A$2:$D$4=A9


就是让区域的所有值跟A9单元格比较，如果是一样的，就显示TRUE，不一样的，就显示FALSE。


    IF($A$2:$D$4=A9,COLUMN(A:D))


就是让满足的值，显示这个值所在的列号，不满足的显示FALSE。


    MAX(IF($A$2:$D$4=A9,COLUMN(A:D)))


因为这是由列号跟FALSE组成的数组，需要再嵌套MAX函数才能获取列号，当然MAX函数也可以换成SUM函数、MIN函数等。


    MAX(IF(条件,值))


也可以改成：


    MAX((条件)*值)


知道列号，嵌套INDEX函数就是获取列号的对应值。

知识扩展

INDEX+MAX组合还可以解决多行多列的查找，在没学习数组之前，需要用到多次VLOOKUP函数才能解决，如果列数多的话还是比较麻烦的。如下图所示，项目跟数据分在三个区域输入，现在要根据J列的项目，查询K列的数据所对应的值。

在K2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制公式。


    =INDEX(A:H,MAX(($A$2:$H$10=J2)*ROW($2:$10)),MAX(($A$2:$H$10=J2)*COLUMN(B:I)))


INDEX函数在多行多列的区域中查找值，首先就要判断在第几行，然后再判断第几列。


    MAX(($A$2:$H$10=J2)*ROW($2:$10))


就是让区域中等于A2单元格的值，返回本身所在的行号。


    MAX(($A$2:$H$10=J2)*COLUMN(B:I))


就是让区域中等于A2单元格的值，返回本身所在的行号向右增加一列，因为数据是在项目右边一列，所以区域从B列开始，而不是A列。

课后练习

如何查找最小数据对应的项目？

 8.11　将外币转换成人民币并汇总金额


疑难：
 购买产品的时候，有部分是使用人民币，有部分是使用外币，如何将外币全部转换成人民币，并汇总金额呢？

妙解：
 选择C9跟C10两个单元格，输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(VLOOKUP(B2:B6,E2:F4,2,0)*C2:C6)

原理分析

VLOOKUP函数正常情况下，每次只能查找一个值，如果是试图一次查找多个值会显示错误。在编辑栏选择VLOOKUP部分内容，按【F9】键解读，就会发现返回错误值。

但有一个有趣的现象就是当VLOOKUP这部分内容输入在多个单元格查找时，却可以返回正确的值。

同理，如果公式在一个单元格内输入，得到的是第一个对应值，也就是人民币的总和而已，其他都无法汇总。这也就是公式为什么要在多个单元格输入的一个原因。

知识扩展

其实VLOOKUP也可以借助T/N+IF{1}技巧建立内存数组，这样就能够在一个单元格输入公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(VLOOKUP(T(IF({1},B2:B6)),E2:F4,2,0)*C2:C6)
    T(IF(1,B2:B6))，这里只得到一个结果，就是B2单元格的对应值。
    IF(1,B2:B6)，这里得到一个区域，也就是B2:B6。
    T(IF({1},B2:B6))，这个得到的结果有5个值，{1}将B2:B6这个区域变成数组。
    IF({1},B2:B6)，得到的结果是：{"人民币";"人民币";"美元";"英镑";"人民币"}，相当于IF({1;1;1;1;1},B2:B6)。
    当IF({1},B2:B6)区域中的值是文本的时候，嵌套T，是数值的时候，嵌套N。


还是那句话，如果查询的结果是数值，用SUM函数家族会变得更简单。在一个单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(SUMIF(E2:E4,B2:B6,F2:F4)*C2:C6)


课后练习

在文本与数字混合的项目，如何转换成效果1：只显示文本，数字显示空白；效果2：数字显示原来的结果，文本显示0。

[image: ]




第9章　求和计数系列


 9.1　计算文本表达式的和


疑难：
 在统计数据的时候忘记输入“=”号，如何计算这些文本表达式的结果呢？

妙解：



STEP 01
 　单击B2单元格，再单击“公式”选项卡的“定义名称”图标，在弹出的“新建名称”对话框中，在“名称”文本框中输入“表达式”，在“引用位置”文本框输入下面的公式，单击“确定”按钮。


    =EVALUATE(A2)

STEP 02
 　在B2单元格输入以下公式，并向下复制。


    =表达式



STEP 03
 　在Excel2007版本以上，保存的时候都会提示“无法在未启用宏的工作簿中保存以下功能：……”这个警告对话框。

单击“否”按钮，将工作簿另存为：Excel启用宏的工作簿，保存起来即可。

原理分析

EVALUATE属于宏表函数。早期低版本Excel中使用的宏表函数，现在已由VBA顶替它的功能，但它仍可以在工作表中使用。不过要特别注意的是：不能直接在单元格中使用，只能在“定义的名称”中使用。还有，一些宏表函数使用后不会自动改变。

EVALUATE函数语法如下：


    EVALUATE(表达式)


对以文字表示的一个公式或表达式求值，并返回结果。

在使用宏表函数或者VBA的时候，必须将工作簿另存为：Excel启用宏的工作簿，否则功能会失效。

知识扩展

有的时候就刚好相反，知道了结果，想获取表达式？

Excel2013提供了一个新函数，就是专门获取表达式。

在B2单元格输入以下公式，并向下复制。


    =FORMULATEXT(A2)


FORMULATEXT函数语法如下：


    FORMULATEXT(公式)


其他版本的Excel可通用宏表函数GET.CELL来获取表达式。

[image: ]



STEP 01
 　单击C2单元格，再单击“公式”选项卡的“定义名称”图标，在弹出的“新建名称”对话框中，在“名称”文本框中输入“表达式1”，在“引用位置”文本框中输入下面的公式，单击“确定”按钮。


    =GET.CELL(6,知识扩展!A2)

STEP 02
 　在C2单元格中输入以下公式，并向下复制。


    =表达式1


GET.CELL函数的作用是返回单元格中的格式信息。

GET.CELL函数语法如下：


    GET.CELL(类型号,单元格(或范围))


其中类型号，即你想要得到的信息的类型号，经实验，范围为1～66，也就是说这个函数可以返回一个单元格里66种信息。类型号这么多，我们也无须去记住，当需要的时候，重新查看即可。

以下是类型号及其代表的意思：


      1——参照储存格的绝对地址
      2——参照储存格的列号
      3——参照储存格的栏号
      4——类似 TYPE 函数
      5——参照地址的内容
      6——文字显示参照位址的公式
      7——参照位址的格式，文字显示
      8——文字显示参照位址的格式
      9——传回储存格外框左方样式，数字显示
      10——传回储存格外框右方样式，数字显示
      11——传回储存格外框方上样式，数字显示
      12——传回储存格外框方下样式，数字显示
      13——传回内部图样，数字显示
      14——如果储存格被设定 locked传回 True
      15——如果公式处于隐藏状态传回 True
      16——传回储存格宽度
      17——以点为单位传回储存格高度
      18——字形名称
      19——以点为单位元传回字号
      20——如果储存格所有或第一个字符为加粗传回 True
      21——如果储存格所有或第一个字符为斜体传回 True
      22——如果储存格所有或第一个字符为单底线传回True
      23——如果储存格所有或第一个字符字形中间加了一条水平线传回 True
      24——传回储存格第一个字符色彩数字，1 至 56。如果设定为自动，传回 0
      25——MS Excel不支持大纲格式
      26——MS Excel不支持阴影格式
      27——数字显示手动插入的分页线设定
      28——大纲的列层次
      29——大纲的栏层次
      30——如果范围为大纲的摘要列则为 True
      31——如果范围为大纲的摘要栏则为 True
      32——显示活页簿和工作表名称
      33——如果储存格格式为多行文字则为 True
      34——传回储存格外框左方色彩，数字显示。如果设定为自动，传回 0
      35——传回储存格外框右方色彩，数字显示。如果设定为自动，传回 0
      36——传回储存格外框上方色彩，数字显示。如果设定为自动，传回 0
      37——传回储存格外框下方色彩，数字显示。如果设定为自动，传回 0
      38——传回储存格前景阴影色彩，数字显示。如果设定为自动，传回 0
      39——传回储存格背景阴影色彩，数字显示。如果设定为自动，传回 0
      40——文字显示储存格样式
      41——传回参照地址的原始公式
      42——以点为单位传回使用中窗口左方至储存格左方水平距离
      43——以点为单位传回使用中窗口上方至储存格上方垂直距离
      44——以点为单位传回使用中窗口左方至储存格右方水平距离
      45——以点为单位传回使用中窗口上方至储存格下方垂直距离
      46——如果储存格有插入批注传回 True
      47——如果储存格有插入声音提示传回 True
      48——如果储存格有插入公式传回 True
      49——如果储存格是数组公式的范围传回 True
      50——传回储存格垂直对齐，数字显示
      51——传回储存格垂直方向，数字显示
      52——传回储存格前缀字符
      53——文字显示传回储存格显示内容
      54——传回储存格数据透视表名称
      55——传回储存格在数据透视表的位置
      56——枢纽分析
      57——如果储存格所有或第一个字符为上标传回True
      58——文字显示传回储存格所有或第一个字符字形样式
      59——传回储存格底线样式，数字显示
      60——如果储存格所有或第一个字符为下标传回True
      61——枢纽分析
      62——显示活页簿和工作表名称
      63——传回储存格的填满色彩
      64——传回图样前景色彩
      65——枢纽分析
      66——显示活页簿名称


课后练习

有些人在输入数据的时候还是用了数学中的思维，乘以就用×，除以就用÷，但这两种符号在Excel中是不被承认的，现在如何对包含这些运算符的文本表达式进行计算呢？

 9.2　对带颜色的项目进行求和


疑难：
 我们在使用Excel时，会遇到这样的情况：一个工作表中某些单元格被填充为某种颜色，要求将填充了某种颜色的单元格进行快速求和。现在保价金额被分别填充成红色跟绿色的背景色，如何分别对这两种背景色进行求和呢？

妙解：



STEP 01
 　单击G2单元格，再单击“公式”选项卡的“定义名称”图标，在弹出的“新建名称”对话框中，在“名称”文本框中输入“颜色”，在“引用位置”文本框中输入下面的公式，单击“确定”按钮。


    =GET.CELL(63,D2)



STEP 02
 　在G2单元格中输入以下公式，并向下复制。


    =颜色



STEP 03
 　将颜色产生的数字依次填入H2和H3单元格。


STEP 04
 　在I2单元格中输入以下公式，并向下复制。


    =SUMIF(G:G,H2,D:D)

STEP 05
 　将工作簿另存为：Excel 启用宏的工作簿。

原理分析

SUMIF函数虽然可以进行条件求和，但不能直接对颜色进行条件求和。需要借助宏表函数GET.CELL获取背景色对应的数字，然后才能求和。

GET.CELL函数中的参数“63”的意思是：单元格填充颜色(背景)编码数字。

知识扩展

其实除了借助宏表函数，用筛选和普通公式也同样可以按颜色统计。

STEP 01
 　在D18单元格中输入以下公式。


    =SUBTOTAL(9,D2:D17)



STEP 02
 　选择A1单元格，单击“数据”选项卡中的“筛选”图标。

STEP 03
 　单击“保价金额”的筛选按钮，再单击“按颜色筛选”，选择红色或者绿色，单击“确定”按钮。

通过筛选后，自动对筛选的颜色进行求和。

课后练习

筛选9月4日的数据，并求出平均的保价金额。

 9.3　对产品进行统计并引用自定义数字格式


疑难：
 产品从不同国家购买，因此使用的货币格式不一样，如何对产品进行条件求和，并引用B列原有的货币格式？

妙解：



STEP 01
 　单击E2单元格，再单击“公式”选项卡的“定义名称”图标，在弹出的“新建名称”对话框中，在“名称”文本框中输入“格式”，在“引用位置”文本框输入下面的公式，单击“确定”按钮。


    =GET.CELL(7,INDEX(Sheet1!$B:$B,MATCH(Sheet1!$D2,Sheet1!$A:$A,0)))



STEP 02
 　在E2单元格中输入以下公式，并向下复制。


    =TEXT(SUMIF(A:A,D2,B:B),格式)



STEP 03
 　将工作簿另存为：Excel 启用宏的工作簿。

原理分析

在名称“格式”中，使用INDEX函数与MATCH函数配合，查询D2在A列对应的值，并返回B列的单元格引用。使用GET.CELL函数，取得INDEX函数返回单元格的数字格式。

使用TEXT函数将SUMIF函数求得的和返回为“格式”的格式，“格式”是指B列对应的格式。

GET.CELL函数中的参数“7”的意思是：用于返回单元格的数字格式。

知识扩展

GET.CELL函数第一参数还有很多方法，请查阅：计算文本表达式的和。

课后练习

如何获取工作簿和工作表的名称，并显示在A1单元格？

 9.4　动态统计金额


疑难：
 这是一份每天出差花费清单，经常要在总金额上面插入行。用SUM函数直接统计有时不会对新增加的金额进行统计，该如何处理呢？

妙解：



STEP 01
 　单击C18单元格，再单击“公式”选项卡的“定义名称”图标，在弹出的“新建名称”对话框中，在“名称”文本框中输入“上一行”，在“引用位置”文本框中输入下面公式，单击“确定”按钮。


    =C17



STEP 02
 　在C18单元格中输入以下公式。


    =SUM(C2:上一行)


原理分析

这个“上一行”的引用为相对引用，每插入一行，引用位置就会动态变化，所以插入行也会自动汇总进去。

有人试过用SUM(区域)就能自动扩展区域，统计正确时就认为任何情况下都可以，但实际上某些情况下还是不会自动扩展区域。如下图所示，就是笔者几年前吃过的一个亏，黄色这一行为后来插入的行，插入行后数量统计正确，笔者就没多想，以为金额肯定正确。后来有一天验证金额的时候，发现问题，金额最后一行没有统计，也就是少统计一个产品的金额。

插入行后公式分别为：


    =SUM(I6:I13)
    =SUM(K6:K12)


小心驶得万年船，宁愿麻烦一点，也要保证数据的准确性。

知识扩展

有的时候，统计总金额并不是在金额的下一行，而是在其他列，这样又该如何统计总金额呢？

在F1单元格中输入以下公式。


    =SUM(C:C)


或者


    =SUM(OFFSET(C2,,,COUNT(C:C)))


正常情况下，金额下面不会出现其他类型的数据，直接引用C:C就可以统计金额。当然也可以借助OFFSET函数获取求和的区域，因为是从求C2开始的区域，也就是从C2开始不偏移行和列，COUNT函数获取数字的个数也就是获取行高，知道求和区域，再嵌套SUM函数就能自动求和。

课后练习

定义一个名称，获取动态数据源，也就是现在增加一行或者一列，都能够体现出来。


 9.5　包含单位的金额求和


疑难：
 这是一份每天出差花费清单，在输入金额的时候会在后面输入单位，导致用SUM函数直接求和得不到正确答案，怎么样才能让含有单位的金额可以求和呢？

妙解：
 在C18单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。

=SUM(--SUBSTITUTE(C2:C17,"元",""))

或者


    =SUM(--LEFT(C2:C17,LEN(C2:C17)-1))


原理分析

因为包含单位“元”，则需要将单位去除掉才能求和。用SUBSTITUTE函数将元替换成空文本或者用LEFT函数提取总字符少1位，也就是只提取数字。

不管是SUBSTITUTE函数还是LEFT函数都属于文本函数，所以得到的数字也属于文本，这里叫做文本数字。数字有两种类型，一种是文本数字，另一种是真正的数字，就是数值。数值可以直接求和，而文本不能求和。如账簿上的数字跟墙上的数字是不同的，前者我们可以用这些数字进行各种分析，后者只能当欣赏用。

那有什么办法还原数字的本质呢？

把文本型转换成数值型，有专用的转换函数VALUE。


    =VALUE("12")，它的结果就是一个数值。
    =VALUE("12")=12，它的结果就是TRUE了。


在函数或公式中，运算过程会自动把文本转换为数值(一个隐含过程)，再与数值进行运算，负值运算(-)也是一种运算，能把文本转换成数值，如下所示：


    -"25"=-25


还记得负负得正吧？


    -(-"25")=-(-25)=25


简写为：


    --"25"=25


--可以把文本转换为数值，但它不是标准的转换方式，是借用负运算的隐含功能。

知识扩展

我们知道含单位对计算很不利，但假如是领导要求这么做，那你也不能违背领导的要求。这里有一种方法可以两全其美，又能容易计算，还能达到领导的要求。


STEP 01
 　利用【Ctrl】+【H】快捷键调出“查找和替换”功能，查找内容输入“元”，单击“全部替换”按钮，将单位去除。

STEP 02
 　选择C列，按【Ctrl】+【1】快捷键调出“设置单元格格式”对话框，在分类框中选择自定义，在类型中输入：G/通用格式"元"，单击“确定”按钮。

返回到工作表，可看见金额全部添加了单位，并且直接用SUM函数也可以自动求和。自定义单元格格式只是表面上看起来有单位，实际上是没有单位的，类似于障眼法。

课后练习

在记录产品重量的时候，有时以斤作为单位，有时以kg作为单位，现在要统一以斤作为单位，怎么求总重量呢？

 9.6　含姓名求总金额


疑难：
 金额跟姓名混合在一起，这样的金额又该如何统计呢？

妙解：
 在E1单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(--RIGHT(A2:A14,2*LEN(A2:A14)-LENB(A2:A14)))


或者


    =SUM(--MIDB(A2:A14,SEARCHB("?",A2:A14),15))


原理分析

因为姓名跟金额的字符数都不确定，所以不能直接提取金额出来。不过可以利用汉字是双字节，数字是单字节的特点来提取金额。

金额在右边可以用RIGHT函数，汉字的个数=总字节−总字符数，而数字的个数=总字符数−汉字的字符数，也就是：


    =LEN(A2)-文字数=LEN(A2)-(LENB(A2)-LEN(A2))=2*LEN(A2)-LENB(A2)


当然也可以用SEARCHB函数跟通配符（?）合作，确定数字的开始位置，因为数字不会超过15位，提取15位肯定能够把所有数字都提取。如果数字不足15位，后面会自动忽略，只提取数字本身位数。

不管是用哪一种方法提取出来的数字都是文本数字，不能够直接求和，需要用负运算将文本数字转换成真正的数字才可以求和。

当然转换成数字也可以用1*、0+、/1、-0等方法，只要让文本数字运算即可。

知识扩展

有的人喜欢用英文名，但只要输入英文名，上面两条公式统统没用，该如何处理呢？

英文跟数字一样都是单字节的，所以直接利用字节和字符的特点提取是行不通的。我们利用其中一条公式来提取数字，英文名会一起被提取出来。


    =MIDB(A2,SEARCHB("?",A2),15)

提取数字用LOOKUP函数最好，即使有英文也照样可以提取出来。


    =-LOOKUP(1,-RIGHT(A2,ROW($1:$15)))

不过LOOKUP函数有一个特点，就是只能提取一个数字，所以如果直接在外面嵌套SUM函数进行求和，会得到错误。


    =SUM(-LOOKUP(1,-RIGHT(A2:A14,ROW($1:$15))))


可以用LOOKUP函数直接通过辅助列将数字提取在B列，然后再通过SUM函数求和，分两步完成。


    =SUM(B2:B14)


课后练习

某老师在输入学生姓名成绩的时候，将姓名跟成绩输入在同一个单元格，输入还很整齐，如何依次求平均分、最低分、最高分呢？

 9.7　含错误值求总数量


疑难：
 数量是从别的地方引用过来的，导致有部分数据是错误值，如何避开这些错误值进行求和呢？

妙解：
 数据源包含错误值，如果直接利用下面的公式会出错。


    =SUM(C2:C17)


在F1单元格中输入以下公式。


    =SUMIF(C:C,"<9E+307")


或者在F1单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(IFERROR(C2:C17,0))


原理分析

Excel允许的最大数值是15位，9E+307是比最大数值还大的值，条件<9E+307就是包含所有数字，这样就可以排除错误值求和。数量中包含逻辑值、文本等也照样可以用这种方法，因为最大的数字都比逻辑值、文本值、错误值还小。将这些内容输入Excel中，点击降序就可以知道数字是最小的。排序：错误值>逻辑值>文本>数字。

错误值也可以用IFERROR函数屏蔽，让错误值显示0，这样就可以直接嵌套SUM函数求和。

知识扩展

针对不规范的数据源，同时出现错误值、逻辑值、文本、文本数字、普通的数字，这样如何进行求和呢？

在F1单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(IFERROR(--C2:C17,0))


看似很复杂，其实逻辑值跟文本数字通过负运算就可以直接转换成能够求和的数字，而文本通过运算转换成错误值，原来的错误值通过运算同样是错误值。这个可以在编辑栏，用F9键抹黑解读。


    =SUM(IFERROR(--C2:C17,0))
    =S U M ( I F E R R O R ({1 0;6;# V A L U E !;6;2;# N/A;1 0;0;2;4;# N/A;1;1;#VALUE!;15;3},0))


最后用IFERROR函数屏蔽各种错误值，让错误值显示0，就可以直接用SUM函数求和。

课后练习

数据源含有错误值，如何按条件统计每个人的数量呢？

 9.8　根据姓氏统计产量


疑难：
 明细表记录了每个操作人员的产量，如何根据姓氏统计产量呢？比如张，就是姓名第一个字是张的所有人员。

妙解：
 说到包含某字符，也就是使用通配符，首先想到的是*，用SUMIF函数进行条件求和。

在E2单元格中输入以下公式，并向下复制。


    =SUMIF(A:A,D2&"*",B:B)


前面说过SUM函数的数组公式，这里也照样可以用数组公式实现。

在E2单元格中输入公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =SUM((LEFT($A$2:$A$11)=D2)*$B$2:$B$11)


原理分析

对姓氏进行统计，也就是提取首个字符，然后再进行条件求和。

提取首字符，用LEFT函数，然后再进行比较，也就是：


    LEFT($A$2:$A$11,1)=D2


LEFT函数在默认情况下就是提取1位字符，参数2可以直接省略掉，就变成：


    LEFT($A$2:$A$11)=D2


因为要让满足条件的值进行统计，这时用IF函数进行判断，让满足条件的值显示本身，否则显示0。


    IF(LEFT($A$2:$A$11)=D2,$B$2:$B$11,0)

其实IF函数的参数了也可以省略掉，省略掉就是FALSE，这样也不会影响计算。

获取了满足条件的数据后，只需嵌套SUM函数即可求和。


    =SUM(IF(LEFT($A$2:$A$11,1)=D2,$B$2:$B$11))


IF函数其实有的时候也可以用*替代。


    =IF(LEFT(A2)="张",B2,0)
    =(LEFT(A2)="张")*B2


这两个公式的结果是一样的，因为满足条件就是TRUE，也就是1，不满足条件就是FALSE，也就是0，让TRUE*B2也就是等于B2的值，让FALSE*B2也就是等于0。

通过简化，最终公式就变成了：


    =SUM((LEFT($A$2:$A$11,1)=D2)*$B$2:$B$11)


再将这公式进行扩展，这样SUM函数条件求和的通用公式就出来了：


    =SUM((条件1)*(条件2)*(条件3)*...*求和区域)


这样以后不管有多少条件的求和，都可以用SUM函数的数组公式解决。

知识扩展

统计QQ号码包含某个数字的上网时间，用SUMIF函数借助通配符统计时出错，该如何解决呢？


    =SUMIF(A:A,"*"&D2&"*",B:B)

SUMIF函数使用通配符是针对文本，如果是数字，可以用SUM函数的数组公式进行解决。

在E2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制公式。


    =SUM(ISNUMBER(FIND(D2,$A$2:$A$13))*$B$2:$B$13)


FIND函数不区分数字跟文本，用FIND函数可以判断数字是否出现，出现了就返回一个数字，不出现就返回错误值。


    =SUM(ISNUMBER({2;#VALUE!;#VALUE!;6;4;#VALUE!;#VALUE!;#VALUE!;9;7;7;#VALUE!})*$B$2:$B$13)


嵌套ISNUMBER函数就是让数字显示TRUE，错误值显示FALSE。


    =SUM({TRUE;FALSE;FALSE;TRUE;TRUE;FALSE;FALSE;FALSE;TRUE;TRUE;TR UE;FALSE}*$B$2:$B$13)


反正就是一句话：哪里不会，用【F9】键抹黑解读！

课后练习

在编码和数量明细表中，求包含2或者4的编码数量之和？

 9.9　同一单元格数据累加


疑难：
 单元格内有长度不一的数据，如何将同一个单元格内的数据累加呢？如123就是1+2+3=6的结果。

妙解：
 在B2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制公式。


    =SUM(--MID(A2,ROW(INDIRECT("1:"&LEN (A2))),1))


或者


    =SUM(--(0&MID(A2,ROW($1:$15),1)))


原理分析

因为都在同一单元格，所以需要分别把数字提取出来，然后再求和。

假如现在位数确定，都是3位数，我们可用：


    =SUM(--MID(A2,ROW($1:$3),1))
    ROW($1:$3)就是提取1到3位的数字。


可是现在字符数是不确定的，也就是1到N，这个N不确定。虽然说不确定，但可以用LEN函数测量长度。要获取1到N，也就是"1:"&LEN(A2)，因为不能直接引用，所以需要嵌套INDIRECT函数才可以，最终就变成ROW(INDIRECT("1:"&LEN(A2)))。

正常的话，数字是不会超过15位，也就是用ROW($1:$15)可以将所有数字提取出来。但在单元格输入下面的公式，按【Ctrl】+【Shift】+【Enter】三键结束，却得到错误值。


    =SUM(--MID(A2,ROW($1:$15),1))


在编辑栏选择MID部分，按【F9】键解读，发现在提取的时候，当超过原来字符数时，提取出来的是空文本。


    =SUM(--MID(A2,ROW($1:$15),1))
    =SUM(--{"1";"2";"3";"";"";"";"";"";"";"";"";" ";"";"";""})


空文本如果直接运算的话，会得到错误值。


    =SUM(--{"1";"2";"3";"";"";"";"";"";"";"";"";" ";"";"";""})
    =SUM({1;2;3;#VALUE!;#VALUE!;#VALUE!;#VALUE!;#VALUE!;#VALUE!;#VA LUE!;#VALUE!;#VALUE!;#VALUE!;#VALUE!;#VALUE!})


含错误值可以用IFERROR函数屏蔽后再进行求和。


    =SUM(IFERROR(--MID(A2,ROW($1:$15),1),0))


当然这里也可以用0&这种办法，将原来的空文本转成0，数字变成01、02这种，大小都不会改变。


    =SUM(--(0&MID(A2,ROW($1:$15),1)))
    =S U M (--({"0 1";"0 2";"0 3";"0";"0";"0";"0";"0";"0"; "0";"0";"0";"0";"0";"0"}))


01、02这种通过运算就变成1、2，而0运算后依然是0。


    =S U M (--({"0 1";"0 2";"0 3";"0";"0";"0";"0";"0";"0"; "0";"0";"0";"0";"0";"0"}))
    =SUM({1;2;3;0;0;0;0;0;0;0;0;0;0;0;0})


知识扩展

有的时候单元格并非单纯的数字，也会混合一些汉字或者字母，这样的情况下该如何对同一单元格的数据累加呢？

在B2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制公式。


    =S U M (--T E X T ( M I D ( A 2 , R OW($1:$15),1),"0;;0;!0"))


直接用MID函数提取出来，会有文本跟字符，需要将非数字类型的强制转换成0，然后再求和。

数据可以分成：正数、负数、零、文本。

TEXT函数正好可以将这4种类型分开，用三个分号（;）隔开。


    正数;负数;零;文本


现在要对数字进行求和，也就是正数就不变，用0表示；这种情况下是没有负数的，因为是一个数字的累计，负数至少要两个字符，比如-2，-这个符号是当做文本处理，也就是负数可以不用考虑，空着；零是有可能出现，为了不影响计算，直接用0表示；文本不参与计算，也就是将文本转换成0，!0就是强制转换成0，感叹号（!）就是强制转换的意思，比如现在要将文本强制转换成1，就可用!1。综合起来就是：0;;0;!0。

课后练习

单元格内有长度不一的数据，现在不是要求和，而是要将所有数据进行乘积？如234就是2×3×4=24的结果。


 9.10　统计销量前5名的和


疑难：
 明细表记录着每个人的销售量，如何统计销量前5名的和呢？

妙解：
 在D2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(LARGE(B2:B17,ROW(1:5)))


原理分析

最大值用MAX函数，前几大的值用LARGE函数，函数语法如下：


    LARGE(区域,N)


比如第2大，N就是2。

现在要前5名，也就是ROW(1:5)。

知识扩展

现在增加一列城市作为条件，统计广州销量前3名的和，该如何做呢？

在E2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(LARGE(IF(A2:A17="广州",C2:C17),ROW(1:3)))


增加城市作为条件，也就是多了一个IF函数进行判断，让符合广州的显示C列对应的值。


    IF(A2:A17="广州",C2:C17)


课后练习

根据销售明细，求销量第2名的姓名？

 9.11　统计各费用项目的计划和实际的合计数


疑难：
 根据1—4月份的明细表，统计各费用项目的计划和实际的合计数。

妙解：
 在B12单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向右和向下复制公式。


    =SUM(($B$2:$I$2=B$11)*$B3:$I3)


或者


    =SUM((MOD(COLUMN($B:$I),2)=COLUMN(A1)-1)*$B3:$I3)原理分析


根据计划和实际统计，可以看成是条件统计，也可以看成是隔列统计。前文提到条件统计很多次，这里不再说明。计划是在B、D、F、H，实际是在C、E、G、I，也就是说计划在偶数列，实际在奇数列。偶数除以2的余数是0，奇数除以2的余数是1，借助这个特点就可以设置公式。

MOD函数就是求余数，语法如下：


    MOD(被除数,除数)


列号通过COLUMN函数获取，对偶数列求和就是：


    =SUM((MOD(COLUMN($B:$I),2)=0)*$B3:$I3)


对奇数列求和就是：


    =SUM((MOD(COLUMN($B:$I),2)=1)*$B3:$I3)


当然0跟1可以用COLUMN(A1)-1获取的。

知识扩展

我们经常可以看到循环生成1～N或者重复N次生成某个值这样的序列，这些是如何做到的呢？

在A1单元格输入以下公式，并向下复制。


    =MOD(ROW(),4)


循环生成1～N
 是利用MOD函数除以N的余数循环的特点，比如1，5，9除以4的余数都是1。

在B1单元格输入以下公式，并向下复制。


    =ROUNDUP(ROW()/4,0)


1～4除以4得到的结果是小于或等于1，5～8除以4得到的结果是小于或等于2，也就是让小数自动向上进位。

ROUNDUP函数就是向上进位，不管小数多少都是向上进一位，比如1.1精确到整数，也就是第二参数为0，得到的是2。

跟ROUNDUP函数相反的就是ROUNDDOWN函数，不管小数多少都是向下退1位，比如1.9精确到整数，也就是第二参数为0，得到的是1。

课后练习

如何隔N
 列求全年销售额合计呢？

 9.12　统计不重复金额的和


疑难：
 在明细表中记录着各个项目负责人消费的金额，金额有很多是重复的，如何统计不重复金额的和呢？比如2出现了4次，实际上只算1次。

妙解：
 在E2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(IF(MATCH(C2:C12,C2:C12,0)=ROW(1:11),C2:C12))


原理分析

MATCH函数是获取每个数字首次出现的位置，比如2虽然出现了很多次，但只是记录首次出现的位置，也就是第1位。


    =SUM(IF(MATCH(C2:C12,C2:C12,0)=ROW(1:11),C2:C12))
    =SUM(IF({1;2;3;1;2;3;2;8;1;8;1}=ROW(1:11),C2:C12))


MATCH=ROW就是判断数字是否首次出现，首次出现显示TRUE，否则显示FALSE。


    =SUM(IF({1;2;3;1;2;3;2;8;1;8;1}=ROW(1:11),C2:C12))
    =SUM(IF({TRUE;TRUE;TRUE;FALSE;FALSE;FALSE;FALSE;TRUE;FALSE;FALS E;FALSE},C2:C12))


IF就是让首次出现的数字返回本身，再嵌套SUM函数进行求和。

知识扩展

实际上很多情况是不重复金额的和这样的，项目名称跟负责人都相同才算重复，比如同样是2元钱，项目1张三出现了3次，项目7王五出现了1次，张三的2元就算重复，而王五的2元不算重复。现在如何根据负责人，统计不重复的项目的金额的和呢？

在F2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =SUM(IFERROR(IF(MATCH(E2&$A$2:$A$12,$B$2:$B$12&$A$2:$A$12,0)=RO W($1:$11),$C$2:$C$12),0))


因为是两个条件，先用&将负责人跟项目名称连接起来，然后用MATCH=ROW判断是否首次出现，用IF让首次出现就返回本身。因为是两个条件的查找，MATCH函数在查找的时候，找不到的对应值会显示错误值，这时需要借助IFERROR函数进行容错，让错误值显示0，然后才能用SUM进行求和。

课后练习

金额会出现一些重复值，如何统计不重复的前三大金额之和？

 9.13　统计每个应聘岗位的人数


疑难：
 在面试明细表中，记录了每一天来面试的岗位的人员信息，如何统计每个应聘岗位的人数呢？

妙解：
 在G2单元格中输入以下公式，并向下复制。


    =COUNTIF(C:C,"*"&F2)


在G2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =SUM(--(RIGHT($C$2:$C$12,2)=F2))


原理分析

正常的话，不管是什么类型的经理，我们都简称为经理，而没有去区分部门。因为经理在部门后面，也就是前面的字符不确定，用COUNTIF函数的时候可用通配符星号（*），也就是"*经理"，如果是包含经理，就用"*经理*"。现在是在单元格使用，就用&将两个内容连接起来。

而SUM函数不支持通配符的使用，应聘岗位的简称都是2位字符，并且在右边，所以用RIGHT函数提取。

前文说过SUM函数的条件求和公式为：


    =SUM((条件1)*(条件2)*(条件3)*...*求和区域)


其实如果把后面的求和区域去除掉，就是SUM函数的条件计数公式：


    =SUM((条件1)*(条件2)*(条件3)*…*(条件n))


而现在只有一个条件，这样是不能直接进行计数的，需要构成另一个条件才可以。


    (RIGHT($C$2:$C$12,2)=F2)


比如用1来做条件，也就是：


    1*(RIGHT($C$2:$C$12,2)=F2)


前文说过数组有自动扩展区域的功能，上面的公式等同于：


    {1;1;1;1;1;1;1;1;1;1;1}*(RIGHT($C$2:$C$12,2)=F2)


有的人可能对数组这样的写法还不习惯，其实这个可以转换成在单元格显示，这样就直观很多。

这样就可以进行条件计数了，其实1*和用--的作用都是一样的。

如果现在是包含简称的内容，而不是在最右边，这时就相对麻烦一点，需要借助两个函数的嵌套才可以。先用FIND函数判断是否出现，接着用ISNUMBER函数判断是否为数字，数字的话就是出现了，显示TRUE，否则就是FALSE。


    =SUM(--(ISNUMBER(FIND(F2,$C$2:$C$12))))


其实像MAX函数、COUNT函数的数组公式用法都跟SUM函数一样，这里是计数的话，也可以直接用COUNT函数处理。COUNT函数只统计数字，忽略其他。


    =COUNT(FIND(F2,$C$2:$C$12))


知识扩展

刚刚是一个应聘岗位条件，现在增加一个应召面试条件，根据这两个条件如何统计人数呢？

在G2单元格中输入以下公式，并向右和向下复制。


    =COUNTIFS($C:$C,"*"&$F2,$D:$D,G$1)


在G2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向右和向下复制。


    =SUM((RIGHT($C$2:$C$12,2)=$F2)*($D$2:$D$12=G$1))


课后练习

如何统计每天面试的人数呢？

 9.14　客户号超过15位的统计次数出错


疑难：
 客户号很长，超过15位，现在用COUNTIF函数统计每个客户号出现的次数，明细表总共才17个客户号，统计出来的次数远远不止17个，这是非常明显的错误，怎么回事呢？

妙解：
 在F2单元格输入以下公式，并向下复制。


    =COUNTIF(C:C,E2&"*")


在F2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =SUM(--($C$2:$C$18=E2))


原理分析

虽然单元格设置为文本格式，但用COUNTIF函数统计次数的时候，还是会将超过15位的部分用0显示，就如下面两个客户号，看起来是不一样的，但用COUNTIF函数统计的时候，会误认为一样，因为15位前面一样，后面全部是0，也一样。

E2&"*"就是强制性将客户号转换成文本，成为真正的文本以后，就得一模一样才会计数，这样就能统计正确。

而用SUM函数的直接比较，不会出现将客户号15位后的转换成0，所以可以直接比较统计。

知识扩展

COUNTIF函数在统计的时候，当出现通配符如星号（*）、比较符号如大于（>）时，如果直接统计同样会出错，需要做一些处理才行。

A*A实际上是2个，而用COUNTIF函数统计出来却是3个。因为星号（*）是通配符，代表所有字符的意思，也就是只要最开始是A，最后是A，中间什么字符都没关系。

[image: ]


在基础操作中，查找替换掉通配符，前面要加~，用COUNTIF函数也一样，前面加~就代表统计的是本身。函数中的替换就是用SUBSTITUTE函数。

在D2单元格中输入以下公式。


    =C O U N T I F ( A:A , S U B S T I T U T E (C2,"*","~*"))


同样道理，包含比较符的情况下，也可以用类似的方法解决统计出错的问题。

加=后，>就不是当运算符号处理，而把这个>20当成文本，所以就可以统计次数。

在D2单元格中输入以下公式。


    =COUNTIF(A:A,"="&C2)


当然对于这种，也可以直接用SUM函数的数组公式，用直接比较就可以。


    =SUM(--($A$2:$A$4=C2))
    =SUM(--($A$2:$A$7=C2))


课后练习

第1列是代号，第2列是数量，依次循环有许多列。现在需要统计全部数量中大于600的个数，该如何做呢？

 9.15　统计不重复产品名称的个数


疑难：
 产品名称有一些出现了很多次，不管出现多少次，只记1次，如何统计不重复产品名称的个数呢？

妙解：
 在单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(1/COUNTIF(A2:A18,A2:A18))


原理分析

将COUNTIF部分在编辑栏用鼠标选择，按【F9】键解读。意思就是统计区域中每个单元格出现的次数。

在编辑栏将1/{}这部分用鼠标选择，按【F9】键解读。数据太多，密密麻麻的一大堆，不太直观。

将公式输入在单元格，并将单元格格式改成分数的格式，这样就看起来直观了很多。出现2次的就变成2个1/2，出现14次的就变成14个1/14，出现1次就变成1个1，也就是不管怎么样，加起来都是1。N*(1/N)=1，这样就可以统计每个产品名称出现的次数。用SUM函数求和即可。

知识扩展

有的时候，产品名称对应的贷款用途是空白的，这又该如何统计不重复的贷款用途个数呢？

直接在单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，返回的是一个错误值。


    =SUM(1/COUNTIF(B2:B18,B2:B18))

将COUNTIF函数部分用【F9】解读，发现空白单元格统计出来的是0。我们知道1/0得到的是错误值，包含错误值是不能直接求和的。

前文提到有错误值就用IFERROR函数，这里也可以嵌套这个函数，让错误值显示0。


    =SUM(IFERROR(1/COUNTIF(B2:B18,B2:B18),0))


当然这里也可以用其他办法，比如下面这个公式：


    =SUM(1/COUNTIF(B2:B18,B2:B18&""))-1


用区域&""，这样一来空单元格也可以统计出现的次数，因为多了一个空单元格，最后再减去1就刚好是实际贷款用途的个数。

课后练习

如何统计不重复的客户号，客户号超过15位呢？

 9.16　统计组别为A3的不重复型号个数


疑难：
 组别和型号都有很多种，其中有很多都是重复的。现在如何统计组别为A3单元格，其对应的不重复型号有多少个呢？

妙解：
 在单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =COUNT(1/(MATCH(C2:C19&"A3",C2:C19&B2:B19,0)=ROW(1:18)))


原理分析

因为现在增加了一个条件，需要用&将内容组合起来，本来按理来说用COUNTIF函数就可以。


    =SUM(1/COUNTIF(C2:C19&"A3",C2:C19&B2:B19))


但是，COUNTIF函数的第一参数只能是区域的引用，而不能用C2:C19&"A3"或者{"NEW9-942BA3";"NEW9-948BA3"}，所以这个公式不成立，输入后就提示出错，只能通过其他函数来完成。

正确方法就是看C列的型号符合A3单元格在C列跟B列组合的新列的排位，有出现就算第一次出现的位置，没有出现就报错。


    MATCH(C2:C19&"A3",C2:C19&B2:B19,0)
    {# N/A;# N/A;# N/A;# N/A;# N/A;6;# N/A;# N/A;9;9;11;11;11;11;15;15;15;15}


利用MATCH只计算第一次出现位置的特点，通过跟ROW比较，如果相同就是不重复。


    MATCH(C2:C19&"A3",C2:C19&B2:B19,0)=ROW(1:18)
    {#N/A;#N/A;#N/A;#N/A;#N/A;TRUE;#N/A;#N/A;TRUE;FALSE;TRUE;FALSE;FALSE;FALSE;TRUE;FALSE;FALSE;FALSE}


到这里可以看出，只有不重复的项目才显示TRUE，其他不是FALSE就是错误，这样也不好统计TRUE的数量，那怎么办呢？干脆把不符合条件都显示错误值，那样才好统计，我们都知道1/0就是错误值，在某些程度上FALSE等同于0。


    1/(MATCH(C2:C19&"A3",C2:C19&B2:B19,0)=ROW(1:18))
    {#N/A;#N/A;#N/A;#N/A;#N/A;1;#N/A;#N/A;1;#DIV/0!;1;#DIV/0!;#DIV/0!;#DIV/0!;1;#DIV/0!;#DIV/0!;#DIV/0!}


然后再统计数字的个数即可，COUNT函数就是专门统计数字的个数。

知识扩展

现在如果要统计各种型号和组别组成的不重复，可用下面的数组公式。


    =COUNT(1/(MATCH(C2:C19&B2:B19,C2:C19&B2:B19,0)=ROW(1:18)))


这个原理跟上面的例子一样。


    =SUM(1/COUNTIFS(C2:C19,C2:C19,B2:B19,B2:B19))


我们知道COUNTIF函数是单条件计数，而COUNTIFS函数是多条件计数，现在既然是两个条件，就可以用COUNTIFS函数。

课后练习

组别的字母有大小写之分，比如C8跟c8这两个组别算不同。在区分大小写的情况下，如何统计不重复组别的次数呢？

 9.17　SUM函数的取代函数SUMPRODUCT


疑难：
 SUM函数的数组公式很强大，可以进行各种求和，但有一个比较麻烦的地方，就是每次输入公式以后都要按【Ctrl】+【Shift】+【Enter】三键结束。刚开始还经常忘记按这三键，导致结果出错，有没有其他方法来取代SUM函数的数组公式呢？

SUM函数的条件求和通用公式为：


    =SUM((条件1)*(条件2)*(条件3)*...*求和区域)


现在要统计每种俗称的尾数。

[image: ]



妙解：
 在I2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(($B$2:$B$16=H2)*$E$2:$E$16)


其实这里可以用SUMPRODUCT函数来取代SUM函数，因为SUMPRODUCT函数本身支持数组间运算，所以不需要按这三键结束。


    =SUMPRODUCT(($B$2:$B$16=H2)*$E$2:$E$16)


知识扩展

有的时候，会有将0写成零的情况，然后统计每个俗称的尾数。

在I2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =SUM(($B$2:$B$16=H2)*IF(ISNUMBER($E$2:$E$16),$E$2:$E$16))


用SUMPRODUCT函数：


    =SUMPRODUCT(--($B$2:$B$16=H2),$E$2:$E$16)


SUMPRODUCT函数除了连乘这种方法外，还有一种方法就是用逗号（,）代替乘号（*）。用逗号还有一个好处，就是SUMPRODUCT函数将非数值型的数组元素作为0处理。但用逗号必须是数字跟数字间的运算才可以，($B$2:$B$16=H2)得出来的是逻辑值，需要转换成数字才行，如用--($B$2:$B$16=H2)。SUMPRODUCT函数的更多说明见帮助：

课后练习

根据销售明细表，如何统计每天的销售金额（数量*单价）呢？

第10章　提取不重复值及筛选符合条件的项目


 10.1　提取不重复公司名称


疑难：
 这是一份销售明细表，记录着每一家公司的销售情况，但有很多公司会重复出现，如何提取不重复的公司名称，方便后期做汇总呢？

妙解：
 在H2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =OFFSET($A$1,MATCH(0,COUNTIF(H$1:H1,$A$2:$A$19),0),0)&""


原理分析

次数公式为=COUNTIF(H$1:H1,$A$2:$A$19)，也就是统计条件区域$A$2:$A$19，在原来区域H$1:H1的次数，因为全部没有出现，所以都是0。

排位公式为=MATCH(0,J2:J19,0)，MATCH函数是查找首次出现的0的位置，也就是第1位。

公司名称公式为=OFFSET(A1,L2,0)，OFFSET函数获取A1向下1个的单元格，就是鑫宝。

提取第1个公司名称后，公式下拉区域就发生了一个小小的变化，H$1:H1变成H$1:H2，因为鑫宝已经出现了，所以次数就显示1，没出现的依然显示0，这样用MATCH函数就顺利找到新的0的位置，也就是第2位，嵌套OFFSET函数就能获取相应的对应值。

依次类推，就能够获取所有的对应值。

COUNTIF函数的条件区域是$A$2:$A$19而不是$A$2:$A$18，也就是比原始区域大，这个很关键。因为当不重复的公司都提取完的时候，用COUNTIF函数统计，也就是所有都是1，这时用MATCH函数来查找就会返回错误值。而区域大一点，因为是空单元格，所以最后一个永远是0，这样就可以用MATCH查找到。我们知道A19是空单元格，但如果输入=A19会得到0，而如果输入=A19&""会得到一个空单元格。同理，OFFSET后面&""就是为了让0显示空单元格。

知识扩展

有的时候购买产品的是一些个人而不是公司，这时就没有公司名称，用前文的数组公式就行不通，遇到这种情况又该如何处理呢？

[image: ]


在H2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =INDEX(A:A,MIN(IF((COUNTIF(H$1:H1,$A$2:$A$18)=0)*($A$2:$A$18<>""),ROW($2:$18),4^8)))&""


区域中有空单元格就出错，也就是说要增加一个判断，让区域不能为空。


    $A$2:$A$18<>""


同时加上原来的条件，COUNTIF=0，因为是两个条件，所以用*将两个条件连接起来。


    (COUNTIF(H$1:H1,$A$2:$A$18)=0)*($A$2:$A$18<>"")



注意
 　在用数组公式的时候同时满足不能用AND函数，要用*；同理，用数组公式的时候或者不能用OR函数，要用+。


用IF函数让满足条件的值显示本身行号，不满足则显示4^8。4^8可以换成其他比较大的数，只要比原来区域大就可以，比如99。

MIN函数就是获取没有出现的最小行号，而INDEX函数就是获取行号所对应的值，公式最后&""就是让0显示为空单元。

课后练习

正常我们提取不重复值都是按原顺序排序的，如是反过来，从最后一个开始提取不重复值，又该如何做呢？

 10.2　将不重复数据从大到小排序


疑难：
 在一列数据清单中，如何提取不重复值，并将数据从大到小排序呢？

妙解：
 在D2单元格中输入以下公式，并向下复制，直到出现第一个错误值为止。


    =LARGE($A$2:$A$15,COUNTIF($A$2:$A$15,">="&D1)+1)

原理分析


    =COUNTIF($A$2:$A$15,">="&D1)


将所有值跟D1进行比较，因为D1是文本，没有一个值能满足条件，也就是所有值都小于D1，最大值就是第一大，也就是COUNTIF得出来的结果+1。


    =LARGE($A$2:$A$15,COUNTIF($A$2:$A$15,">="&D1)+1)


也就是说，第一个返回最大值为9。


    =COUNTIF($A$2:$A$15,">="&D2)


因为有两个9，所以≥9的值有两个，COUNTIF的结果+1就是第二大所对应的值为8。

依次类推，就得到所有值。

知识扩展

如果是从小到大排序又该如何做呢？

在D2单元格中输入以下公式，并向下复制，直到出现第一个错误值为止。


    =SMALL($A$2:$A$15,COUNTIF($A$2:$A$15,"<="&D1)+1)

与LARGE函数相反的就是SMALL，从大到小是用>=的符号，从小到大就用<=，用相反的就没错。

这样虽然做出来了，但有错误值，看起来还是不顺眼。之前提到错误值，首先想到的都是用IFERROR这个函数来进行容错，这里是否同样可以呢？不妨来试试。

结果发现嵌套IFERROR函数是行不通的，空了一个值后又重复从小到大显示。

其实我们也可以换个思路，比如有6个不重复值被提取完，从第7个开始就想让它不显示。也就是行号大于不重复值的总数就不显示。

统计不重复值的公式，前面已经提过：


    =SUMPRODUCT(1/COUNTIF($A$2:$A$15,$A$2:$A$15))


行号与不重复值的总数比较就用：


    =ROW(A1)>SUMPRODUCT(1/COUNTIF($A$2:$A$15,$A$2:$A$15))


综合起来就是：


    =IF(ROW(A1)>SUMPRODUCT(1/COUNTIF($A$2:$A$15,$A$2:$A$15)),"",SMA LL($A$2:$A$15,COUNTIF($A$2:$A$15,"<="&D1)+1))


课后练习

在比赛中计算成绩时都会扣除一些异常值，比如前几个最大值跟前几个最小值，然后根据中间这些分数再进一步获取评分。现在如何获取去除前2个最大值跟前2个最小值后的平均分？

 10.3　提取只去过1次的学校


疑难：
 公司规定每家学校在指定的时间内，必须去2次以上（含2次），否则这些学校的业务会重新分配。如何将不满足条件的学校提取出来，也就是提取只去过1次的学校呢？

妙解：
 在E2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =INDEX(B:B,SMALL(IF(COUNTIF($B$2:$B$18,$B$2:$B$18)=1,ROW($2:$18 ),4^8),ROW(A1)))&""


原理分析


    COUNTIF($B$2:$B$18,$B$2:$B$18)


统计每个学校出现的次数。


    COUNTIF($B$2:$B$18,$B$2:$B$18)=1


判断次数是否等于1，满足显示TRUE，不满足显示FALSE。


    IF(COUNTIF($B$2:$B$18,$B$2:$B$18)=1,ROW($2:$18),4^8)


让满足条件的值显示本身的行号，不满足显示4^8。


    SMALL(IF(COUNTIF($B$2:$B$18,$B$2:$B$18)=1,ROW($2:$18),4^8),ROW (A1))


得到第一小的值的行号，通过下拉依次获取第二小、第三小、第N小值的行号。


    INDEX(B:B,SMALL(IF(COUNTIF($B$2:$B$18,$B$2:$B$18)=1,ROW($2:$18) ,4^8),ROW(A1)))&""


让行号返回本身的对应值，&""就是让0显示空白单元格，进行容错处理。

知识扩展

在很多时候为了记录方便都是用二维表格记录，有记录就打勾，这种形式的表格又该如何处理呢？

在B19单元格中输入以下公式，并向右复制。


    =COUNTIF(B2:B18,"√")


在L2单元格中输入公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =INDEX($1:$1,SMALL(IF($B$19:$J$19=1,COLUMN(B:J),200),ROW (A1)))&""


先借助辅助列，用COUNTIF函数统计每个学校去的次数，再用数组公式获取对应值。现在学校是第1行，也就是要返回列号才可以获取，这时就用COLUMN函数获取列号。用列号的时候一般都不会有太多列，所以让不满足的显示200，这个数字就足够大，其他的跟前面一样。

当然不使用辅助列也可以，不过公式会变得复杂很多。


    =INDEX($1:$1,SMALL(IF(MMULT(TRANSPOSE(ROW(2:18)^0),--($B$2:$J$1 8="√"))=1,COLUMN(B:J),200),ROW(A1)))&""


在实际工作中尽量以简单为主，不建议使用复杂的公式解决问题。

课后练习

将只去过1次的学校在C列进行备注。

 10.4　查询工程师对应的所有奖金


疑难：
 在明细表中记录了不同职称对应的奖金，现在想查询某个职称对应的所有奖金，如工程师，该如何处理呢？

[image: ]



妙解：
 在E2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =IFERROR(INDEX(B:B,SMALL(IF($A$2:$A$11=$D$2,ROW($2:$11)),ROW (A1))),"")


原理分析


    $A$2:$A$11=$D$2


将职称区域的每个值跟D2进行比较，如果符合条件，就返回TRUE，否则返回FALSE。


    IF($A$2:$A$11=$D$2,ROW($2:$11))


嵌套IF函数，让符合条件的值显示本身行号。

可能有读者会产生疑问，前面一个例子是用下面这个公式，让符合条件的显示本身行号，不符合的显示4^8。


    IF($A$2:$A$11=$D$2,ROW($2:$11),4^8)


那就沿用旧思路先做说明，如果用4^8，那后面就是：


    =INDEX(B:B,SMALL(IF($A$2:$A$11=$D$2,ROW($2:$11),4^8),ROW (A1)))&""


先用SMALL跟ROW依次获取行号，然后嵌套INDEX函数获取行号的对应值，最后用&""进行容错处理。

这时会出现一个小小的问题，因为使用了&""导致奖金全部左对齐，默认情况下，左对齐就是文本，右对齐就是数值。也就是说现在的奖金全部变成文本了，正常情况下文本是不能求和的。

在全部是文本的时候采用不符合条件返回4^8，在公式后面的&""进行容错。当数据源出现数值或者日期的时候，就得采用其他方法。

因为反正都会出错，那就直接用：


    IF($A$2:$A$11=$D$2,ROW($2:$11))


接着就用SMALL跟ROW依次返回行号，再用INDEX函数获取对应值。因为不满足的条件会返回错误值，这时在最外面嵌套IFERROR函数进行容错，这个函数不会改变数据的本质。

知识扩展

现在职称多了几个，假如要获取工程师跟会计2个职称的所有对应资金，又该如何做呢？

正常我们的反应就是用OR函数将2个条件组合起来，然后用公式。


    OR($A$2:$A$14="工程师",$A$2:$A$14="会计")


在D2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下和向右复制。


    =IFERROR(INDEX(A:A,SMALL(IF(OR($A$2:$A$14="工程师",$A$2:$A$14="会计"),ROW($2:$14)),ROW(A1))),"")


这时奇怪的一幕发生了，所有明细表的数据都出现了，怎么回事呢？

我们先将OR部分在单元格输入，发现全部返回TRUE，而不是每个单元格进行判断。也就是OR函数不能用在数组上，这时要采用别的方法来替代。在普通公式上，我们可以用+代替OR，我们这里也尝试用+看看结果怎么样。

可以发现只要是工程师或会计的都显示1，其他显示0。1等同于TRUE，0等同于FALSE。

也就是说这里将OR换成+即可。


    =IFERROR(INDEX(A:A,SMALL(IF(($A$2:$A$14="工程师")+($A$2:$A$14="会计"),ROW($2:$14)),ROW(A1))),"")


课后练习

在职称跟奖金明细表中，获取职称为工程师，奖金大于或等于100所对应的所有数据。

 10.5　依次查找每个店面对应的品牌和数量


疑难：
 在销售明细表中，依次查找每个店面对应的品牌和数量，该如何做呢？

妙解：
 在F2单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下和向右复制。


    =IFERROR(INDEX(B:B,SMALL(IF($A$2:$A$17=$E2,ROW($2:$17)),COUNTIF ($E$2:$E2,$E2))),"")


原理分析

正常查询对应值，都会想到用LOOKUP函数或者VLOOKUP函数，但LOOKUP函数如果有多个对应值时只返回最后一个满足条件的值，VLOOKUP函数如果有多个对应值时只返回第一个满足条件的值。


    =LOOKUP(1,0/($A$2:$A$17=$E2),B$2:B$17)
    =VLOOKUP($E2,$A:$C,COLUMN(B1),0)

如果是只有一个对应值，用这两个函数确实是不错的选择，但现在都出错了。

出现第几次可以用COUNTIF函数进行累计：


    =COUNTIF($E$2:$E2,$E2)


通过下拉，区域不断变大，依次变成$E$2:$E3、$E$2:$E4……$E$2:$E8。


    =IFERROR(INDEX(B:B,SMALL(IF($A$2:$A$17=$E2,ROW($2:$17)),COUNTIF ($E$2:$E2,$E2))),"")


筛选公式最大的好处就是模式固定，只需改变条件或者返回第N个值两部分，只要理解了一个，其他问题就能很好解决。

知识扩展

除了上面的依次查找对应值外，筛选公式还有一种特殊的用法，就是按指定次数重复文本。如按B列的次数，重复显示店面。

在D2单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，并向下复制。


    =INDEX(A:A,SMALL(IF($B$2:$B$5>=COLUMN($A:$Z),ROW($2:$5),4^8),RO W(A1)))&""

这里关键是构成条件$B$2:$B$5>= COLUMN($A:$Z)，看B列的区域大于列号的有几个数，COLUMN($A:$Z)就是代表1～26，因为重复次数小于10，范围写大点比较有保证。

课后练习

在销售明细表中，查询每个店铺对应的所有品牌。

第三篇　活运用

函数与公式很强大，可以处理Excel中绝大部分的数据，同时还能辅助图表、数据透视表等。但在实际工作中，要根据实际情况而决定解决方案，哪一种方案简单快捷，就用哪一种解决，千万不能为了使用函数与公式而将原本简单的问题复杂化。函数与公式只是给你提供了一种解决问题的思路而已。



第11章　数据处理

Excel有四个作用：数据储存、数据处理、数据分析、数据展现。函数与公式的主要用途就是对储存的数据进行处理，从而辅助图表展现、分析数据。这一节主要是介绍利用函数与公式对工作中的案例进行处理，从而获得需要的结果。


 11.1　标准日期格式大转换


案例：
 每个公司都有自己一套表示日期的方法，但如果日期不参与运算的话，怎么表示都无所谓，一旦要参与运算，不规则的日期会带来很大的不便。先看看笔者公司的日期表示法：

说到日期，其实有专门的函数，NOW，TODAY，YEAR，MONTH，DAY，DATEDIF等，不过这里主要说明TEXT函数的运用。


效果1：
 日期以分数效果显示，只有月跟日。


    =TEXT(B4,"m/d")



效果2：
 日期显示成年月日各2位数。


    =TEXT(B5,"yymmdd")



效果3：
 日期显示成年月日各2位数，但年份是实际年份的2倍，这是日本的特殊年份表示法。


    =2*TEXT(B6,"yy")&TEXT(B6,"mmdd")



效果4：
 日期显示年为1位，月日为2位数，后面加720，这个是包装部的标志。


    =TEXT(B7,"ymmdd!7!2!0")
    =TEXT(B7,"ymmdd”)&"720"



效果5：
 日期显示成年月日各2位数+当天日期的月日各2位数。


    =TEXT(B8,"yymmdd")&TEXT(NOW(),"mmdd")
    =TEXT(B8,"yymmdd")&TEXT(TODAY(),"mmdd")



效果6：
 下面这种也属于标准日期的一种。


    =TEXT(B9,"e年mm月dd日")
    =TEXT(B9,"yyyy年mm月dd日")



效果7：
 日期以“.”分割符隔开。


    =TEXT(B10,"e.mm.dd")
    =TEXT(B10,"yyyy.mm.dd")


原理分析

TEXT函数的第二参数其实在这里的作用是设置“单元格格式”，通过不断地改变第二参数得到各种各样的效果。

y其实是year的缩写，代表年份，1个y就代表年份为1位，2个y就代表年份为2位。

m其实是month的缩写，代表月份。

d其实是day的缩写，代表日。

e这里的意思是代表4位数的年份，等同于yyyy。

now是获取系统时间，today是获取系统日期。系统时间就是你计算机设置的时间，只有当系统时间设置正确的时候，才等同于当天的时间。

感叹号（!）是干什么用的呢？其实在这里是强制显示一些没法显示的字符，如720在这里本来是无法显示的，直接写720返回错误值，这时在前面添加“!”就可以显示。

这么多单元格格式，记不下怎么办呢？

其实这些可以不用记的，借助自定义单元格格式能减轻我们的记忆负担。

利用【Ctrl】+【1】快捷键调出“自定义单元格格式”对话框，通过设置单元格格式，从而获取自定义格式代码。

假如现在要将日期显示为星期几的形式，操作方法如下。


STEP 01
 　单击“日期”→星期三。

STEP 02
 　单击“自定义”就可以看到我们刚刚设置的单元格格式代码。

STEP 03
 　复制代码，然后设置公式，就可以获取星期几。


    =TEXT(B4,"[$-804]aaaa;@")


以后就不用辛辛苦苦去记这些了。

知识扩展

每个公司都有自己的一套表示日期的方法，上面几种就是笔者公司所有日期的形式。如果不处理数据，这样也没什么问题，假如用数据透视表，这种格式的日期是没办法自动组合的，会造成很大的麻烦！

Q：
 怎么将下面的日期格式转换成2014-10-09这种格式的日期呢？


A：
 大部分不规范的日期都可以通过分列变成标准日期。


STEP 01
 　选择区域A2:A6，单击“数据”选项卡→“分列”。

STEP 02
 　弹出“文本分列向导”对话框，保持默认不变，连续单击2次“下一步”按钮。列数据格式选择“日期”，更改目标区域为$B$2，单击“完成”按钮。

STEP 03
 　选择区域B2:B6单元格，将单元格格式改为“短日期”。

经过这3步处理，日期就变成了标准日期，可以进行后期统计，如数据透视表按年月组合。

 11.2　统计指定日期的出货数


案例：
 卢子每天都会将产品的出货检查记录在下表，然后根据明细表进行统计，最后将统计的结果以月报的形式交给领导。

效果1：
 统计2010年和2011年的出货数

在H3单元格输入以下公式，下拉到H4单元格。


    =SUMPRODUCT((YEAR($A$2:$A$35)=G3)*$D$2:$D$35)
    =SUMPRODUCT((--TEXT($A$2:$A$35,"e")=G3)*$D$2:$D$35)



效果2：
 统计2011年4月份～12月份的出货数

在L3输入以下公式，双击单元格填充公式。


    =SUMPRODUCT((--TEXT($A$2:$A$35,"emm")=K3)*$D$2:$D$35)



效果3：
 统计2010年和2011年4～12月份的出货数

在P3单元格输入以下公式，向右拉到Q3，双击单元格填充公式。


    =SUMPRODUCT((TEXT($A$2:$A$35,"e年")=P$2)*(TEXT($A$2:$A$35,"m月")=$O3)*$D$2:$D$35)

原理分析

这里涉及了一个条件求和的通用公式：


    =SUMPRODUCT((条件1)*(条件2)* (条件3)*…*求和区域)


通过TEXT函数得到的年月都是文本型数字，跟直接录入到单元格的数值不同。文本型数字就等同于写在黑板上的艺术字，我们关心的是它显示的效果而不是其本身的大小。数值就等同于记录在账目上的数字，我们关心的是它的大小，要用这些来进行统计分析。也就是说，这两种形式的数字，表面上看似一样，实质是不同的。

一起来了解“减负”运算：

在函数与公式中，运算过程会自动把文本型数字转换为数值（一个隐性过程），再与数值进行运算。负值运算（-）其实也是一种运算，能把文本型数字转换成数值。


    -"201104"=-201104


而负负得正


    -(-"201104")=-( -201104)=201104


也就是说


    --"201104"=201104


其实负负运算称为减负运算更好，即减去数字的负担，还原数字的本质。

知识扩展


Q：
 除了TEXT函数外，还有哪些函数得到的数字是文本型数字，数值又如何转换成文本型数字呢？


A：
 通过文本函数得到的数字都是文本型数字，单击“公式”选项卡→“文本”，出现的这些函数都是文本函数。常用的就是LFET、MID、RIGHT三个字符提取函数。

数值转换成文本型数字只要用&任意字符都可以，最常用的就是&""。

 11.3　年度数据统计


案例：
 月度统计表一般都只是交给上级领导，而年度统计表则是要交给老板，性质就不同了。在统计的时候，卢子都是尽可能全面统计，让老板可以随着查看到所需要的数据。

效果：根据明细表统计2011年每个月份的检查天数、检查时间（h）、出货数、检查数、不良数、不良率。

检查天数是根据年月统计不重复的天数，直接统计稍微有难度，这里采用辅助列方法。

在F2单元格中输入以下公式，双击填充公式。


    =COUNTIF(A$2:A2,A2)=1


借助COUNTIF的累积计数法，出现1次就是首次出现，跟1比较，首次出现就显示TRUE，非首次出现就显示FALSE。

检查天数：在I3单元格中输入以下公式，双击填充公式。


    =SUMPRODUCT((TEXT($A$2:$A$35,"e年m月")=$H$2&H3)*$F$2:$F$35)


检查时间（h）：在J3单元格中输入以下公式，双击填充公式。


    =R O U N D ( S U M P R O D U C T ( ( T E X T ($A$2:$A$3 5 ,"e年m月")=$H$2&H3)*$B$2:$B$35)/60,0)


出货数：在K3单元格中输入以下公式，双击填充公式。


    =SUMPRODUCT((TEXT($A$2:$A$35,"e年m月")=$H$2&H3)*$D$2:$D$35)


检查数：在L3单元格中输入以下公式，双击填充公式。


    =SUMPRODUCT((TEXT($A$2:$A$35,"e年m月")=$H$2&H3)*$E$2:$E$35)


不良数：在M3单元格中输入以下公式，双击填充公式。


    =L3-K3


不良率：在N3单元格中输入以下公式，双击填充公式。


    =IF(L3=0,"",TEXT(M3/L3,"0.0%"))

原理分析

这里出现了三个函数：COUNTIF、ROUND、IF，下面来看这些函数的语法。

COUNTIF函数语法：


    =COUNTIF(去哪里数，按什么条件数)


如统计番号40065出现的次数，也就是去C:C这个区域找，条件为40065。


    =COUNTIF(C:C,40065)


ROUND函数语法：


    =ROUND(表达式或单元格,N)


N就是精确到小数点N位，并四舍五入。以0为分界点，0就是没有小数点，2就是小数点2位，-2就是十位数的四舍五入。负数用得比较少，举个例子来说明。


    =ROUND(123.45,-2)


-2表示十位数四舍五入，十位数为2，小于5，所以舍去，结果为100。


    =ROUND(153.45,-2)


同理，结果为200。

IF函数语法：


    =IF(条件判断，如果条件成立的值，反之（即条件不成立）的值)


IF函数是一个使用频率非常高的函数，经常借助这个函数进行判断。如成绩大于80的显示优秀，其他显示一般。


    =IF(A1>80,"优秀","一般")


知识扩展


Q：
 SUMPRODUCT函数这么牛，各种条件求和都能搞定，那它可以进行条件计数吗？


A：
 答案是肯定的，SUMPRODUCT函数是一个非常好用的函数，求和计数都可以。称它为条件求和、计数之王也不过分。

这里涉及了一个条件计数的通用公式：


    =SUMPRODUCT((条件1)*(条件2)* (条件3)*...*(条件N))

1．根据明细表统计科目划分的笔数，在I2单元格中输入以下公式，双击填充公式。


    =SUMPRODUCT(--($E$2:$E$53=H2))


2．根据明细表统计各部门科目划分的笔数，在J9单元格中输入以下公式，双击填充公式。


    =SUMPRODUCT(($D$2:$D$53=H9)*($E$2:$E$53=I9))


其实条件计数的函数是COUNITF，多条件计数的函数是COUNTIFS。设置公式不一定要用某个函数，而是要用最合适（熟悉）的函数。


 11.4　年度数据核对


案例：
 完成年度统计并不是就可以了，还需要对数据进行核对，小心驶得万年船。

1．核对明细表的数据录入有没差错。

2．核对汇总的数据是否正确。

现在是统计2011年的每个月数据，也就是只要筛选出2011年的数据即可，然后根据筛选出来的数据进行逐条核对。


STEP 01
 　单击A1单元格，然后单击“数据”选项卡→“筛选”。

STEP 02
 　因为这里的日期是标准日期，会直接按年月进行筛选，方便很多。单击“日期”的筛选按钮，取消“2010”的勾选，单击“确定”按钮。这样就只显示2011年的数据。

现在问题来了，如何对这些筛选出来的数据进行统计呢？

SUBTOTAL函数就是专门针对筛选的数据进行统计的。

在B36单元格输入以下公式，向右拖动公式到E36单元格，然后删除C36的公式。


    =SUBTOTAL(9,B2:B35)


如果显示要统计4月份的，只需对4月份的数据进行筛选即可。

原理分析

SUBTOTAL函数语法：


    =SUBTOTAL(N,区域)



N
 就是代表各种汇总的形式，也就是说这个函数在筛选的状态下可以取代11个函数的功能，如1就是平均值AVERAGE，9就是求和SUM。

包含隐藏值和忽略隐藏值在这里举个例子说明一下，将数据复制到新表，然后对其中的一些行进行隐藏。参数为9就是不管你隐藏与否都会自动统计，而参数为109就是只统计看得见的数据。

知识扩展


Q：
 如何在筛选状态下，获取连续序号呢？

A：
 SUBTOTAL函数专门针对筛选这种状态，也就是说这里可以借助这个函数来达到目的。生成序号，其实就是累计筛选状态下有多少个非空单元格，也就是参数1设置为3，或统计非空单元格COUNTA函数。

在A2单元格中输入以下公式，双击填充公式。


    =SUBTOTAL(3,B$2:B2)


不过在对部门进行筛选状态下会出现一个小小的问题，那就是在最后一行出现了一个其他部门。

Excel也会有出错的时候。


解决方案1
 　在最后一行录入一个空格，然后再进行筛选。


解决方案2
 　对公式做一个小小的变动，在后面加*1。


    =SUBTOTAL(3,B$2:B2)*1


如果涉及隐藏行的话，就将3改成103。


    =S U B T O T A L ( 1 0 3 , B$2:B2)*1


笔者比较推崇用种方案2，在能不更改数据源的情况下尽量不要更改。

 11.5　隆成小天使订单跟出货情况


案例：
 笔者自从负责隆成小天使业务后，每次都会将出货明细一条条清楚地记录下来。每一个订单的数量是固定的，出货数不能大于订单数，也不能小于订单数。只要完成就显示OK，否则就显示NG。这样每回出货前笔者心里就有数，知道还有那些订单没完成，还剩下多少尾数。

要制作这份表会涉及很多问题：

问题1：制作日期要动态显示，每天更新，还要显示星期几。


    =TEXT(NOW(),"e.mm.dd aaaa")


问题2：根据番号获取俗称。


    =IFERROR(VLOOKUP(B6,表1[#全部],2,),"")


问题3：统计出货数。


    =IF(F6="","",SUM(F6:L6))


问题4：统计剩下的尾数。


    =IF(M6="","",D6-M6)


问题5：判断完成情况，尾数为0就是OK，否则NG。


    =IF(N6="","",IF(N6=0,"OK","NG"))原理分析


这里涉及了IFERROR和VLOOKUP函数。

IFERROR函数语法：


    =IFERROR(表达式,错误显示值)


也就是说IFERROR函数是为了进行容错处理，让错误值显示成自己想要的结果。

举个简单的例子来说明一下，现在要计算产品的完成率，当订单数还没有输入时，直接在D2输入以下公式，下拉到D5就显示错误值。


    =C2/B2


这时就可以用IFERROR函数来处理，让错误值显示“未录入订单数”。


    =IFERROR(C2/B2,"未录入订单数")


[image: ]


当然对这种数据进行容错还有很多种方法，这里就不一一罗列。

VLOOKUP函数语法：


    =VLOOKUP(要查找的值, 到哪里去找, 带回什么,匹配方式（精确/模糊）)


VLOOKUP函数是一个很神奇的函数，要必须掌握。下面通过一个简单的例子来说明一下。

根据学生姓名在学生成绩表查找总分。


    =VLOOKUP(F3,$B$3:$D$16,2,0)

参数1：
 要查找的值就是姓名，也就是F3。


参数2：
 到哪里去找就是学生成本表，也就是$B$3:$D$16这个区域。


参数3：
 带回什么就是总分，总分在$B$3:$D$16这个区域是第2列，也就是写2。


参数4：
 匹配方式（精确/模糊），正常情况都是精确查找，也就是写0。

说明：查找的值要在查找的区域中的首列，否则查找不到对应值，所以区域没写$A$3:$D$16而是从B列写起。

经过了前面2个例子的说明，下面这个公式的意思就可以知道了：B6在表1[#全部]的第2列进行查找，查找不到对应值显示空白。


    =IFERROR(VLOOKUP(B6,表1[#全部],2,),"")


知识扩展


Q：
 查找了整个工作簿，只有3个工作表：小天使、对应表、小天使总体情况，都没找到这个“表1[#全部]”区域在哪里，怎么回事呢？

A：
 这个是Excel中的“表格”功能，跟传统的工作表不一样。只要用鼠标选取“表格”的区域，就会自动出现“表1[#全部]”。

单击A1单元格，单击“插入”选项卡→“表格”，弹出“创建表”对话框，保持默认不变单击“确定”按钮。这样就创建了一个“表格”。

“表格”的三大妙用：

1．实现隔行填充色，防止看错行。

2．利用汇总行，实现可以统计。

单击“表格”任意单元格，然后单击“设计”选项卡，勾选“汇总行”，自动对俗称进行计数。公式跟我们手写的SUBTOTAL函数差不多。


    =SUBTOTAL(103,[俗称])

单击公式所在单元格，还可以选择各种统计方式，如平均值、最大值、最小值。

[image: ]


3．利用“表格”创建“数据透视表”，实现动态数据源。


 11.6　各产品订单跟出货汇总


案例：
 完成对隆成小天使订单跟出货的录入只是第一步，对各产品订单跟出货进行汇总才是目的，这样可以更直观地看出各产品的总体情况。

要完成各产品订单跟出货的汇总，需要搞定下面几个问题：

1．获取不重复产品，也就是俗称。

提取不重复的方法有很多，一般笔者都是采用“删除重复项”这个功能，方便快捷。

复制“俗称”这一列，放在新表格的B列，粘贴成值。单击“数据选项卡”→“删除重复项”，弹出“删除重复项”对话框，单击“确定”按钮。这样就获取了不重复的番号。

其实不管是什么方法都是针对项目比较多的，而小天使这边的产品非常少，实际笔者是采用手工输入。

2．完成数量就是指对出货数量的汇总，在C2输入以下公式并向下复制。


    =SUMIF(小天使!C:C,B2,小天使!M:M)


3．订单数就是指对订单数的汇总，在D2输入以下公式并向下复制。


    =SUMIF(小天使!C:C,B2,小天使!D:D)


4．尾数就是完成数量减去订单数，在E2输入以下公式并向下复制。


    =D2-C2


5．完成批数就是统计产品有多少批是OK的，在F2输入以下公式并向下复制。


    =COUNTIFS(小天使!C:C,B2,小天使!O:O,"OK")


6．批数就是统计产品有多少批，在G2输入以下公式并向下复制。


    =COUNTIF(小天使!C:C,B2)


7．未完成批就是批数减去完成批数，在H2输入以下公式并向下复制。


    =G2-F2


8．合计就是对上面统计的东西进行汇总，在C8输入以下公式并向右复制。


    =SUM(C2:C6)


原理分析

这里涉及了SUMIF、COUNTIFS函数，当然也可以直接用SUMPRODUCT函数取代，不过这里不需要根据日期统计，用这个并不是最便利。

SUMIF函数语法：


    =SUMIF(去哪里,找什么,把哪里加起来)


COUNTIFS函数语法：


    =COUNTIFS(去哪里数1,按什么条件数1, 去哪里数2,按什么条件数2,…, 去哪里数n,按什么条件数n)


这里两个函数比较简单，这里就不举例说明。

知识扩展


Q：
 在不使用SUMPRODUCT函数的前提下，能用SUMIF函数进行多条件求和吗？如果不能可以用什么函数进行多条件求和呢？


A：
 正确情况下SUMIF函数是不能实现多条件求和的，但也有特殊情况。如统计发生额在30到40之间的总发生额。

    =SUMIF(F:F,">=30",F:F)-SUMIF(F:F,">40",F:F)
    =SUMIF(F:F,">=30")-SUMIF(F:F,">40")


当参数1跟参数3的区域一样时，可以将参数3省略点。

公式理解：[30,无穷大)跟(40,无穷大)的交集是(40,无穷大)，[30,无穷大)减去交集(40,无穷大)刚好是[30,40]这个区间，所以>=30的总和减去>40的总和就是30至40的总和。

这种属于特殊情况，其实多条件求和用SUMIFS更合适。


    =SUMIFS(F:F,F:F,">=30",F:F,"<40")


SUMIFS函数语法：


    =SUMIFS(把哪里加起来,去哪里1,找什么1,去哪里2,找什么2,…, 去哪里n,找什么n)


再来看一个例子加深印象，统计各部门科目划分的发生额。


    =SUMIFS(F:F,D:D,H2,E:E,I2)


[image: ]



 11.7　根据AQL抽样基准确认抽检数量


案例：
 AQL是ACCEPTANCE QUALITY Level 接收质量限的缩写，即当一个连续系列批被提交验收时，可允许的最差过程平均质量水平。 AQL普遍应用于出口服装、纺织品检验上，不同的AQL标准应用于不同物质的检验上。在AQL 抽样时，抽取的数量相同，而AQL后面跟的数值越小，则允许的瑕疵数量就越少，说明品质要求越高，检验就相对较严。

验货抽样依据是根据AQL值决定抽样的数量和合格与不合格产品的数量。

问题：如何用AQL抽样基准确认每个出货数量的抽检数量呢？如果出货数少于20，实行全数检查。

在A列建立一个辅助列，输入出货数量区间的下限值。

在W4单元格中输入以下公式，并向下复制。


    =MIN(V4,VLOOKUP(V4,$A$5:$H$9,6))


原理分析

VLOOKUP函数的第四参数省略，就是模糊查找，也就是按区间查找对应值。如123大于1，小于501，就返回1的对应值，也相当于是返回1～500这个区间的对应值。同理，1000大于501，小于1201，就返回501的对应值，也相当于501～1200这个区间的对应值。

在VLOOKUP外面嵌套MIN函数有什么特殊含义呢？


    =VLOOKUP(V4,$A$5:$H$9,6)


如果没有MIN函数，当出货数小于20时，依然返回20。如出货数量只有15，抽检数量是不可能大于出货数量了。

MIN函数是最小值的意思，15跟20比较就返回15。


    =MIN(15,20)

有的人也许会想到直接用出货数量这个区间进行查找，但出货区间是文本，Excel并不能识别为数值，也就是说无法跟数值比较。如果直接查询，会返回错误值。


    =VLOOKUP(V4,$B$5:$H$9,5)

知识扩展


Q：
 如何用公式提取区间的下限，也就是“～”之前的数字？

A：
 字符提取一般都是借助LFET（左边）、MID（中间）、RIGHT（右边）三个函数提取，现在要提取～之前的数字，也就是要从左边提取，就得用LEFT函数。


    =--LEFT(A2,FIND("～",A2)-1)


LEFT函数语法：


    =LEFT(字符串,N)


也就是从左边开始连续提取N位字符，现在的N是由～确定的，也就是知道这个才能确定N。

FIND函数语法：


    =FIND(查找的字符,字符串)


作用是查找某个字符在字符串中的位置，FIND("～",A2)就是查找～在A2的位置。数字的位置就是～之前的1位，也就是减去1。

前面说过用文本函数提取出来的数字叫文本数字，通过减负（--）运算可以转换成数值。


 11.8　根据AQL抽样基准确认允许最大不良数


案例：
 确认完抽检数量还不行，还得根据AQL抽样基准确认各不良等级允许最大不良数，根据这个才能确认产品是否合格，合格了才能出货。

在W4单元格中输入以下公式，并向右和向下复制。


    =VLOOKUP($V4,$A$5:$T$9,MATCH(W$3&"*",$A$3:$T$3,0))


原理分析

VLOOKUP函数这里返回的列数不确定，是根据等级来判断的，如A等级就返回区域第9列，B等级就返回区域第13列，C等级就返回区域第17列。如果等级很少的情况下，也可以逐一写公式。


    =VLOOKUP($V4,$A$5:$T$9,9)
    =VLOOKUP($V4,$A$5:$T$9,13)
    =VLOOKUP($V4,$A$5:$T$9,17)


不过用人工数列数，需要逐个更改公式，难免会出差错。这时MATCH函数就派上用场了，这个函数是返回某个等级在数据源的排位。

MATCH函数语法：


    =MATCH(要查找的值,上哪里去找,匹配方式（模糊/精确）)


这个函数类似于VLOOKUP函数，只不过MATCH函数返回的是排位，而不是对应值。

如现在要查找等级在区域中的排位，C就是第2位。


    =MATCH(C2,$A$2:$A$5,0)

这样就不用去数第几列，即使数据源的顺序改变，也可以得到正确值。

回到原公式，W$3&"*"后面的星号（*）其实就是通配符，代表所有字符，也就是代表A开头的所有字符。

如果是以A结尾的，就用"*"&W$3

如果是包含A，就用"*"&W$3&"*"。

A等级在AQL基准表是这样表示——“A·重欠点AQL=1.0”，其实就是A开头的。

说明：这里的W$3，对应的单元格的值为A。

知识扩展


Q：
 通配符除了星号（*）之外，还有其他吗？如果有，具体怎么用？


A：
 通配符还有问号（?），这个代表一个字符。

下面举2个例子来说明一下：

例1　统计销售店铺为3个字符的销售收入


    =SUMIF(B:B,"???",D:D)


例2　产品编码倒数第2位是C的销售收入


    =SUMIF(C:C,"*C?",D:D)



 11.9　供应商分类管理


案例：
 供应商多了，就得分类管理。根据供应商的品质进行划分，设定一套符合供应商的检查基准。

1．根据供应商，依次查找到地区、城市、检查基准。

在H4单元格中输入以下公式，向下和向右复制。


    =VLOOKUP($G4,$B$3:$E$23,COLUMN(B1),0)


2．根据检查基准，获取所有对应的供应商。

在H12单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束，向下复制。


    =IFERROR(INDEX(B:B,SMALL(IF($G$12=$E$4:$E$23,ROW($4:$23)),ROW (A1))),"")


原理分析

ROW、COLUMN函数是获取行列数用的，向下生成序号用ROW函数，向右生成用COLUMN函数。

VLOOKUP函数的第三参数是依次向右生成序号，所以用COLUMN函数。

将重点放在第二条公式上，这个难度比较大。当公式比较长时，一般都是借助【F9】键进行理解的，哪里不会抹黑哪里，公式剖析So Easy！

将鼠标放在编辑栏，选择$G$12=$E$4:$E$23，会出现黑色背景色，这个俗称抹黑。

按【F9】键出现由FALSE跟TRUE组合的一个结果，满足条件的显示TRUE，不满足条件的显示FALSE。

按【Ctrl】+【Z】键取消，要不然公式就变了，然后选择IF($G$12=$E$4:$E$23, ROW($4:$23))抹黑。

按【F9】键满足条件的显示本身行号，不满足条件的显示FALSE。

再按【Ctrl】+【Z】取消，然后选择SMALL(IF($G$12=$E$4:$E$23,ROW($4:$2&nbsp;3)),ROW(A1))抹黑。

按【F9】键得到5，也就是最小的序号。

再按【Ctrl】+【Z】键取消，然后选择INDEX(B:B,SMALL(IF($G$12=$E$4:$E$23,RO W($4:$23)),ROW(A1)))抹黑。

按【F9】键，得到南海建泰，也就是序号5对应的值。

这时大概知道了这条长公式的运算过程，按【Ctrl】+【Z】取消。接着继续测试其他单元格的公式，对于长公式而言，不要奢望一次就能了解整个公式的含义，需要不断地配合【Ctrl】+【Z】和【F9】键多次测试才可以。

单击H16单元格，在编辑栏选择跟第一个单元格不同的地方，也就是包含ROW(A5)这部分的公式，测试就要抹黑不同的。将SMALL(IF($G$12=$E$4:$E$23,ROW($4:$23)),ROW(A5))抹黑。

按【F9】键得到20。刚才的IF函数抹黑后去除掉FALSE就是{5;13;19;20;21;22}，20就是排在这组数字的第5位，也就是说SMALL是依次获取前N位的对应值。

到这里这条公式的意思就基本明白了，如果还不明白，可以借助另外一种方法：在单元格输入公式，这样会比用【F9】键相对直观一点。

如现在不明白IF($G$12=$E$4:$E$23,ROW($4:$23))这一段的意思，复制出来，然后选择区域N4:N23单元格，这个区域的大小要跟ROW($4:$23)一样大，否则会出错。因为是数组公式，输入公式后按【Ctrl】+【Shift】+【Enter】三键结束。

剖析公式最常用就是这两种办法，根据自己的实际情况进行选择。

知识扩展


Q：
 什么是数组公式？为什么没有输入【Ctrl】+【Shift】+【Enter】三键会得不到正确值？


A：
 假如Excel划分成九级，你学好普通公式最多也就到第三级，而数组公式即使再差也算第四级。也就是说数组公式比普通公式高级。

什么是数组？

如四大名著是一个整体，“西游记”就是其中一个组。只要笔者一点“西游记”，就会一次性出现：唐僧、孙悟空、猪八戒、沙僧。再如点“三国演义”，里面的四个人也会一起出现，而不用逐个点名，这就是跟普通公式的差异。

一句话：数组一次搞定，普通公式多次搞定。

注：“西游记”等为实现定义好的名称。

什么是数组公式？

数组公式可以执行多项计算并返回一个或多个结果。数组公式必须按【Ctrl】+【Shift】+【Enter】三键结束，在输入数组公式时，Excel 会自动在大括号 { } 之间插入该公式。目的是告诉Excel，我不是普通人，要对我特别招待，一副盛气凌人的样子。

一般情况下，能力跟脾气是成正比的，能力越强，脾气越大。正因为数组公式能力比普通公式强，所以脾气大点，动不动就耍脾气。

需谨记于心的内容：

你不是说一喊“西游记”就所有人出来吗，怎么只出来一个唐僧，其他人跑哪去了？

你把我当成普通人，当然不理你，选择区域，记得要按【Ctrl】+【Shift】+【Enter】三键结束。

注意
 　编辑栏的变化，三键结束就会自动生成{}，直接按回车键的话就没有。

“西游记”只有四人，如果区域多选的话也会出错。

“西游记”四人是一个整体，不能更改其中一个，要改就得全部改。

在Excel2013版中，连让你回车都不允许，在其他版本会显示“不能更改数组的某一部分”。可以点击“×”按钮，然后选择整个区域，删除（修改）公式。

单击包含多单元格数组的任意单元格，使用【Ctrl】+【/】快捷键就可以快速选中整个数组区域。

只要你摸清数组公式的脾气，以后注意点，就没事。



第12章　辅助图表展现

大部分的图表并不能直接根据明细表来绘制，而需要经过函数与公式进行处理，才可以创建图表。这一节就介绍函数与公式如何辅助图表，为创建图表提供合适的数据源。


 12.1　产品不良柏拉图

什么是柏拉图（排列图）

根据搜集到的数据，按不良原因、不良状况、不良发生位置等不同区分标准，以寻求占最大比率的原因、状况或位置的一种图形。柏拉图又叫排列图。它是将质量改进项目从最重要到最次要顺序排列而采用的一种图表。柏拉图由一个横坐标、两个纵坐标、几个按高低顺序（“其他”项例外）排列的矩形和一条帕累托曲线组成。

柏拉图（排列图）格式。

柏拉图的主要用途

1．按重要顺序显示出每个质量改进项目对整个质量问题的作用。

2．识别进行质量改进的机会。(即识别对质量问题最有影响的因素,并加以确认)

以上是对品质管理工具之一的柏拉图的介绍，其中柏拉图遵循二八原则，即：


	80%的问题由20%的原因引起；

	80%的索赔发生在20%的生产线上；

	80%的销售额由20%的产品带来；

	80%的品质成本由20%的品质问题造成；

	80%的品质问题由20%的人员引起。



柏拉图是每个品质人员必须掌握的一种手法，非常重要。

现在要根据产品不良的汇总数据，制作柏拉图。

制作产品不良柏拉图，首先需要对不良数进行降序排序，根据不良数获取影响度、累积影响度等。

1．获取影响度，即每项不良数占总不良数的比例。

在D3单元格中输入以下公式，并向下复制。


    =C4/SUM($C$3:$C$9)


2．获取累积影响度，即每个项目的影响度累加。

在E3单元格中输入以下公式，并向下复制。


    =SUM($D$3:D3)


将D列和E列的单元格格式设置为百分比格式，同时将D3:E3的字体改成白色。

3．标示第一个大于80%的项目。

在F3单元格中输入以下公式，并向下复制。


    =IF(ROW()-2=MATCH(0.8,$E$3:$E$9)+1,E3,NA())


原理分析

前面两条公式很好理解，下面将重点放在最后一条公式上。

用0.8在A列进行模糊查找，返回小于或等于0.8的最大值的排位。这里的0.67刚好是小于0.8的最大值，排在第3位。


    =MATCH(0.8,A:A)


而现在我们需要的是大于0.8的第一个值，也就是MATCH排位+1。


    =MATCH(0.8,A:A)+1
    =IF(ROW()-2=MATCH(0.8,$E$3:$E$9)+1,E3,NA())


这条公式的含义就是让等于第一个大于0.8的值显示本身，其他显示错误值#N/A，在做图表的时候经常会借助错误值#N/A。

知识扩展


Q：
 Excel中还有哪些常见错误，可以取代#N/A这种制作图表吗？


A：
 Excel常用错误及解决方案：

1．出现错误值#####！

导致这种错误的最常见原因是输入到单元格中的数值太长或公式产生的结果太长，导致单元格容纳不下。可以通过修改列的宽度来解决此问题。另外，对日期或时间做减法时若产生了负值，Excel也会在整个单元格中显示#####！。

2．出现错误值#VAUE！

这种错误是因为使用了错误的参数或运算对象类型。比如在需要输入数字或逻辑值时，却输入了文本；在需要赋予单一数值的运算符或函数时，却赋予一个数值区域。解决方法分别是：确认运算符或参数正确，且公式引用的单元格中包含有效数值；将数值区域改为单一数值。

3．出现错误值#DIV/0！

当除数“0”或引用了空单元格时(Excel通常将空单元格解释为“0”)，会出现此种错误。请确定函数或公式中的除数不为“0”且不是空值。

4．出现错误值#NAME？

当Excel不能识别公式中使用的文本时，就出现错误值"#NAME？"。向公式中输入文本时，要将文本括在""中，否则Excel会将其解释为名称，导致出错。另外，公式中使用的名称已经被删除或使用了不存在的名称以及名称拼写错误，也能产生这种错误值。请确认公式中使用的名称存在并且是正确的。

5．出现错误值#N/A

这种错误产生的原因是函数或公式中没有可用的数值。解决方法是在没有数值的单元格中输入"#N/A"，这样，公式在引用这些单元格时，将不进行数值计算，而是直接返回"#N/A"，从而避免了错误的产生。

6．出现错误值#REF！

当引用的单元格无效时会产生这种错误。请确认所引用的单元格是否存在。

7．出现错误值#NUM！

产生这种错误的原因是函数或公式中的数字有问题。比如函数中使用了不正确的参数类型；公式产生的数字太大或太小等。请检查函数中使用的参数类型是否正确，或修改公式使其结果能让Excel正确表示。

用其他错误值取代#N/A做出来的图表很难看。

 12.2　产品使用寿命直方图


案例：
 直方图(Histogram)又称质量分布图。它是一种统计报告图，由一系列高度不等的纵向条纹或线段表示数据分布的情况。一般用横轴表示数据类型，纵轴表示分布情况。

现在要根据产品使用寿命制作直方图。

制作直方图是一个很烦琐的过程，需要准备很多辅助的计算公式。

STEP 01
 　计算最大值、最小值、极差、分组数、分组组距。


    =MAX(A2:A101)
    =MIN(A2:A101)
    =D1-D2
    =R O U N D U P ( S Q R T ( C O U N T ( A 2:A101)),0)
    =ROUND(D3/D4,0)

STEP 02
 　根据计算出来的结果就可以设置分组区间和分组数。再计算频率和正态分布曲线。

频率公式是数组公式，选择I2:I12单元格，

输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =FREQUENCY(A2:A101,H2:H12)


正态分布曲线，在J2单元格输入以下公式，并下拉复制。


    =NORMDIST(H2,AVERAGE($A$2:$A$101),STDEV($A$2:$A$101),0)

原理分析

这里出现了大量数学和统计函数，其中最大值MAX、最小值MIN、平均值AVERAGE、计数COUNT，这4个函数跟SUM语法基本一样，只要知道其中一个的用法，其他函数就会用。

ROUND跟ROUNDUP这两个函数语法也一样，ROUND是四舍五入，而ROUNDUP是向上进位。如对12.2进行舍进，ROUND得到12，而ROUNDUP得到13，也就是不管小数点是什么都向上进一位。


    =ROUND(12.2,0)
    =ROUNDUP(12.2,0)


NORMDIST跟STDEV这两个函数如果不是专业人士，可以不用掌握，毕竟使用频率非常低。

FREQUENCY函数语法：

=FREQUENCY(数据源,分段点)

结果：

以分段点为间隔，统计数据源值在各段出现的频数。

其中：

数据源：为对一行/一列单元格或一个连续的单元格区域的引用。

分段点：为对一行/一列单元格或一个连续的单元格区域的引用。

数据引用支持跨工作表、工作簿。

公式输入方法：

以多单元格数组方式输入，且必须是纵向数组。

为了更好地理解这个函数，我们将数据源区域变小，进行统计各区间的个数。

选择C2:C5单元格输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =FREQUENCY(A2:A10, B2:B4)

统计4个区间，就等同于用COUNTIF和COUNTIFS两个函数写4条公式，它是非常有用的一个函数。


    =COUNTIF(A2:A10,"<=70")
    =COUNTIFS(A2:A10,">=71",A2:A10,"<=79")
    =COUNTIFS(A2:A10,">=80",A2:A10,"<=89")
    =COUNTIF(A2:A10,">=90")


知识扩展


Q：
 FREQUENCY函数是否可以按条件统计各区间的个数，如统计1班在各区间的分数？COUNTIFS函数虽然可以实现，但过于烦琐，而且容易写错。


    =COUNTIFS(A2:A15,"1班",C2:C15,"<=70")
    =COUNTIFS(A2:A15,"1班",C2:C15,">=71",C2:C15,"<=79")
    =COUNTIFS(A2:A15,"1班",C2:C15,">=80",C2:C15,"<=89")
    =COUNTIFS(A2:A15,"1班",C2:C15,">=90")

A：
 其实FREQUENCY函数也可以实现，只需要增加一个条件判断，让符合1班的人返回本身的成绩。

在E2:E5单元格中输入以下公式，按【Ctrl】+【Shift】+【Enter】三键结束。


    =FREQUENCY(IF(A2:A15="1班",C2:C15),D2:D4)


FREQUENCY函数忽略空白跟文本，用IF进行判断，符合条件的显示成绩，否则显示FALSE，FALSE自动被忽略。

 12.3　供应商出货数量动态图


案例：
 交互式图表，很多人称为“动态图表”，它能进行简单的交互功能，但并不是真正意义上的动态。

原理：图表的表现依赖于数据，数据变化，图表就会随之变化。所以图表的动态变化不是变动图表本身，而是变动数据源。

现在要制作反映单个供应商的每个月出货数量动态图，数据源的供应商一改变，图表也跟着改变。

根据数据源重新构造数据区域，供应商设置“数据验证”（低版本叫数据有效性）→“序列”，就可以通过下拉选择供应商。在B13输入以下公式，并向右复制。


    =VLOOKUP($A13,$A$3:$E$8,COLUMN(),0)

原理分析

VLOOKUP函数语法：


    =VLOOKUP(要查找的值, 到哪里去找, 带回什么,匹配方式（精确/模糊）)


现在是要查找供应商的数据源的1～4月份的出货数，也就是要精确查找，就是第四参数为0。1～4月份依次是2～5列，通过COLUMN函数可以获取列号。正常情况下是用COLUMN(B1)带有参数，但这里刚好要返回的列号跟数据源一样，所以参数可以省略不写。

知识扩展


Q：
 除了数据有效性+VLOOKUP函数这种方法，常见的还有哪些制作动态图表的方法呢？


A：
 常用的动态图表法：直接筛选法、数据透视表+切片器、OFFSET函数+控件。这里说一下OFFSET函数+控件制作动态图表的方法。

创建控件，让A11单元格跟控件关联，在A13单元格输入以下公式，向右复制。


    =OFFSET($A$2,$A$11,COLUMN()-1)

OFFSET函数语法：


    =OFFSET(指定单元格为基准,下移多少行,右移多少列,引用多少行,引用多少列）


参数4、参数5可选，现在就不管这两个。

要从1到11这个位置，需要下移2行，右移2列。


    =OFFSET(A1,2,2)

再如从11到23，需要下移3行，右移0列。


    =OFFSET(C3,3,0)

回到原来公式，现在是5，也就是说A2单元格下移5列就得到建泰，跟控件显示的一样，建泰属于供应商，也就是不用向右偏移，就是0。COLUMN()-1在A列就等于0，其他依次类推。


    =OFFSET($A$2,$A$11,COLUMN()-1)




第13章　辅助数据透视表分析

Excel数据分析一般会借助排序、筛选等技巧，但更多的时候是借助数据透视表，数据透视表是非常强大的数据分析工具。但单纯靠数据透视表，也有一些特殊情况还是处理不了，这时就是函数与公式派上用处的时候了。


 13.1　统计每个等级的次数


案例：
 为出货数划分等级，如出货数>500，就显示很多，出货数在200~500之间显示多，出货数小于200显示少，根据划分的结果统计各等级的次数。

划分等级可以借助数据透视表的手工组合，但没有使用公式方便。

在F列增加一列辅助列，在F2单元格中输入以下公式，并向下复制公式。


    =IF(D2>500,"很多",IF(D2>=200,"多","少"))

原理分析

IF函数语法：


    =IF(条件判断,如果条件成立的值,反之（即条件不成立）的值)


IF函数的嵌套，其实第2个IF函数可以先不要管，然后看成这样：D2>500这个条件，就返回很多，否则返回公式。


    =IF(D2>500,"很多",公式)


如果返回公式，再对公式里面进行判断。如果D2>=200，就返回多，否则返回少。


    IF(D2>=200,"多","少")


通过拆分，就很轻松地理解这个公式了。

知识扩展


Q：
 IF函数在进行判断时，如果条件太多，公式容易写错，有没有函数可以取得IF函数呢？


A：
 一般条件超过3个，笔者都是采用别的函数取代IF，如用VLOOKUP函数。

在B2单元格中输入以下公式，并向下复制。


    =VLOOKUP(A2,$D$2:$F$9,3)


其实就是借助VLOOKUP函数的模糊查找，返回区间的对应值。


 13.2　统计每种颜色的出货数


案例：
 在笔者公司番号的第5位代表颜色，也就是最后一位。那现在笔者要统计每个颜色的总出货数怎么办呢？

在F列建立辅助列，在F2单元格中输入以下公式，并向下复制。


    =RIGHT(C2,1)

原理分析

RIGHT函数语法：


    =RIGHT(字符串,N)


意思就是从字符串中，从右边提取N位。最后一位数是番号，也就是从右边提取1位。

知识扩展


Q：
 RIGHT函数为什么第二参数可以省略，只写字符串呢？


A：
 默认情况下，如果第二参数省略，就是提取一位字符。


    =RIGHT(A2)

像LEFT函数也一样，如果要提取左边一位字符，可用：


    =LEFT(A2)



 13.3　统计每个月的检查时间

笔者公司算工资的时候是从每个月的26号开始算，比如2月26日到3月25日就算3月份。如何统计每个月份的出货检查时间呢？

在F列创建一列辅助列，在F2单元格中输入以下公式，并向下复制。


    =IF(MONTH(A2)+IF(DAY(A2)>25,1,0)>12,1,MONTH(A2)+IF(DAY (A2)>25,1,0))

原理分析

MONTH函数是获取月份，IF函数是判断日是否大于25，如果大于25就加1。也就达到2月26日显示3月份的效果。


    =MONTH(A2)+IF(DAY(A2)>25,1,0)


但这条公式存在着小小的缺陷，因为月份只有12个月，当出现12月26日时，会显示13，这明显就不符合实际情况。

这时就要再加一个判断，让大于12的显示1，其他不变。


    =IF(公式>12,1,公式)


将公式代替进去就成了：


    =IF(MONTH(A2)+IF(DAY(A2)>25,1,0)>12,1,MONTH(A2)+IF(DAY (A2)>25,1,0))


知识扩展


Q：
 这里只有13这种情况，假如有14，15，…，怎么转变成1～12月份呢？


A：
 其实这相当于月份除以12，取余数一样。

13除以12的余数为1。

14除以12的余数为2。

……

23除以12的余数为11。

24除以12的余数为0。

这里除了12的整数倍的余数为0外，其他都可以实现转换，我们可以将24减去1，取余数11，然后再加1，这样就可以转变成12了。

Excel中获取余数的函数为MOD。

MOD函数语法：


    =MOD(被除数,除数)


综合上面的思路，可设置公式为：


    =MOD(A2-1,12)+1

其实如果是高版本的话，可以用下面的公式获取月份：


    =MONTH(EDATE(A2,IF(DAY(A2)>25,1,0)))




第四篇　解疑难

笔者早期基本上都是在各大Excel论坛、Excel交流群免费帮人解答疑难。后来听从一个朋友的建议，在淘宝上开了一家帮人解答疑难的店铺。在这期间，我发现一个现象，付费的人普遍比以前那些人更有礼貌。就因为这个原因，笔者慢慢喜欢上在淘宝上帮人解答问题，做自己喜欢的事，还能挣点零花钱，何乐而不为。下文挑选了一些笔者在淘宝帮人解答的问题，进行讲解。



第14章　表格模板设计


 14.1　动态汇总每个区域每个月的收入


案例：
 现在有4个明细表，表格的格式完全一样，只是行数不一样而已，分别是成都、深圳、海口、武汉，在最后的汇总表对这4个明细表进行统计。原先是先用公式在明细表统计好，再引用过去。

统计C列的性别，男女人数分别多少：


    =COUNTIF(C3:C40,"男")
    =COUNTIF(C3:C40,"女")


统计F列客户状态，新客户跟旧客户人数分别为多少：


    =COUNTIF(F3:F40,"新客户")
    =COUNTIF(F3:F40,"旧客户")


统计各种金额，消费金额、应收、扣款项目、实收：


    =SUM(G3:G40)
    =SUM(J3:J40)
    =SUM(K3:K40)
    =SUM(L3:L40)


4个明细表设置的公式都一样，只是区域不一样而已。

分别统计好后，就在汇总表依次引用，如C3单元格的公式为：


    =成都!B41


这样做会出现2个问题：

1．明细表的区域限制死了，如果后续有数据添加，不能自动统计。

2．这种统计只是针对1个月份，如果有多个月份也不能统计。

看到这里，笔者大概就有了一个思路，就问lee：你每个明细表下面的统计能否删除掉？

lee：这个是为了方便统计用的，如果有更好的方法，可以删除。

得到了一个肯定的说法后，笔者就对这个表格进行了小小的变动。

将明细表的各种统计删除。

在日期后面插入一列，获取月份，如下所示。


    =MONTH(A3)&"月"

用同样的方法，将其他明细表也这样操作。

为了方便测试，笔者在成都这个表添加了几行2月份的数据。

同时也在汇总表添加了2月份的区域，并将原来的公式删除。

男性、女性、新客户、旧客户人数所使用的公式一样，只是区域跟条件略作更改而已。


    =COUNTIFS(INDIRECT(B3&"!d:d"),"男",INDIRECT(B3&"!b:b"),A3)
    =COUNTIFS(INDIRECT(B3&"!d:d"),"女",INDIRECT(B3&"!b:b"),A3)
    =COUNTIFS(INDIRECT(B3&"!g:g"),"新客户",INDIRECT(B3&"!b:b"),A3)
    =COUNTIFS(INDIRECT(B3&"!g:g"),"旧客户",INDIRECT(B3&"!b:b"),A3)


消费金额、应收、扣款项目所使用的公式一样，只是区域跟条件略做更改而已。


    =SUMIF(INDIRECT(B3&"!b:b"),A3,INDIRECT(B3&"!h:h"))
    =SUMIF(INDIRECT(B3&"!b:b"),A3,INDIRECT(B3&"!k:k"))
    =SUMIF(INDIRECT(B3&"!b:b"),A3,INDIRECT(B3&"!l:l"))


实收：


    =H3-K3


尾款：


    =H3-I3-K3


本来做完后笔者就只是把最终的附件发给lee，但lee实在水平有限，典型的菜鸟，看了以后一头雾水，啥都不懂，一定要笔者给她解释。为了解释这2条公式，笔者足足花了一个小时。


lee：
 我的本和笔都准备好了。


卢子：



    =COUNTIF(C3:C40,"男")
    =COUNTIF(区域,条件)


这个语法能看懂吗？

因为你原来区域是固定的C3:C40单元格，现在你区域不固定，也就是区域要大一点，那你干脆就引用整列，写C:C。

就是统计C列性别为男的次数，像其他统计新旧客户人数的也是一样的道理。

现在进入核心部分，因为有很多表，不可能每个表单独设置公式。


lee：
 嗯。


卢子：
 你当然希望同一列下拉就全部表格汇总出来。


lee：
 对的。


卢子：


正常的话，成都就用=COUNTIF(成都!C:C,"男")

下面的深圳就改成=COUNTIF(深圳!C:C,"男")

这样好麻烦。


lee：
 是，要是有100张那要改100次。


卢子：


你看，这样下拉就全部出来了：


    =B3&"!c:c"


所有表格的区域都出来了，对吧？

lee：
 慢点，出来区域和后面汇总分表数据啥关系？


卢子：
 要制作区域出来，这个能理解吧。如统计成都就用=COUNTIF(成都!C:C,"男")。


lee：
 你的意思是不是，汇总表区域下面城市就用，=B3&"!c:c"这个公式来设置。


lee：
 这个公式翻译成汉语是啥意思？


卢子：
 刚刚那个b3&"!c:c"就是把内容合并起来而已，就是在区域后面增加!c:c，成都!c:c这样。


lee：
 喔，懂了。


卢子：
 因为我们是通过&连接起来的，不是直接用区域生成的，所以需要增加一个函数INDIRECT，才能进行计算：


    =COUNTIF(INDIRECT(B3&"!c:c"),"男")


也就是这条公式的由来。


lee：
 INDIRECT这个函数是啥意思？


卢子：
 因为你不是直接引用，是用公式生成，所以在那里增加这个函数。


lee：
 直接引用和公式生成哪个方便点啊？


卢子：
 刚刚不是解释了吗，如果有100张表，你就得直接引用100次。


lee：
 喔喔喔喔，明白了。

大神，前面分表是一个月的，总表汇是一个月，那如果有多个月，需要怎么做呢？


卢子：
 那就用COUNTIFS函数，COUNTIF函数是单条件计数，COUNTIFS函数是多条件计数。

COUNTIFS函数语法：


    COUNTIFS(条件区域1,条件1, 条件区域2,条件2……)


刚刚我在你的表格增加了一列计算月份，这样有了前面COUNTIF函数的基础，设置公式就变得很简单。两个函数的语法基本一样，只是COUNTIFS可以多条件计数而已。


    =C O U N T I F S ( I N D I R E C T ( B 3&"! d:d") ,"男",INDIRECT(B3&"!b:b"),A3)


男女人数跟新老客户用的公式都一样，其他只要更改区域就可以。

前面都是计数，后面的是求和。单条件求和用SUMIF函数，多条件求和用SUMIFS函数。现在都是单条件求和，那就用SUMIF函数，语法如下：

SUMIF(条件区域,条件,求和区域)

消费金额的公式为：


    =SUMIF(INDIRECT(B3&"!b:b"),A3,INDIRECT(B3&"!h:h"))


其他的也相应改变区域跟条件即可。大概就这样，你自己好好理解下。最终效果如图：

[image: ]



lee：
 不打扰你了，我自己好好琢磨下。

知识扩展

正常表格模板都是要发送给其他人填写，为了防止其他人不小心删除或者修改公式导致出错，最好对表格中的公式进行保护。


STEP 01
 　单击全选，借助【Ctrl】+【1】快捷键，调出“设置单元格格式”对话框，单击“保护”，取消“锁定”的勾选，单击“确定”按钮。

STEP 02
 　选择公式区域，借助【Ctrl】+【1】快捷键，调出“设置单元格格式”对话框，单击“保护”，勾选“锁定”跟“隐藏”，单击“确定”按钮。

STEP 03
 　单击“审阅”→“保护工作表”，输入密码，这里以不输入密码为例，在下面的“允许此工作表的所有用户进行操作”，允许的内容就在前面勾选，勾选完单击“确定”按钮。

设置保护后，工作表的公式是看不到的，同时也不允许修改，一旦修改就会弹出警告对话框。

以后如果自己要重新更改公式，就“撤销工作表保护”就可以。

 14.2　提取42个表格名称并汇总数量


案例：



帕萨特鲁byc076：
 能帮做表格汇总吗？


卢子：
 怎么汇总法？


帕萨特鲁byc076：
 应该一共有42个工作表，我想把所有的工作表的B和C列统计出来。

卢子：
 货号这列有没有一个现成的总表呢？


帕萨特鲁byc076：
 什么意思？


卢子：
 就是把货号这些全部列举出来。


帕萨特鲁byc076：
 这个样子？


卢子：
 是，有的话我就用公式，不用VBA。那你把这个总货号给我，我就省了一个步骤提取所有货号。

帕萨特鲁byc076随后就把相关的表格发送过来，卢子就开始设置表格了。

STEP 01
 　按【Ctrl】+【F3】键调出定义名称对话框，或者单击“公式”→“定义名称”调出定义名称对话框。

STEP 02
 　在“名称”框里输入一个定义名称“表名”，方便在工作表中引用。在“引用位置”处输入=GET.WORKBOOK(1)，设置完毕后单击“确定”按钮。

STEP 03
 　在B1输入以下公式，这时就返回了包括工作簿名称的工作表名称。


    =INDEX(表名,COLUMN())

而我们现在需要的是工作表名称，也就是“]”后面的内容，下面需要对公式略作更改。

输入以下公式，向右拉，直到所有工作表名称出现为止。


    =MID(INDEX(表名,COLUMN()),FIND("]",INDEX(表名,COLUMN()))+1,9)

STEP 04
 　在B2单元格中输入以下公式，并向右和向下复制公式。


    =SUMIF(INDIRECT(B$1&"!B:B"),$A2,INDIRECT(B$1&"!C:C"))

在保存的时候会弹出一个警告对话框，单击“否”。

将工作簿另存为启用宏的工作簿。

表格做完后，卢子就发送给帕萨特鲁byc076核对。

知识扩展

提取工作表名称用宏表函数有点麻烦，需要定义名称才能使用，一般笔者都是采用VBA生成的。

按【Alt】+【F11】快捷键，插入一个模块，输入代码后，单击运行，所有工作表就自动生成了。


    Sub 表名()
    For i = 2 To Sheets.Count
    Cells(1, i) = Sheets(i).Name
    Next
    End Sub


[image: ]




第15章　函数的瓶颈


 15.1　多个工作簿合并汇总


案例：
 现在有6～11月的表格明细表，1个月份有1张工作簿，表格格式一样，现在要汇总每个商户名称每个月份的交易金额。

[image: ]


要将6个工作簿的商户名称合并后并提取不重复，用函数基本上是做不到的。对于个多工作簿，卢子习惯性会找好友无言的人帮忙，因为这个需要借助VBA合并。

无言的人提供了多个工作簿的第一个工作表合并代码：


    Sub 合并()
    Dim MyPath$, MyName$, sh As Worksheet, sht As Worksheet
    Dim MyRcou As Integer, AbRcou As Integer
    Application.ScreenUpdating = False
    Set sh = ThisWorkbook.Sheets(1)
    MyPath = ThisWorkbook.Path & "\"
    MyName = Dir(MyPath & "*.xls")
    [a1].CurrentRegion.Offset(1).Clear
    Do While MyName <> ""
    If MyName <> ThisWorkbook.Name Then
    With GetObject(MyPath & MyName)
        AbRcou = sh.Cells(Rows.Count, 1).End(xlUp).Row
        Workbooks(MyName).Worksheets(1).[a1]. CurrentRegion.Offset(IIf(AbRcou = 1, 0, 1)).Copy sh.Cells(1, 1).Offset(IIf(AbRcou = 1, 0, AbRcou))
        .Close False
    End With
    End If
    MyName = Dir
    Loop
    Application.ScreenUpdating = True
    MsgBox "ok"
    End Sub


在模块中粘贴以上代码，单击“运行”。

这个工作簿要跟前面的6个工作簿在同一个文件夹才可以运行。

[image: ]


运行后所有工作簿就瞬间合成在一张工作簿中，生成后单击“确定”。

用VBA生成的数据顺序与实际略有差异，现在是10月份排在最前面。不过这个对数据统计没有任何影响。

统计数据在这里就不采用函数了，用数据透视表。


STEP 01
 　单击数据源任意单元格如A1，“插入”→“数据透视表”，弹出“创建数据透视表”对话框，因为区域是连续的，Excel会帮你选好区域，保持默认不变，单击“确定”。

STEP 02
 　将商品名称拉到行字段，月份拉到列字段，交易金额拉到值字段。

经过简单的操作，就轻松实现了汇总。

函数虽然很强大，但合理地借助VBA或者数据透视表处理问题，会使问题变得更简单，不要执着于函数。


 15.2　几十万行数据的查询


案例：
 有A、B两个表格，A表有4万行数据，B表有35万行数据，现在要根据B表的客户编号在A表查询对应值，并将有对应值的项目放在另外一个表格中。

这个问题看起来挺简单的，就是用VLOOKUP函数查询对应值，然后筛选非错误值，复制到新表格。笔者尝试了输入公式，跟往常一样向右复制，双击向下填充公式，结果傻眼了：表格基本属于死机的状态，一直显示正在计算。到这里只能强制关闭表格。

grace52545：
 我自己就电脑不行，下午已经匹配出了，保存不了，太大。


卢子：
 我刚刚尝试了用公式，根本操作不了，看来需要借助Access才行，我先导入后再处理。


grace52545：
 麻烦你了。

新建一个Access文件，打开Access后，单击“外部数据”，导入并链接选择Excel，在弹出的“获取外部数据”的对话框中，浏览到指定的文件位置，单击“确定”按钮。

弹出“导入数据表向导”对话框，保持默认不变，直接单击“下一步”。

勾选第一行包含列标题，单击“下一步”。

不做任何处理，直接单击“下一步”。

[image: ]


不做任何处理，直接单击“完成”。

导入后在左边会出现一个Sheet的表，导入的步骤虽然烦琐，但并不难，直接根据向导就能很轻松导入。至于数据导出就简单很多，直接选择按Excel导出就行。

选择存储位置，单击“确定”。

知道了导入跟导出的方法后，其他的Excel表格也就可以参照这个方法操作。

将2个表格导入Access中后就开始进行查询操作。

单击“创建”选项卡的“查询设计”。

单击Sheet添加，Sheet2添加，添加后左边就出现相应的表格及字段名字，最后单击“关闭”。

用鼠标将2个表的客户编号连起来，添加要显示的字段，*就是代表所有，因为是2个表格的所有字段都要显示，单击“运行”。

瞬间就产生了结果，这个结果是2个表格共同有的客户编号。

因为我们现在需要的是B表（Sheet2）的所有记录，这里需要做一个小小的变动。

右键单击“查询1”，然后单击“SQL试图”。

这时会看到一段SQL语句，里面有一个关键词：INNER JOIN，这个就是代表2个表共同有的。如果查询不到对应值还要显示出来右边的表格所有记录，就用RIGHT JOIN，相反，如果是左边的就用LEFT JOIN。

这里因为要显示右边的表Sheet2的所有记录，所以改成RIGHT JOIN。修改后单击“运行”。

这样有对应值的就会显示相应的值，没有的就显示空白。

查询后就保存起来。

Access看起来好复杂的样子。其实在Access中，我们常用的功能就是查询，而查询大部分功能都可以通过基础操作生成SQL语句。对于一个新软件的基础操作，多操作几次就熟悉了，这需要一个适应过程，就像当初学Excel一样。


Excel疑难妙解（函数版）：http://pan.baidu.com/s/1bp3tOJX

